

Learn Real World Haskell

Jacob Bishop

2023-08-20

1 Getting started

略

2 Types and functions

略

3 Defining types, streamlining functions

略

4 Functional programming

略

5 Writing a library: working with JSON data

在 Haskell 中表示 JSON

首先是在 Haskell 中定义 JSON 这个数据，这里使用代数数据类型来表达 JSON 类型的范围。

```
1 data JValue
2   = JString String
3   | JNumber Double
4   | JBool Bool
5   | JNull
6   | JObject [(String, JValue)]
7   | JArray [JValue]
8   deriving (Eq, Ord, Show)
```

对于每种 JSON 类型，我们都提供了独立的值构造函数。测试：

```
1 ghci> :l SimpleJSON
2 [1 of 1] Compiling Main             ( SimpleJSON.hs, interpreted )
3 Ok, one module loaded.
4 ghci> JString "foo"
5 JString "foo"
6 ghci> JNumber 2.7
7 JNumber 2.7
8 ghci> :type JBool True
9 JBool True :: JValue
```

构造一个从 `JValue` 获取字符串的函数：

```
1 getString :: JValue -> Maybe String
2 getString (JString s) = Just s
3 getString _ = Nothing
```

测试：

```
1 ghci> :reload
2 [1 of 1] Compiling Main             ( SimpleJSON.hs, interpreted )
3 Ok, one module loaded.
4 ghci> getString (JString "hello")
5 Just "hello"
6 ghci> getString (JNumber 3)
7 Nothing
```

接下来是其它类型的访问函数：

```
1 getInt (JNumber n) = Just n
2 getInt _ = Nothing
3
4 getDouble (JNumber n) = Just n
5 getDouble _ = Nothing
```

```
6
7  getBool (JBool n) = Just n
8  getBool _ = Nothing
9
10 getObject (JObject o) = Just o
11 getObject _ = Nothing
12
13 getArray (JArray a) = Just a
14 getArray _ = Nothing
15
16 isNull v = v == JNull
```

`truncate` 函数可以让浮点类型或者有理数去掉小数点后变为整数：

```
1 ghci> truncate 5.8
2 5
3 ghci> :module +Data.Ratio
4 ghci> truncate (22 % 7)
5 3
```

Haskell 模块详解

一个 Haskell 源文件包含了单个模块的定义。模块允许我们在它其内部进行定义，并由其它模块访问：

```
1 module SimpleJSON
2   ( JValue (..),
3     getString,
4     getInt,
5     getDouble,
6     getBool,
7     getObject,
8     getArray,
9     isNull,
10  )
11 where
```

如果省略了导出（即圆括号以及其所包含的名称），那么该模块中的所有名称都会被导出。

编译 Haskell 源

编译一个源文件：

```
1 ghc -c SimpleJSON.hs
```

`-c` 选项告诉 `ghc` 仅生成对象代码。如果省略了该选项，那么编译器则会尝试生成一整个可执行文件。这会导致失败，因为我们并没有一个 `main` 函数，即 GHC 所认为的一个独立程序的执行入口。

编译后会得到两个新文件：`SimpleJSON.hi` 与 `SimpleJSON.o`。前者是一个接口 *interface* 文件，即 `ghc` 以机器码的形式存储模块导出的名称信息；后者是一个对象 *object* 文件，其包含了生产的机器码。

生成一个 Haskell 程序，导入模块

添加一个 `Main.hs` 文件，其内容如下：

```
1 module Main where
2
3 import SimpleJSON
4
5 main = print (JObject [("foo", JNumber 1), ("bar", JBool False)])
```

与原文 `module Main () where` 的不同之处在于，现在的 `Main` 后不再需要一个 `()`。接下来是编译 `main` 函数：

```
1 ghc -o simple Main.hs
```

与原文 `ghc -o simple Main.hs SimpleJSON.o` 不同，现在没了 `SimpleJSON.o` 这个文件，加上后会报错重复 symbol 的编译错误（因为在 `Main.hs` 中已经做了 `import SimpleJSON` 导入了）。

这次省略掉了 `-c` 选项，因此编译器尝试生成一个可执行。生成可执行的过程被称为链接 *linking*（与 C++ 一样），即在一次编译中链接源文件与可执行文件。

这里给了 `ghc` 一个新选项 `-o`，其接受一个参数：可执行文件的名称，这里是 `simple`，执行它：

```
1 ./simple
2 JObject [{"foo",JNumber 1.0},{"bar",JBool False}]
```

打印 JSON 数据

现在我们将 Haskell 的值渲染成 JSON 数据，创建一个 `PutJSON.hs` 文件：

```
1 module PutJSON where
2
3 import Data.List (intercalate)
4 import SimpleJSON
5
6 renderJValue :: JValue -> String
7 renderJValue (JString s) = show s
8 renderJValue (JNumber n) = show n
9 renderJValue (JBool True) = "true"
10 renderJValue (JBool False) = "false"
11 renderJValue JNull = "null"
12 renderJValue (JObject o) = "{" ++ pairs o ++ "}"
```

```

13     where
14         pairs [] = ""
15         pairs ps = intercalate ", " (map renderPair ps)
16         renderPair (k, v) = show k ++ ": " ++ renderJValue v
17     renderJValue (JArray a) = "[" ++ values a ++ "]"
18     where
19         values [] = ""
20         values vs = intercalate ", " (map renderJValue vs)

```

好的 Haskell 风格需要分隔纯代码与 I/O 代码。我们的 `renderJValue` 函数不会与外界交互，但是仍然需要一个打印的函数：

```

1 putJValue :: JValue -> IO ()
2 putJValue = putStrLn . renderJValue

```

类型推导是把双刃剑

假设我们编写了一个自认为返回 `String` 的函数，但是并不为其写类型签名：

```

1 upcaseFirst (c:cs) = toUpper c -- forgot ":cs" here

```

这里希望单词首字母大写，但是忘记了将剩余的字符放进结果中。我们认为函数的类型是 `String -> String`，但是编译器则会将其视为 `String -> Char`。假设我们尝试在其他地方调用该函数：

```

1 camelCase :: String -> String
2 camelCase xs = concat (map upcaseFirst (words xs))

```

那么当我们尝试编译该代码或者是加载进 `ghci`，我们并不会得到明显的错误信息：

```

1 ghci> :load Trouble
2 [1 of 1] Compiling Main                ( Trouble.hs, interpreted )
3
4 Trouble.hs:9:27:
5   Couldn't match expected type `[Char]' against inferred type `Char'
6     Expected type: [Char] -> [Char]
7     Inferred type: [Char] -> Char
8     In the first argument of `map', namely `upcaseFirst'
9     In the first argument of `concat', namely
10        `(map upcaseFirst (words xs))'
11 Failed, modules loaded: none.

```

注意这里的报错是在 `upcaseFirst` 函数处，那么假设我们认为 `upcaseFirst` 的定义与类型是正确的，那么查找到真正的错误可能会花掉一些时间。

更加泛用的渲染

我们将更为泛用的打印模块称为 `Prettify`，那么其源文件即 `Prettify.hs`。

为了使 `Prettify` 满足实际需求，我们还要一个新的 JSON 渲染器来使用 `Prettify` 的 API。在 `Prettify` 模块中将使用一个抽象类型 `Doc`。基于建立在抽象类型的泛用渲染库，我们可以选择灵活高效的实现。

`PrettyJSON.hs` 示例：

```
1 renderJValue :: JValue -> Doc
2 renderJValue (JBool True) = text "true"
3 renderJValue (JBool False) = text "false"
4 renderJValue JNull = text "null"
5 renderJValue (JNumber num) = double num
6 renderJValue (JString str) = string str
```

这里的 `text`，`double` 以及 `string` 都会在 `Prettify` 模块中提供。

开发 Haskell 代码而不发疯

一个用于快速开发程序框架的技巧就是编写占位符，或者类型与函数的根 *stub* 版本。例如上述 `string`，`text` 以及 `double` 函数将被 `Prettify` 模块提供。如果我们没有提供这些函数或者 `Doc` 类型，那么“早点编译，经常编译”这个尝试就会失败。为了避免这个问题现在让我们编写一个不做任何事情的根本程序。

```
1 import SimpleJSON
2
3 data Doc = ToBeDefined deriving (Show)
4
5 string :: String -> Doc
6 string str = undefined
7
8 text :: String -> Doc
9 text str = undefined
10
11 double :: Double -> Doc
12 double num = undefined
```

特殊值 `undefined` 有一个 `a` 类型，无论在哪里使用它，总是会有类型检查。如果尝试计算，则会使程序崩溃：

```
1 ghci> :type undefined
2 undefined :: a
3 ghci> undefined
4 *** Exception: Prelude.undefined
5 ghci> :type double
6 double :: Double -> Doc
7 ghci> double 3.14
8 *** Exception: Prelude.undefined
```

尽管还不能运行根代码，但是编译器的类型检查器可以确保我们的程序类型正确。

漂亮的打印字符串

当需要打印一个漂亮的字符串时，我们必须遵循 JSON 的转义规则。字符串就是一系列被包裹在引号中的字符们。 `PrettyJSON.hs`：

```
1 string :: String -> Doc
2 string = enclose '"' ' ' . hcat . map oneChar
```

以及 `enclose` 函数将一个 `Doc` 值简单的包裹在一个开始与结束字符之间：

```
1 enclose :: Char -> Char -> Doc -> Doc
2 enclose left right x = char left <> x <> char right
```

这里提供的 `<>` 函数定义在 `Prettify` 库中，它需要两个 `Doc` 值，即 `Doc` 版本的 `(++)`。还是先在根文件中定义：

```
1 (<>) :: Doc -> Doc -> Doc
2 a <> b = undefined
3
4 char :: Char -> Doc
5 char c = undefined
```

我们的库还需要提供 `hcat`，将若干 `Doc` 值合成成为一个（类似于列表的 `concat`）：

```
1 hcat :: [Doc] -> Doc
2 hcat xs = undefined
```

我们的 `string` 函数应用 `oneChar` 函数到字符串中的每个字符，将它们连接，然后用引号包装。而 `oneChar` 函数则是用来转义或包装一个独立的字符。在 `PrettyJSON.hs` 中：

```
1 oneChar :: Char -> Doc
2 oneChar c = case lookup c simpleEscapes of
3   Just r -> text r
4   Nothing
5     | mustEscape c -> hexEscape c
6     | otherwise -> char c
7   where
8     mustEscape c = c < ' ' || c == '\x7f' || c > '\xff'
9
10  simpleEscapes :: [(Char, String)]
11  simpleEscapes = zipWith ch "\b\n\f\r\t\\\"/" "bnfrt\\\"/"
12  where
13    ch a b = (a, ['\\', b])
```

这里的 `simpleEscapes` 是一个列表的二元组，我们称其为关联 *association* 列表，或者简称 `alist`。每个 `alist` 的元素都将字符关联了它的转义表达，测试：

```
1 ghci> take 4 simpleEscapes
2 [('b', "\\b"), ('n', "\\n"), ('f', "\\f"), ('r', "\\r")]
```


我们的 `case` 表达式尝试查看字符是否匹配关联列表。如果匹配则返回匹配值，如若不然则需要以更复杂的方式来转义该字符。只有当两种转义都不需要时，才会返回普通字符。保守起见，我们输出的唯一未转义字符是可打印的 ASCII 字符。

更复杂的转义包含了将一个字符转为字符串 `"\u"` 并跟随四个十六进制的字符用于表达 Unicode 字符的数值。仍然是 `PrettyJSON.hs`：

```
1 smallHex :: Int -> Doc
2 smallHex x =
3   text "\\u"
4   <> text (replicate (4 - length h) '0')
5   <> text h
6   where
7     h = showHex x ""
```

这里的 `showHex` 函数需要从 `Numeric` 库中加载，其用于返回一个值的十六进制：

```
1 ghci> showHex 114111 ""
2 "1bdbf"
```

`replicate` 函数则是由 Prelude 提供：

```
1 ghci> replicate 5 "foo"
2 ["foo","foo","foo","foo","foo"]
```

`smallHex` 提供的四位编码只能表示最大 `0xffff` 的 Unicode 字符，而有效的 Unicode 字符的范围可以达到 `0x10ffff`。为了正确的表达一个超出了 `0xffff` 的 JSON 字符串，我们遵循一些复杂的规则将其分为两部分。这使我们有机会对 Haskell 的数执行一些位级操作。还是 `PrettyJSON.hs`：

```
1 astral :: Int -> Doc
2 astral n = smallHex (a + 0xd800) <> smallHex (b + 0xdc00)
3   where
4     a = (n `shiftR` 10) .&. 0x3ff
5     b = n .&. 0x3ff
```

这里 `shiftR` 函数和 `(.&.)` 函数都源自 `Data.Bits` 模块，前者将一个数移动右一位，后者则是执行一个字节层面的两值 *and* 操作。

```
1 ghci> 0x10000 `shiftR` 4    :: Int
2 4096
3 ghci> 7 .&. 2              :: Int
4 2
```

现在有了 `smallHex` 与 `astral`，我们可以提供 `hexEscape` 的定义了：

```
1 hexEscape :: Char -> Doc
2 hexEscape c
3   | d < 0x10000 = smallHex d
4   | otherwise = astral (d - 0x10000)
```

```

5   where
6     d = ord c

```

其中 `ord` 由 `Data.Char` 模块提供。

数组与对象，以及模块头

相比于字符串的漂亮打印，数组和对象就是小菜一碟了。我们已经知道了它们两者其实很相似：都是由起始字符开始，接着是一系列由逗号分隔的值，最后接上结束字符。让我们编写一个函数捕获数组与对象的共同结构：

```

1   series :: Char -> Char -> (a -> Doc) -> [a] -> Doc
2   series open close item = enclose open close . fsep . punctuate (char ',') . map item

```

让我们首先从函数类型开始。它接受起始与结束字符，一个打印某些未知类型 `a` 值的函数，以及一个类型为 `a` 的列表，返回一个类型为 `Doc` 的值。

注意尽管我们的类型签名提及了四个参数，在函数定义中仅列出了三个。这遵循了简化定义的规则，如 `myLength xs = length xs` 等同于 `myLength = length`。

我们已经有了之前编写过的 `enclose`，即包装一个 `Doc` 值进起始与结束字符之间。那么 `fsep` 函数则位于 `Prettify` 模块中，其结合一个 `Doc` 值列表成为一个 `Doc`，在输出不适合单行的情况下还需要换行。

```

1   fsep :: [Doc] -> Doc
2   fsep xs = undefined

```

那么现在，遵循上述提供的例子，你应该能够定义你自己的 `Prettify.hs` 的根文件了。这里不再显式的定义更多的根了。

`punctuate` 函数同样位于 `Prettify` 模块中：

```

1   punctuate :: Doc -> [Doc] -> [Doc]
2   punctuate p [] = []
3   punctuate p [d] = [d]
4   punctuate p (d : ds) = (d <> p) : punctuate p ds

```

通过 `series` 的定义，漂亮打印一个数组就非常直接了：

```

1   renderJValue (JArray ary) = series '[' ']' renderJValue ary

```

对于对象而言，还需要额外的一些工作：每个元素同时需要处理名称与值：

```

1   renderJValue (JObject obj) = series '{' '}' field obj
2   where
3     field (name, val) =
4       string name
5         PrettyStub.<> text ":"
6         PrettyStub.<> renderJValue val

```

编写一个模块头

现在已经有了 `PrettyJSON.hs` 文件，我们需要回到其顶部添加模块声明：

```
1 module PrettyJSON (renderJValue) where
```

这里导出了一个名称：`renderJValue`，也就是我们的 JSON 渲染函数。模块中其它的定义都是用于支持 `renderJValue` 的，因此没有必要对其它模块可见。

充实我们的漂亮打印库

在 `Prettify` 模块中，提供了 `Doc` 类型作为一个代数数据类型：

```
1 data Doc
2   = Empty
3   | Char Char
4   | Text String
5   | Line
6   | Concat Doc Doc
7   | Union Doc Doc
8   deriving (Show)
```

观察可知 `Doc` 类型实际上是一颗树。`Concat` 与 `Union` 构造函数根据其它两个 `Doc` 值创建一个内部节点，而 `Empty` 以及其它简单的构造函数用于构建叶子。

在模块的头部，我们导出该类型的名称，而不是它们的构造函数：这样可以防止使用 `Doc` 的构造函数被用于创建与模式匹配 `Doc` 值。

相反的，要创建一个 `Doc`，用户需要调用我们在 `Prettify` 模块中所提供的函数：

```
1 empty :: Doc
2 empty = Empty
3
4 char :: Char -> Doc
5 char = Char
6
7 text :: String -> Doc
8 text "" = Empty
9 text s = Text s
10
11 double :: Double -> Doc
12 double = text . show
```

`Line` 构造函数代表一个换行，其创建的是一个 *hard* 换行，即总是会在漂亮的打印中出现。有时我们想要一个 *soft* 换行，即只会在窗口或者页面上过长显示时才会换行。稍后将会介绍 `softline` 函数。

```
1 line :: Doc
2 line = Line
```

另外就是用于连接两个 `Doc` 值的 `(<>)` 函数（这里使用 `(<+>)`，因为 `(<>)` 在 Prelude 中已经有定义了）：

```
1 (<+>) :: Doc -> Doc -> Doc
2 Empty <+> y = y
3 x <+> Empty = x
4 x <+> y = x `Concat` y
```

测试：

```
1 ghci> text "foo" <> text "bar"
2 Concat (Text "foo") (Text "bar")
3 ghci> text "foo" <> empty
4 Text "foo"
5 ghci> empty <> text "bar"
6 Text "bar"
```

接下来是用于连接 `Doc` 列表的 `hcat` 函数：

```
1 hcat :: [Doc] -> Doc
2 hcat = fold (<+>)
3
4 fold :: (Doc -> Doc -> Doc) -> [Doc] -> Doc
5 fold f = foldr f empty
```

以及 `fsep` 函数，它还依赖若干其它函数：

```
1 fsep :: [Doc] -> Doc
2 fsep = fold (</>)
3
4 (</>) :: Doc -> Doc -> Doc
5 x </> y = x <+> softline <+> y
6
7 softline :: Doc
8 softline = group line
```

这里需要解释一下，`softline` 函数应该在当前行特别宽的时候进行换行，否则插入空格。如果我们的 `Doc` 类型不包含任何关于渲染的信息，那么该如何呢？答案就是每次遇到一个软换行，通过 `Union` 构造函数来维护两个可选项：

```
1 group :: Doc -> Doc
2 group x = flatten x `Union` x
```

`flatten` 函数将一个 `Line` 替换为空格，将两行转换为一个更长的行。

```
1 flatten :: Doc -> Doc
2 flatten (x `Concat` y) = flatten x `Concat` flatten y
3 flatten Line = Char ' '
4 flatten (x `Union` _) = flatten x
5 flatten other = other
```

注意总是调用 `flatten` 在 `Union` 的左元素上：每个 `Union` 的左侧总是大于等于右侧宽度（字符距离）。

紧密渲染

我们需要频繁的使用包含尽可能少字符的数据。例如通过网络连接发送 JSON 数据就没有必要美观：软件并不在乎数据的美观与否，添加很多空格只会带来性能下降。

因此我们提供了一个紧密渲染的函数：

```
1 compact :: Doc -> String
2 compact x = transform [x]
3 where
4     transform [] = ""
5     transform (d : ds) = case d of
6         Empty -> transform ds
7         Char c -> c : transform ds
8         Text s -> s ++ transform ds
9         Line -> '\n' : transform ds
10        a `Concat` b -> transform (a : b : ds)
11        _ `Union` b -> transform (b : ds)
```

`compact` 函数将其参数包裹成一个列表，然后将帮助函数 `transform` 应用至该列表。`transform` 函数视其参数为堆叠的项用于处理，列表的第一个元素即堆的顶部。

`transform` 函数的 `(d:ds)` 模式将堆顶部的元素取出 `d` 并留下 `ds`。在 `case` 表达式中，前面几个分支都在 `ds` 上递归，每次递归消费堆顶部的元素；最后两个分支则是在 `ds` 之前添加项：`Concat` 添加两个元素至堆，而 `Union` 分支则忽略它左侧元素，调用的是 `flatten`，再将其右侧元素至堆。

测试 `compact`：

```
1 ghci> let value = renderJValue (JObject [("f", JNumber 1), ("q", JBool True)])
2 ghci> :type value
3 value :: Doc
4 ghci> putStrLn (compact value)
5 {"f": 1.0,
6  "q": true
7  }
```

为了更好的理解代码是如何运作的，让我们用一个更简单的例子来展示细节：

```
1 ghci> char 'f' <> text "oo"
2 Concat (Char 'f') (Text "oo")
3 ghci> compact (char 'f' <> text "oo")
4 "foo"
```

当我们应用 `compact` 时，它会将它的参数转为一个列表后应用函数 `transform`。

- 接下来 `transform` 函数接受了一个单例列表，然后进行模式匹配 `(d:ds)`，这里的 `d` 是 `Concat (Char 'f') (Text "oo")`，而 `ds` 则是一个空列表。

由于 `d` 的构造函数是 `Concat`，其模式匹配就在 `case` 表达式中。那么在右侧，添加 `Char 'f'` 与 `Text "oo"` 值堆，然后递归的应用 `transform`。

- `transform` 函数接受了一个包含两项的列表，继续匹配 `d:ds` 模式。此时变量 `d` 绑定到了 `Char 'f'`，而 `ds` 则是 `[Text "oo"]`。
`case` 表达式匹配到了 `Char` 分支。那么在右侧，使用 `(:)` 来构建一个列表，其头部为 `'f'`，其余部分则是递归应用 `transform` 后的结果。
- * 递归的调用接受到一个单例列表，其变量 `d` 绑定至 `Text "oo"`，`ds` 绑定至 `[]`。
`case` 表达式匹配 `Text` 分支。那么在右侧，使用 `(++)` 来连接 `"oo"` 与递归调用 `transform` 后的结果。
 * · 最后的调用，`transform` 得到一个空列表，返回一个空字符串。
 * 结果是 `"oo" ++ ""`
- 结果是 `'f' : "oo" ++ ""`

真实的漂亮打印

我们的 `compact` 函数对于机器之间的交流是有帮助的，但是它的结果对人类而言并不易读：每行的信息很少。为了生成一个更可读的输出，我们将要编写另一个函数 `pretty`。相比于 `compact`，`pretty` 接受一个额外的参数：一行的最大宽度。

```
1 pretty :: Int -> Doc -> String
```

确切来说，`Int` 参数控制了 `pretty` 在遇到一个 `softline` 时的行为。在一个 `softline` 时，`pretty` 会选择继续留在当前行还是另起一行。其余情况下，必须严格遵守漂亮打印函数所设定的指令。

以下是实现的代码：

```
1 pretty width x = best 0 [x]
2 where
3   best col (d : ds) = case d of
4     Empty -> best col ds
5     Char c -> c : best (col + 1) ds
6     Text s -> s ++ best (col + length s) ds
7     Line -> '\n' : best 0 ds
8     a `Concat` b -> best col (a : b : ds)
9     a `Union` b -> nicest col (best col (a : ds)) (best col (b : ds))
10  best _ _ = ""
11  nicest col a b
12    | (width - least) `fits` a = a
13    | otherwise = b
14  where
15    least = min width col
```

`best` 帮助函数接受两个参数：当前行使用了的列数，以及剩余的仍需处理的 `Doc` 列表。

在简单的情况下，随着消费输入 `best` 直接更新了 `col` 变量。其中 `Concat` 情况也很明显：将两个连接过的部分推至堆叠，且不触碰 `col`。

有趣的情况在 `Union` 构造函数。回想一下之前将 `flatten` 应用至左侧元素，且不对右侧做任何操作。还有就是 `flatten` 将新行替换成空格。因此我们则需要检查这两种布局，`flatten` 后的还是原始的，更适合我们的宽度限制。

为此需要编写一个小的帮助函数来决定 `Doc` 值的一行是否适合给定的长度：

```
1 fits :: Int -> String -> Bool
2 w `fits` _ | w < 0 = False
3 w `fits` "" = True
4 w `fits` ('\n' : _) = True
5 w `fits` (c : cs) = (w - 1) `fits` cs
```

遵循漂亮打印

为了理解代码如何工作的，首先考虑一个简单的 `Doc` 值：

```
1 ghci> empty </> char 'a'
2 Concat (Union (Char ' ') Line) (Char 'a')
```

我们将应用 `pretty 2` 在该值。当我们首先应用 `best`，`col` 值为零。它匹配 `Concat` 模式，接着将 `Union (Char ' ') Line` 与 `Char 'a'` 推至堆，接着是递归应用自身，它匹配了 `Union (Char ' ') Line`。

现在忽略 Haskell 通常的计算顺序，两个子表达式，`best 0 [Char ' ', char 'a']` 与 `best 0 [Line, Char 'a']`，前者计算得到 `" a"`，而后者得到 `"na "`。接着将它们替换到外层的表达式中，得到 `nicest 0 " a" "\na"`。

为了明白 `nicest` 的结果，我们做一个小小的替换。`width` 以及 `col` 分别是 0 与 2，那么 `least` 就是 0，`width - least` 就是 2。这里用 `ghci` 来计算一下 `2 `fits` " a"`：

```
1 ghci> 2 `fits` " a"
2 True
```

计算得到 `True`，那么这里的 `nicest` 结果就是 `" a"`。

如果将 `pretty` 函数应用到之前同样的 JSON 数据上，可以看到根据提供的最大长度，它会得到不同的结果：

```
1 ghci> putStrLn (pretty 10 value)
2 {"f": 1.0,
3  "q": true
4  }
5 ghci> putStrLn (pretty 20 value)
6 {"f": 1.0, "q": true
7  }
8 ghci> putStrLn (pretty 30 value)
9 {"f": 1.0, "q": true }
```

创建一个库

Haskell 社区构建了一个标准工具库，名为 Cabal，其用于构建，安装，以及分发软件。Cabal 以 `包package` 的方式管理软件，一个包包含了一个库，以及若干可执行程序。

编写一个包的描述

要使用包，Cabal 需要一些描述。这些描述保存在一个文本文件中，以 `.cabal` 后缀命名。该文件应位于项目的根目录。

包描述由一系列的全局属性开始，其应用于包中所有的库以及可执行。

```
1 name:           pretty-json
2 version:        0.1.0.0
```

包名称必须是唯一的。如果你创建并安装了一个同名包在你的系统上，GHC 会感到迷惑。

```
1 synopsis:       My pretty printing library, with JSON support
2 description:
3   A simple pretty printing library that illustrates how to
4   develop a Haskell library.
5 author:         jacob xie
6 maintainer:     jacobbishopxy@gmail.com
```

这里还要有 license 信息，大多数 Haskell 包都使用 BSD license，Cabal 称为 BSD3。

另外就是 Cabal 的版本：

```
1 cabal-version:  2.4
```

在一个包中要描述一个独立的库，需要 *library* 这个部分。注意缩进在这里很重要。

```
1 library
2   default-language: Haskell2010
3   build-depends:    base
4   exposed-modules:
5     Prettify
6     PrettyJSON
7     SimpleJSON
```

`exposed-modules` 字段包含了一个模块列表，用于暴露给使用该包的用户导入。另一个可选字段 `other-modules` 包含了一个内部 *internal* 模块的列表，它们提供给 `exposed-modules` 内的模块使用，而对用户不可见。

`build-depends` 字段包含了一个逗号分隔的包列表，它们是我们库所需的依赖。`base` 包中包含了 Haskell 很多核心模块，例如 Prelude，所以它总是必须的。

GHC 的包管理器

GHC 包含了一个简单的包管理器用于追踪安装了哪些包，以及这些包的版本号。一个名为 `ghc-pkg` 的命令行工具提供了包数据库的管理。

这里说数据库 *database* 是因为 GHC 区分了系统级别的包，即对所有用户可用；以及用户可见的包，即仅对当前用户可用。后者可以避免管理员权限来安装包。

ghc-pkg 提供了不同的子命令，多数时候我们仅需两个命令：**ghc-pkg list** 用于查看已安装的包；当需要删除包时则使用 **ghc-pkg unregister**。

设置，构建与安装

除了一个 **.cabal** 文件，一个包还必须包含一个 *setup* 文件。在包需要的情况下，它允许 Cabal 的构建过程中包含大量的定制。

Setup.hs 示例：

```
1  #!/usr/bin/env runhaskell
2
3  import Distribution.Simple
4
5  main = defaultMain
```

一旦有了 **.cabal** 与 **Setup.hs** 文件，那么就剩下三步了。

指导 Cabal 如何构建以及在哪里安装包，只需一个简单命令：

```
1  runghc Setup configure
```

这可以确保我们所需要的包都是可用的，并且保存设定为了之后的 Cabal 命令。

如果没有为 **configure** 提供任何参数，Cabal 则会将包安装在系统层的包数据库。要在 home 路径安装则需要提供一些额外的信息：

```
1  runghc Setup configure --prefix=$HOME --user
```

接下来就是包的构建：

```
1  runghc Setup build
```

如果成功了，我们就可以安装这个包了。我们无需指定其安装的位置：Cabal 会使用我们在 **configure** 步骤中所提供的配置。即安装在我们自己的路径，并更新 GHC 的用于层包数据库。

```
1  runghc Setup install
```

6 Using typeclasses

Typeclasses 是 Haskell 中最强大的特性。它们允许我们定义通用性的接口，为各种类型提供公共特性集。Typeclasses 是一些语言特性的核心，例如相等性测试和数字运算符。

对 Typeclasses 的需求

假设在没有相等性测试 `==` 的情况下需要构建一个简单的 `color` 类型，那么相等测试就应该如下：

```
1 data Color = Red | Green | Blue
2
3 colorEq :: Color -> Color -> Bool
4 colorEq Red Red = True
5 colorEq Green Green = True
6 colorEq Blue Blue = True
7 colorEq _ _ = False
```

现在假设我们想为 `StringS` 增加一个相等性测试。由于 Haskell 的 `String` 是字符列表，我们可以编写一个简单的函数用于测试。这里为了简化我们使用 `==` 操作符用于说明。

```
1 stringEq :: [Char] -> [Char] -> Bool
2 stringEq [] [] = True
3 stringEq (x:xs) (y:ys) = x == y && stringEq xs ys
4 stringEq _ _ = False
```

现在已经发现问题了：我们必须为每个不同的类型使用不同名称的比较函数，这是很低效且令人讨厌的。因此需要一个通用的函数可用于比较任何东西。此外，当新的数据类型之后被创建时，已经存在的代码不能被改变。

Haskell 的 typeclasses 就是设计用来解决上述问题的。

什么是 typeclasses

Typeclasses 定义了一系列的函数，它们可以根据给定的数据类型有不同的实现。

首先我们必须定义 typeclass 本身。我们希望一个函数接受同样类型的两个参数，返回一个 `Bool` 来表示它们是否相等。我们无需在意类型是什么，但需要两个参数的类型相同。下面是 typeclass 的第一个定义：

```
1 class BasicEq a where
2   isEqual :: a -> a -> Bool
```

通过 `ghci` 的类型检查 `:type` 可以得知 `isEqual` 的类型：

```
1 ghci> :type isEqual
2 isEqual :: (BasicEq a) => a -> a -> Bool
```

现在可以为特定类型定义 `isEqual` :

```
1 instance BasicEq Bool where
2   isEqual True True = True
3   isEqual False False = True
4   isEqual _ _ = False
```

测试:

```
1 ghci> isEqual False False
2 True
3 ghci> isEqual False True
4 False
5 ghci> isEqual "Hi" "Hi"
6
7 <interactive>:1:0:
8   No instance for (BasicEq [Char])
9     arising from a use of `isEqual' at <interactive>:1:0-16
10  Possible fix: add an instance declaration for (BasicEq [Char])
11  In the expression: isEqual "Hi" "Hi"
12  In the definition of `it': it = isEqual "Hi" "Hi"
```

注意在尝试比较两个字符串时, `ghci` 发现我们并未给 `String` 提供 `BasicEq` 的实例。因此 `ghci` 并不知道该如何对 `String` 进行比较, 同时提议我们可以通过为 `[Char]` 定义 `BasicEq` 实例来解决这个问题。

下面是定义一个包含了两个函数的 `typeclass`:

```
1 class BasicEq2 a where
2   isEqual2 :: a -> a -> Bool
3   isNotEqual2 :: a -> a -> Bool
```

虽然 `BasicEq2` 的定义没有问题, 但是它让我们做了额外的事情。就逻辑而言, 如果我们知道了 `isEqual` 或 `isNotEqual` 中的一个, 我们便知道了另一个。那么与其让用户定义 `typeclass` 中两个函数, 我们可以提供一个默认的实践。

```
1 class BasicEq3 a where
2   isEqual3 :: a -> a -> Bool
3   isEqual3 x y = not (isNotEqual3 x y)
4
5   isNotEqual3 :: a -> a -> Bool
6   isNotEqual3 x y = not (isEqual3 x y)
```

声明 `typeclass` 实例

现在知道了如何定义 `typeclasses`, 接下来就是直到如何定义 `typeclasses` 的实例。回忆一下, 类型是由一个个特定 `typeclass` 所构成的实例所赋予了意义, 这些 `typeclasses` 又实现了必要的函数。

之前为我们的 `Color` 类型创建了相等性测试，现在让我们试试让其成为 `BasicEq3` 的实例：

```
1 instance BasicEq3 Color where
2   isEqual3 Red Red = True
3   isEqual3 Green Green = True
4   isEqual3 Blue Blue = True
5   isEqual3 _ _ = False
```

注意这里提供了与之前定义的一样的函数，实际上实现是完全相同的。不过在这种情况下，我们可以使用 `isEqual3` 在任何定义了 `BasicEq3` 实例的类型上。

另外注意 `BasicEq3` 定义了 `isEqual3` 与 `isNotEqual3`，而我们仅实现了它们其中一个。这是因为 `BasicEq3` 包含了默认实现，因此没有显式定义 `isNotEqual3` 时，编译器自动的使用了 `BasicEq3` 中的默认实现。

重要的内建 Typeclasses

Show

`Show` typeclass 用于将值转换为 `String`。可能最常用于将数字转换为字符串，很多类型都有它的实例，因此还可以用于转换更多的类型。如果自定义类型实现了 `Show` 的实例，那么就可以在 `ghci` 上展示或者在程序中打印出来。

`Show` 中最重要的函数就是 `show`，它接受一个参数：用于转换的数据，返回一个 `String` 来表示该数据。

```
1 ghci> :type show
2 show :: (Show a) => a -> String
```

一些其它的例子：

```
1 ghci> show 1
2 "1"
3 ghci> show [1, 2, 3]
4 "[1,2,3]"
5 ghci> show (1, 2)
6 "(1,2)"
```

`ghci` 展示的结果与输入到 Haskell 程序中的结果相同。表达式 `show 1` 返回的是单个字符的字符串，其包含数字 `1`，也就是说引号并不是字符串本身。通过 `putStrLn` 可以更清楚的看到：

```
1 ghci> putStrLn (show 1)
2 1
3 ghci> putStrLn (show [1,2,3])
4 [1,2,3]
```

我们也可以使用 `show` 在字符串上：

```

1 ghci> show "Hello!"
2 "\"Hello!\""
3 ghci> putStrLn (show "Hello!")
4 "Hello!"
5 ghci> show ['H', 'i']
6 "\"Hi\""
7 ghci> putStrLn (show "Hi")
8 "Hi"
9 ghci> show "Hi, \"Jane\""
10 "\"Hi, \\\"Jane\\\"\""
11 ghci> putStrLn (show "Hi, \"Jane\"")
12 "Hi, \"Jane\""

```

现在为我们自己的类型实现 `Show` 的实例：

```

1 instance Show Color where
2   show Red = "Red"
3   show Green = "Green"
4   show Blue = "Blue"

```

Read

`Read` typeclass 基本就是相反的 `Show`：它定义了函数接受一个 `String`，分析它，并返回属于 `Read` 成员的任何类型的数据。

```

1 ghci> :type read
2 read :: (Read a) => String -> a

```

以下是一个 `read` 与 `show` 的例子：

```

1 main = do
2   putStrLn "Please enter a Double:"
3   inpStr <- getLine
4   let inpDouble = read inpStr :: Double
5   putStrLn $ "Twice" ++ show inpDouble ++ " is " ++ show (inpDouble * 2)

```

`read` 的类型：`(Read a) => String -> a`。这里的 `a` 是每个 `Read` 实例的类型。也就是说特定的解析函数是根据预期的 `read` 返回类型所决定的。

```

1 ghci> (read "5.0")::Double
2 5.0
3 ghci> (read "5.0")::Integer
4 *** Exception: Prelude.read: no parse

```

在尝试解析 `5.0` 为 `Integer` 时异常。当预期返回值的类型是 `Integer` 时，`Integer` 的解析函数并不接受小数，因此异常被抛出。

`Read` 提供了一些相当复杂的解析器。你可以通过实现 `readsPrec` 函数定义一个简单的解析。该实现在解析成功时，返回只包含一个元组的列表，如果解析失败则返回空列表。下面是一个实现 `Color` 的 `Read` 实例的例子：

```

1  instance Read Color where
2  -- readsPrec is the main function for parsing input
3  readsPrec _ value =
4      -- We pass tryParse a list of pairs. Each pair has a string
5      -- and the desired return value. tryParse will try to match
6      -- the input to one of these strings
7      tryParse [("Red", Red), ("Green", Green), ("Blue", Blue)]
8      where
9          -- If there is nothing left to try, fail
10         tryParse [] = []
11         tryParse ((attempt, result) : xs) =
12             -- Compare the start of the string to be parsed to the
13             -- text we are looking for.
14             if take (length attempt) value == attempt
15             then -- If we have a match, return the result and the remaining input
16                 [(result, drop (length attempt) value)]
17             else -- If we don't have a match, try the next pair in the list of attempts.
18                 tryParse xs

```

测试：

```

1  ghci> (read "Red")::Color
2  Red
3  ghci> (read "Green")::Color
4  Green
5  ghci> (read "Blue")::Color
6  Blue
7  ghci> (read "[Red]")::[Color]
8  [Red]
9  ghci> (read "[Red,Red,Blue]")::[Color]
10 [Red,Red,Blue]
11 ghci> (read "[Red, Red, Blue]")::[Color]
12 *** Exception: Prelude.read: no parse

```

注意最后一个例子的异常。这是因为我们的解析器并没有聪明到能处理空格。

Tip

Read 并没有大范围的被使用

虽然可以使用 `Read` typeclass 来构建复杂的解析器，但许多人发现使用 `Parsec` 会更容易，且仅依赖 `Read` 来处理简单的任务。在第 16 章会详细介绍 `Parsec`。

通过 Read 和 Show 进行序列化

我们经常需要存储一个内存中的数据结构至硬盘供未来使用或者通过网络发送出去，那么将内存中数据转换为一个平坦的字节序列用于存储的这个过程就被称为序列化 *serialization*。

Tip

解析大型字符串

在 Haskell 中字符串的处理通常都是惰性的，因此 `read` 与 `show` 可以被用于处理很大的数据结构而不发生异常。Haskell 内建的 `read` 与 `show` 实例是高效的，并且都是纯 Haskell。如何处理解析异常的更多细节将会在第 19 章中详细介绍。

测试：

```
1 ghci> let d1 = [Just 5, Nothing, Nothing, Just 8, Just 9]::[Maybe Int]
2 ghci> putStrLn (show d1)
3 [Just 5,Nothing,Nothing,Just 8,Just 9]
4 ghci> writeFile "test" (show d1)
```

再是反序列：

```
1 ghci> input <- readFile "test"
2 "[Just 5,Nothing,Nothing,Just 8,Just 9]"
3 ghci> let d2 = read input
4
5 <interactive>:1:9:
6   Ambiguous type variable `a' in the constraint:
7     `Read a' arising from a use of `read' at <interactive>:1:9-18
8   Probable fix: add a type signature that fixes these type variable(s)
9 ghci> let d2 = (read input)::[Maybe Int]
10 ghci> print d1
11 [Just 5,Nothing,Nothing,Just 8,Just 9]
12 ghci> print d2
13 [Just 5,Nothing,Nothing,Just 8,Just 9]
14 ghci> d1 == d2
15 True
```

这里解释器并不知道 `d2` 的类型，因此抛出了异常。

以下是一些稍微复杂点的数据结构：

```
1 ghci> putStrLn $ show [( "hi", 1), ( "there", 3)]
2 [( "hi",1),( "there",3)]
3 ghci> putStrLn $ show [[1, 2, 3], [], [4, 0, 1], [], [503]]
4 [[1,2,3],[],[4,0,1],[],[503]]
5 ghci> putStrLn $ show [Left 5, Right "three", Left 0, Right "nine"]
6 [Left 5,Right "three",Left 0,Right "nine"]
7 ghci> putStrLn $ show [Left 0, Right [1, 2, 3], Left 5, Right []]
8 [Left 0,Right [1,2,3],Left 5,Right []]
```

数值类型

Haskell 拥有强大的数值类型。

选定的数值类型	
类型	描述
Double	双精度浮点数。浮点数的通常选择。
Float	单精度浮点数。通常用于与 C 交互。
Int	带方向的确定精度整数；最小范围 $[-2^{29}..2^{29} - 1]$ 。
Int8	8-bit 带方向的整数
Int16	16-bit 带方向的整数
Int32	32-bit 带方向的整数
Int64	64-bit 带方向的整数
Integer	带方向的任意精度整数；范围仅受机器限制。很常用。
Rational	带方向的任意精度的有理数。以两个 <code>Integers</code> 进行存储。
Word	无方向的确定精度整数；存储大小与 <code>Int</code> 一致
Word8	8-bit 无方向的整数
Word16	16-bit 无方向的整数
Word32	32-bit 无方向的整数
Word64	64-bit 无方向的整数

有很多不同的数值类型。有些运算，比如加法对所有类型都适用；还有其它类型的计算例如 `asin`，仅适用于浮点类型。

选定的数值函数与常量			
项	类型	模块	描述
(*)	$\text{Num } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	加法
(-)	$\text{Num } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	减法
(*)	$\text{Num } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	乘法
(/)	$\text{Fractional } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	除法
(**)	$\text{Floating } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	幂
()	$(\text{Num } a, \text{Integral } b) \Rightarrow a \rightarrow b \rightarrow a$	Prelude	非负数的幂
(i	$(\text{Fractional } a, \text{Integral } b) \Rightarrow a \rightarrow b$	Prelude	分数的幂
(%)	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow \text{Ratio } a$	Data.Ratio	比例
(.&.)	$\text{Bits } a \Rightarrow a \rightarrow a \rightarrow a$	Data.Bits	Bitwise 和
(. .)	$\text{Bits } a \Rightarrow a \rightarrow a \rightarrow a$	Data.Bits	Bitwise 或
abs	$\text{Num } a \Rightarrow a \rightarrow a$	Prelude	绝对值
approxRational	$\text{RealFrac } a \Rightarrow a \rightarrow a \rightarrow \text{Rational}$	Data.Ratio	基于分数分子与分母的近似有理组合
cos	$\text{Floating } a \Rightarrow a \rightarrow a$	Prelude	Cosine, 还有 acos, cosh, 与 acosh
div	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	整数除法, 总是向下取整, quot 同理
fromInteger	$\text{Num } a \Rightarrow \text{Integer} \rightarrow a$	Prelude	从 Integer 类型转换为任意数值类型
fromIntegral	$(\text{Integral } a, \text{Num } b) \Rightarrow a \rightarrow b$	Prelude	比上述更泛化的转换
fromRational	$\text{Fractional } a \Rightarrow \text{Rational} \rightarrow a$	Prelude	从 Rational 转换, 可能有损
log	$\text{Floating } a \Rightarrow a \rightarrow a$	Prelude	自然 log
logBase	$\text{Floating } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	显式底的 log
maxBound	$\text{Bounded } a \Rightarrow a$	Prelude	bound 类型的最大值
minBound	$\text{Bounded } a \Rightarrow a$	Prelude	bound 类型的最小值
mod	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	整数模
pi	$\text{Floating } a \Rightarrow a$	Prelude	数学常数的派
quot	$\text{Integral } a \Rightarrow a \rightarrow a \rightarrow a$	Prelude	整数除法; 商的小数部分向零截断

接上表

项	类型	模块	描述
recip	Fractional a => a -> a	Prelude	倒数
rem	Integral a => a -> a -> a	Prelude	整数除法的余数
round	(RealFrac a, Integral b) => a -> b	Prelude	向最近的整数方向取整
shift	Bits a => a -> Int -> a	Bits	左移指定 bits, 右移可能为负
sin	Floating a => a -> a	Prelude	Sine, 另 asin, sinh, 与 asinh
sqrt	Floating a => a -> a	Prelude	平方根
tan	Floating a => a -> a	Prelude	Tangent, 令 atan, tanh, 与 atanh
toInteger	Integral a => a -> Integer	Prelude	转换任意 Integral 至一个 Integer
toRational	Real a => a -> Rational	Prelude	转换 Real 至 Rational
truncate	(RealFrac a, Integral b) => a -> b	Prelude	向零截断数值
xor	Bits a => a -> a -> a	Data.Bits	Bitwise 的异或

数值类型的 Typeclass 实例								
项	Bits	Bounded	Floating	Fractional	Integral	Num	Rea	RealFrac
Double			X	X		X	X	X
Float	X	X			X	X	X	
Int	X	X			X	X	X	
Int16	X	X			X	X	X	
Int32	X	X			X	X	X	
Int64	X	X			X	X	X	
Integer	X				X	X	X	
Rational or any Ratio				X		X	X	X
Word	X	X			X	X	X	
Word16	X	X			X	X	X	
Word32	X	X			X	X	X	
Word64	X	X			X	X	X	

在数值类型间转换是另一个常用的需求。

数值类型之间的转换				
原始类型	目标类型			
	Double, Float	Int, Word	Integer	Rational
Double, Float	fromRational . toRational	truncate *	truncate *	toRational
Int, Word	fromIntegral	fromIntegral	fromIntegral	fromIntegral
Integer	fromIntegral	fromIntegral	N/A	fromIntegral
Rational	fromRational	truncate *	truncate *	N/A

自动化派生

对于很多简单的数据类型, Haskell 编译器可以自动的为我们派生出 `Read` , `Show` , `Bounded` , `Enum` , `Eq` 以及 `Ord` 的实例。这将大大的节省用户编写代码的时间。

```
1 data Color = Red | Green | Blue
2 deriving (Read, Show, Eq, Ord)
```

测试:

```
1 ghci> show Red
2 "Red"
3 ghci> (read "Red")::Color
4 Red
5 ghci> (read "[Red,Red,Blue]")::[Color]
6 [Red,Red,Blue]
7 ghci> (read "[Red, Red, Blue]")::[Color]
8 [Red,Red,Blue]
9 ghci> Red == Red
10 True
11 ghci> Red == Blue
12 False
13 ghci> Data.List.sort [Blue,Green,Blue,Red]
14 [Red,Green,Blue,Blue]
15 ghci> Red < Blue
16 True
```

自动派生也不总是能成功。例如如果定义了一个类型 `data MyType = MyType (Int -> Bool)` , 编译器将不能为其派生 `show` 的实例, 因为它不知道如何渲染一个函数。这种情况下我们会得到一个编译错误。

当我们自动派生某些 `typeclass` 的实例时, 在数据声明中引用的类型也必须是该 `typeclass` 的实例 (手动或自动的)。

```
1 data CannotShow = CannotShow
2 -- deriving (Show)
```

```

3
4  -- will not compile, since CannotShow is not an instance of Show
5  data CannotDeriveShow = CannotDeriveShow CannotShow
6      deriving (Show)
7
8  data OK = OK
9
10 instance Show OK where
11 show _ = "OK"
12
13 data ThisWorks = ThisWorks OK
14     deriving (Show)

```

注意 `CannotShow` 的 `deriving (Show)` 是被注释掉的，因此 `CannotDeriveShow` 才无法派生 `Show`。

Typeclasses 实战：令 JSON 使用起来更方便

上一章讲述的 `JValue` 处理 JSON 并不容易。例如，下面是一个实际 JSON 数据的删减整理的片段：

```

1  {
2    "query": "awkward squad haskell",
3    "estimatedCount": 3920,
4    "moreResults": true,
5    "results":
6    [{
7      "title": "Simon Peyton Jones: papers",
8      "snippet": "Tackling the awkward squad: monadic input/output ...",
9      "url": "http://research.microsoft.com/~simonpj/papers/marktoberdorf/",
10     },
11     {
12       "title": "Haskell for C Programmers | Lambda the Ultimate",
13       "snippet": "... the best job of all the tutorials I've read ...",
14       "url": "http://lambda-the-ultimate.org/node/724",
15     }
16   ]
17 }

```

以及在 Haskell 中的表达：

```

1  import SimpleJSON
2
3  result :: JValue
4  result = JObject [
5    ("query", JString "awkward squad haskell"),
6    ("estimatedCount", JNumber 3920),
7    ("moreResults", JBool True),
8    ("results", JArray [
9      JObject [

```

```

10     ("title", JString "Simon Peyton Jones: papers"),
11     ("snippet", JString "Tackling the awkward ..."),
12     ("url", JString "http://.../marktoberdorf/")
13 ])
14 ]

```

因为 Haskell 并不支持列表中包含不同类型的值，我们不能直接表示一个包含了多个类型的 JSON 对象。我们必须将每个值通过 `JValue` 构造函数来进行包装。这限制了我们的灵活性：如果想要更换数值 `3920` 成为一个字符串 `"3,920"`，我们必须更换构造函数，即 `JNumber` 变为 `JString`。

Haskell 的 typeclasses 为此类问题提供了一个诱人的解决方案：

```

1  type JSONError = String
2
3  class JSON a where
4      toJValue :: a -> JValue
5      fromJValue :: JValue -> Either JSONError a
6
7  instance JSON JValue where
8      toJValue = id
9      fromJValue = Right

```

现在无需再将类似 `JNumber` 这样的构造函数应用在值上将值包裹，直接应用 `toJValue` 函数在该值上即可。

我们同样提供了一个 `fromJValue` 函数，用于将一个 `JValue` 转换成一个我们所期望的类型。

更多有帮助的错误

让我们构造一下自己的 `Maybe` 与 `Either`：

```

1  data Maybe a
2  = Nothing
3  | Just a
4  deriving (Eq, Ord, Read, Show)
5
6  data Either a b
7  = Left a
8  | Right b
9  deriving (Eq, Ord, Read, Show)

```

在 `Bool` 值实例中尝试一下：

```

1  instance JSON Bool where
2      toJValue = JBool
3      fromJValue (JBool b) = Right b
4      fromJValue _ = Left "not a JSON boolean"

```

通过类型同义词来创建一个实例

```

1 instance JSON String where
2   toJValue = JString
3   fromJValue (JString s) = Right s
4   fromJValue _ = Left "not a JSON string"

```

活在开放世界

Haskell 的 typeclasses 允许我们在任何合适的时候创建新的 typeclass 实例。

```

1 doubleToJValue :: (Double -> a) -> JValue -> Either JSONError a
2 doubleToJValue f (JNumber v) = Right (f v)
3 doubleToJValue _ _ = Left "not a JSON number"
4
5 instance JSON Int where
6   toJValue = JNumber . realToFrac
7   fromJValue = doubleToJValue round
8
9 instance JSON Integer where
10  toJValue = JNumber . realToFrac
11  fromJValue = doubleToJValue round
12
13 instance JSON Double where
14  toJValue = JNumber
15  fromJValue = doubleToJValue id

```

我们还希望转换一个列表成为 JSON，暂时用 `undefined` 作为实例的方法。

```

1 instance (JSON a) => JSON [a] where
2   toJValue = undefined
3   fromJValue = undefined

```

对象亦是如此：

```

1 instance (JSON a) => JSON [(String, a)] where
2   toJValue = undefined
3   fromJValue = undefined

```

合适重叠实例会导致问题

如果我们将这些定义放入一个原文件中，并加载至 `ghci`，每个初始化看起来都没问题：

```

1 ghci> :load BrokenClass
2 [1 of 2] Compiling SimpleJSON      ( ../ch05/SimpleJSON.hs, interpreted )
3 [2 of 2] Compiling BrokenClass      ( BrokenClass.hs, interpreted )
4 Ok, modules loaded: SimpleJSON, BrokenClass.

```

然而当我们尝试使用元组的列表时，错误便发生了。

```

1 ghci> toJValue [("foo", "bar")]
2
3 <interactive>:1:0:
4   Overlapping instances for JSON [(Char), (Char)]
5   arising from a use of `toJValue' at <interactive>:1:0-23
6   Matching instances:
7     instance (JSON a) => JSON [a]
8       -- Defined at BrokenClass.hs:(44,0)-(46,25)
9     instance (JSON a) => JSON [(String, a)]
10      -- Defined at BrokenClass.hs:(50,0)-(52,25)
11   In the expression: toJValue [("foo", "bar")]
12   In the definition of `it': it = toJValue [("foo", "bar")]

```

重叠实例的问题是 Haskell 的开放世界假设的结果。下面用一个更简单的例子来说明到底发生了什么：

```

1 class Borked a where
2   bork :: a -> String
3
4 instance Borked Int where
5   bork = show
6
7 instance Borked (Int, Int) where
8   bork (a, b) = bork a ++ ", " ++ bork b
9
10 instance (Borked a, Borked b) => Borked (a, b) where
11   bork (a, b) = ">>" ++ bork a ++ " " ++ bork b ++ "<<"

```

我们有两个 typeclass `Borked` 的 pairs 实例：一对是 `Int`，另一对是其它任意值。

假设我们希望 `bork` 一对 `Int` 值，那么编译器必须选择一个实例来使用。由于这些实例相邻，那么看起来就能够简单的选择更加明确的实例。

然而 GHC 默认是保守的，它会坚持必须只有一个可能的实例。因此最终在使用 `bork` 时会抛出异常。

Tip

什么时候重叠的实例会有影响？

正如我们之前提到的那样，我们可以将一个 typeclass 的实例散落在若干模块中。GHC 并不会抱怨重叠实例的存在。而真正抱怨的时候就是当我们尝试使用这个受影响 typeclass 的方法时，也就是当强制要求选择哪一个实例需要使用的時候。

放宽 typeclasses 的某些限制

通常而言，我们不可以编写一个特定多态类型的 typeclass 实例。例如 `[Char]` 类型就是 `[a]` 指定 `Char` 类型的多态。因此禁止将 `[Char]` 声明为 typeclass 的实例。这个非常的不

方便，因为字符串在真实代码中处处存在。

`TypeSynonymInstances` 语言扩展移除了这个限制，允许我们编写上述的实例。

GHC 还支持另一个有用的语言扩展，`OverlappingInstances`，专门用于处理重叠实例。当存在若干重叠实例需要选择时，该扩展使编译器选择最明确的那个。

我们通常将该扩展与 `TypeSynonymInstances` 一起使用。例如：

```
1 import Data.List
2
3 class Foo a where
4     foo :: a -> String
5
6 instance {-# OVERLAPS #-} Foo a => Foo [a] where
7     -- foo = concat . intersperse ", " . map foo
8     foo = intercalate ", " . map foo
9
10 instance {-# OVERLAPS #-} Foo Char where
11     foo c = [c]
12
13 instance {-# OVERLAPS #-} Foo String where
14     foo = id
15
16 main :: IO ()
17 main = do
18     putStrLn $ "foo: " ++ foo "SimpleClass"
19     putStrLn $ "foo: " ++ foo ["a", "b", "c"]
```

与原文不同之处在于，注解 `{-# LANGUAGE OverlappingInstances #-}` 在 6.8.1 后被弃用，现在则是在 `instance` 后使用 `{-# OVERLAPS #-}` 来表示重叠的实例，详见文档。

如果我们将 `foo` 应用至一个 `String`，编译器将使用 `String` 指定的实现。尽管存在 `[a]` 与 `Char` 的 `Foo` 实例，但是 `String` 的实例更加的明确，因此 GHC 会选择它。

如何给类型一个新的身份

略。

```
1 data DataInt = D Int
2     deriving (Eq, Ord, Show)
3
4 newtype NewtypeInt = N Int
5     deriving (Eq, Ord, Show)
```

略。

总结：三总命名类型的方法：

- `data` 关键字引入了一个真实的新的代数数据类型。

- `type` 关键字给予了已存在的类型一个同义词。我们可以替换的使用类型与其同义词。
- `newtype` 关键字给予已存在类型一个新的身份。原始类型和新的类型是不可替换的。

JSON typeclasses 没有重叠的实例

现在需要帮助编译器区分 JSON 数组的 `[a]`，以及 JSON 对象的 `[(String, [a])]`。它们是造成重叠问题的原因。我们将列表类型包裹起来，这样编译器则不会视其为列表：

```
1 newtype JAry a = JAry {fromJAry :: [a]}
2 deriving (Eq, Ord, Show)
```

当需要将其从模块中导出时，我们需要导出该类型的所有细节。我们模块头看起来像是这样：

```
1 module JSONClass (JAry (...)) where
```

接下来是另一个包装类型用于隐藏 JSON 对象：

```
1 newtype JObj a = JObj {fromJObj :: [(String, a)]}
2 deriving (Eq, Ord, Show)
```

有了这些类型定义，那么就可以对 `JValue` 类型做点小修改：

```
1 data JValue
2   = JString String
3   | JNumber Double
4   | JBool Bool
5   | JNull
6   | JObject (JObj JValue) -- was [(String, JValue)]
7   | JArray (JAry JValue) -- was [JValue]
8   deriving (Eq, Ord, Show)
```

这个改动并不会影响已经写过的 `JSON` typeclass 实例，不过还是需要为 `JAry` 与 `JObj` 类型编写 `JSON` typeclass 实例。

```
1 jaryFromJValue :: (JSON a) => JValue -> Either JSONError (JAry a)
2
3 jaryToJValue :: (JSON a) => JAry a -> JValue
4
5 instance (JSON a) => JSON (JAry a) where
6   toJValue = jaryToJValue
7   fromJValue = jaryFromJValue
```

让我们慢慢的看一下将 `JAry a` 转换成 `JValue` 的每个步骤。给定一个列表，我们知道其所有元素都是 `JSON` 实例，将其转换成一个列表的 `JValue` 很简单。

```
1 listToJValues :: (JSON a) => [a] -> [JValue]
2 listToJValues = map toJValue
```

有了上述代码后，将其变为一个 `JArray JValue` 就仅仅是应用 `newtype` 类型构造函数即可：

```
1 jvaluesToJArray :: [JValue] -> JArray JValue
2 jvaluesToJArray = JArray
```

（记住这并没有性能损耗，仅仅只是告诉编译器隐藏我们在使用一个列表的事实。）将其转换成一个 `JValue`，我们应用赢一个类型构造函数：

```
1 jarrayOfJValuesToJValue :: JArray JValue -> JValue
2 jarrayOfJValuesToJValue = JArray
```

将这些部分用函数组合的方式集成起来，就获得了一个简洁的一行代码转换成一个 `JValue`：

```
1 jarrayToJValue = JArray . JArray . map toJValue . fromJArray
```

从 `JValue` 转换至一个 `JArray a` 则需要更多的工作，我们还是将其分解成可复用的部分。基础函数很直接：

```
1 jarrayFromJValue (JArray (JArray a)) = whenRight JArray (mapEithers fromJValue a)
2 jarrayFromJValue _ = Left "not a JSON array"
```

`whenRight` 函数检查它的参数：如果是由 `Right` 构造函数构建的它则调用一个函数，如果是由 `Left` 构造则保留值不变：

```
1 whenRight :: (b -> c) -> Either a b -> Either a c
2 whenRight _ (Left err) = Left err
3 whenRight f (Right a) = Right (f a)
```

更复杂的是 `mapEithers`，它的行为类似于普通的 `map` 函数，但是如果遇到一个 `Left` 值，它会立刻返回，而不是继续累积一个 `Right` 值的列表。

```
1 mapEithers :: (a -> Either b c) -> [a] -> Either b [c]
2 mapEithers f (x : xs) = case mapEithers f xs of
3   Left err -> Left err
4   Right ys -> case f x of
5     Left err -> Left err
6     Right y -> Right (y : ys)
7 mapEithers _ _ = Right []
```

由于隐藏在 `JObj` 类型的列表中的元素有少许结构，因此它与 `JValue` 之间的转换会变得更复杂。幸运的是我们可以复用刚刚定义好的函数：

```
1 instance (JSON a) => JSON (JObj a) where
2   toJValue = JObject . JObj . map (second toJValue) . fromJObj
3   fromJValue (JObject (JObj o)) = whenRight JObj (mapEithers unwrap o)
4   where
5     unwrap (k, v) = whenRight (k,) (fromJValue v)
6   fromJValue _ = Left "not a JSON object"
```

7 Input and output

Haskell 中的经典 I/O

略

处理文件以及句柄

Haskell 为 I/O 定义了相当多的基本函数，其中许多函数与其它编程语言中的函数相似。`System.IO` 库中提供了所有的基础 I/O 函数。

使用 `openFile` 会返回一个文件的 `Handle`，它用于对文件执行指定的操作。Haskell 提供了例如 `hPutStrLn` 这样的函数，其类似于 `putStrLn`，不同在于接受一个额外的参数 – 一个 `Handle` – 指定哪个文件被操作。当我们结束时，需要用 `hClose` 来结束 `Handle`。这些函数都定义在 `System.IO` 中，“h”开头的函数对应了几乎所有的非“h”开头的函数；例如 `print` 打印在屏幕上，而 `hPrint` 打印至一个文件。

一个例子：

```
1 import Data.Char (toUpper)
2 import System.IO
3
4 main :: IO ()
5 main = do
6     inh <- openFile "input.txt" ReadMode
7     outh <- openFile "output.txt" WriteMode
8     mainloop inh outh
9     hClose inh
10    hClose outh
11
12    putStrLn "whatever"
13
14 mainloop :: Handle -> Handle -> IO ()
15 mainloop inh outh = do
16     ineof <- hIsEOF inh
17     if ineof
18     then return ()
19     else do
20         inpStr <- hGetLine inh
21         hPutStrLn outh (map toUpper inpStr)
22         mainloop inh outh
```

`mainloop` 首先检查输入是否结束（EOF），如果没有则读取一行，将该行转为大写后写入输出文件中，再递归的调用 `mainloop`。

略

可能的 IOMode 值				
IOMode	可读?	可写?	起始位置	说明
ReadMode	Yes	No	文件起始	文件必须存在
WriteMode	No	Yes	文件起始	文件存在时会被清空
ReadWriteMode	Yes	Yes	文件起始	文件不存在时创建；否则现有数据保留
AppendMode	No	Yes	文件尾部	文件不存在时创建；否则现有数据保留

关闭句柄

上述例子中我们已经见到了 `hClose` 用于关闭文件的句柄。在之后的小节中我们将了解缓存 Buffering 这个概念，即 Haskell 在内部为文件维护了缓存。这样做能显著的提升性能，然而这就意味着在对打开的文件调用 `hClose` 之前，数据可能不会被刷新到操作系统中。

另一个原因是打开着的文件消耗系统的资源。如果程序运行的时间很长，开开了很多文件但是并没有关闭它们，那么程序很可能会因为资源枯竭而导致崩溃。

Seek 与 Tell

从一个关联了磁盘文件的 `Handle` 读写时，操作系统会维护一个关于当前位置的内部记录。每当进行下一层读取的时，操作系统返回下一个数据块，其起始点为当前位置，且增加的位置反应了读取了多少数据。

你可以使用 `hTell` 来找到文件中当前的位置。当文件被创建时，它是空的且你的位置将会是 0。在编写 5 个字节后，你的位置则变为了 5，以此类推。`hTell` 接受一个 `Handle` 并返回一个 `IO Integer` 来表示你的位置。

`hTell` 的同伴是 `hSeek`，它允许你更改文件的位置，其三个参数为：`Handle`，`SeekMode` 以及一个位置。

`SeekMode` 有三个不同类型，用于指定如何解析给定的位置。`AbsoluteSeek` 为文件中的精确位置，等同于 `hTell`；`RelativeSeek` 意为当前位置开始的多少位置，正值向后，负值向前；`SeekFromEnd` 则是从文件末尾往前多少的位置。

并不是所有的 `Handle` 都是可以 seek 的。一个 `Handle` 通常关联一个文件，但是它还可以关联其它的东西，比如网络连接，磁带机，或者终端。可以使用 `hIsSeekable` 来查看给定的 `Handle` 是否可以 seek。

标准输入，输出以及错误

较早之前我们指出每个非“h”函数通常都会有与其关联的“h”函数用作于任何 `Handle` 上。实际上非“h”函数只不过是“h”函数的一种缩写。

在 `System.IO` 中有三个预定义的 `Handle` : `stdin` 标准输入, 通常是键盘; `stdout` 标准输出, 通常是显示器; `stderr` 标准错误, 通常也是显示器。

像是 `getLine` 类似的函数可以简单的定义成:

```
1 getLine = hGetLine stdin
2 putStrLn = hPutStrLn stdout
3 print = hPrint stdout
```

删除与文件重命名

`System.Directory` 模块提供了两个比较有用的函数: `removeFile` 接受单个参数, 即文件名, 然后删除该文件; `renameFile` 接受两个文件名, 一个旧名称以及一个新的名称, 如果新的文件名是一个不同的路径, 那么可以认为这是一个移动。旧的文件名必须在调用 `renameFile` 之前就存在, 另外如果新文件已经存在, 则重命名后将其删除。

跟很多其它接受一个文件名的函数一样, 如果“旧”名称不存在, `renameFile` 则会抛出异常。

`System.Directory` 模块中还有很多其他的函数, 像是创建或移除文件夹, 在路径中查找文件列表, 测试文件是否存在, 等等。

临时文件

程序员频繁的需要临时文件。这些文件被用于存储大量等待计算的数据, 可被其它程序所用的数据, 等等。

通过名为 `openTempFile` 的函数 (以及相关联的 `openBinaryTempFile`) 可以帮助我们解决不少问题。

`openTempFile` 接受两个参数: 创建文件的路径, 以及一个“template”用于命名文件。路径可以是 `.` 来代表当前路径, 或者也可以使用 `System.Directory.getTemporaryDirectory` 来找到操作系统所给出的最佳放置临时文件的位置。template 则将一些随机字符添加至文件, 用以确保结果是真正唯一的。实际上它保证了可以在一个唯一的文件名上工作。

`openTempFile` 的返回类型是 `IO (FilePath, Handle)` 。元组的第一部分是被创建文件的名称, 第二部分则是一个模式为 `ReadWriteMode` 的打开了文件的 `Handle` 。当我们使用完文件, 我们希望用 `hClose` 来操作 `Handle` 用于关闭文件, 接着调用 `removeFile` 来删除它。接下来我们会看到一个例子用于展示这些函数的用法。

扩展案例: 函数式 I/O 以及临时文件

```
1 import Control.Exception
2 import System.Directory (getTemporaryDirectory, removeFile)
3 import System.IO
```

```

4
5 main :: IO ()
6 main = withTempFile "mytemp.txt" myAction
7
8 myAction :: FilePath -> Handle -> IO ()
9 myAction tempname temph = do
10     -- Start by displaying a greeting on the terminal
11     putStrLn "Welcome to tempfile.hs"
12     putStrLn $ "I have a temporary file at " ++ tempname
13
14     -- Let's see what the initial position is
15     pos <- hTell temph
16     putStrLn $ "My initial position is " ++ show pos
17
18     -- Now, write some data to the temporary file
19     let tempdata = show [1 .. 10]
20     putStrLn $ "Writing one line containing " ++ show (length tempdata) ++ " bytes: " ++
        tempdata
21     hPutStrLn temph tempdata
22
23     -- Get our new position. This doesn't actually modify pos in memory,
24     -- but makes the name "pos" correspond to a different value for
25     -- the remainder of the "do" block.
26     pos <- hTell temph
27     putStrLn $ "After writing, my new position is " ++ show pos
28
29     -- Seek to the beginning of the file and display it
30     putStrLn "The file content is: "
31     hSeek temph AbsoluteSeek 0
32
33     -- hGetContents performs a lazy read of the entire file
34     c <- hGetContents temph
35
36     -- Copy the file byte-for-byte to stdout, followed by \n
37     putStrLn $ "c: " ++ c
38
39     -- Let's also display it as a Haskell literal
40     putStrLn "Which could be expressed as this Haskell literal:"
41     print c
42
43 {-
44     This function takes two parameters: a filename pattern and another function.
45     It will create a temporary file, and pass the name and Handle of that file to
46     the given function.
47
48     The temporary file is created with openTempFile. The directory is the one
49     indicated by getTemporaryDirectory, or, if the system has no notion of a temporary
50     directory, "." is used. The given pattern is passed to openTempFile.
51

```

```

52     After the given function terminates, even if it terminates due to an exception,
53     the Handle is closed and the file is deleted.
54 -}
55 withTempFile :: String -> (FilePath -> Handle -> IO a) -> IO a
56 withTempFile pattern func = do
57     -- The library ref says that getTemporaryDirectory may raise an exception on
58     -- systems that have no notion of a temporary directory. So, we run
59     -- getTemporaryDirectory under catch. catch takes two functions: one to run,
60     -- and a different one to run if the first raised an exception.
61     -- If getTemporaryDirectory raised an exception, just use "." (the current
62     -- working directory).
63     tempdir <-
64         catch
65             getTemporaryDirectory
66             (\(_ :: IOException) -> return ".") -- explicit annotates exception type
67     (tempfile, temp) <- openTempFile tempdir pattern
68
69     -- Call (func tempfile temp) to perform the action on the temporary file.
70     -- finally takes two actions. The first is the action to run, the second is an action
71     -- to run after the first, regardless of the temporary file is always deleted.
72     -- The return value from finally is the first action's return value.
73     finally
74         (func tempfile temp)
75         ( do
76             hClose temp
77             removeFile tempfile
78         )

```

首先 `withTempFile` 函数证明了 Haskell 在 I/O 时也没有忘记其函数式的天性。该函数接受一个 `String` 与另一个函数。传递给 `withTempFile` 的函数带着临时文件的 `Handle` 被调用，当该函数退出时，临时文件关闭并被删除。

异常处理可以让程序变得更加健壮。通常我们希望在程序结束后删除临时文件，即使程序的执行过程中出现了错误。更多的错误处理会在第十九章讲述。

现在回到程序，`main` 做的事情很简单，调用函数 `withTempFile` 并传入临时文件名以及 `myAction` 函数。

`myAction` 展示了一些信息到终端上，写入了一些数据到文件中，`seek` 到文件的起始位置，通过 `hGetContents` 进行数据读取，接着按照字节展示文件内容，同时通过 `print c` 进行打印。这等同于 `putStrLn (show c)`。

现在看看输出：

```

1 % runhaskell tempfiles.hs
2 Welcome to tempfile.hs
3 I have a temporary file at /var/folders/nb/w1q3ztlj139_vz9pqglmbks80000gn/T/mytemp3709-0.txt
4 My initial position is 0
5 Writing one line containing 22 bytes: [1,2,3,4,5,6,7,8,9,10]
6 After writing, my new position is 23

```

```

7 The file content is:
8 c: [1,2,3,4,5,6,7,8,9,10]
9
10 Which could be expressed as this Haskell literal:
11 "[1,2,3,4,5,6,7,8,9,10]\n"

```

Lazy I/O

Haskell 还提供了另一种方式处理 I/O。由于 Haskell 是 lazy 语言，意味着任何数据只会在其值必须被知道时才会被计算，那么就有了一些新颖的 I/O 处理方法。

hGetContents

一个新颖的 I/O 方式就是 `hGetContents` 函数，其类型 `Handle -> IO String`，这里的 `String` 代表着由文件的 `Handle` 所返回的所有数据。

在严格求值的语言中，使用这样的函数通常是一个坏主意。假设读一个 500GB 的文件，那么就有可能因为内存不足而导致程序崩溃。在这些语言中，传统的做法就是使用循环来处理文件的所有数据。

但是 `hGetContents` 不一样，它返回的 `String` 是惰性的，即在调用 `hGetContents` 时，并没有读取任何数据。只有在处理列表的元素（字符）时才会从 `Handle` 中读取数据。当 `String` 的元素不再使用时，Haskell 的垃圾收集器则会自动释放内存。

来看一个例子：

```

1 import Data.Char (toUpper)
2 import System.IO
3
4 main :: IO ()
5 main = do
6     inh <- openFile "input.txt" ReadMode
7     outh <- openFile "output.txt" WriteMode
8     inpStr <- hGetContents inh
9     let result = processData inpStr
10    hPutStr outh result
11    hClose inh
12    hClose outh
13
14 processData :: String -> String
15 processData = map toUpper

```

测试：

```

1 ghci> :load toupper-lazy1.hs
2 [1 of 1] Compiling Main           ( toupper-lazy1.hs, interpreted )
3 Ok, modules loaded: Main.
4 ghci> processData "Hello, there! How are you?"

```



```
5 "HELLO, THERE! HOW ARE YOU?"
6 ghci> :type processData
7 processData :: String -> String
8 ghci> :type processData "Hello!"
9 processData "Hello!" :: String
```

Warning

如果我们在上述例子中尝试在使用 `inpStr` 的地方（对 `processData` 的调用）之后继续使用 `inpStr`，那么程序则会失去其内存效率。这是因为编译器将会强制保留 `inpStr` 值在内存中供未来使用。在这里编译器知道 `inpStr` 不会再被使用，那么就在其使用结束后尽快的释放内存。这里需要记住：内存只有在最后一次使用后才会被释放。

这段代码有点繁琐，我们可以让它变得更简洁一些：

```
1 import Data.Char (toUpper)
2 import System.IO
3
4 main :: IO ()
5 main = do
6   inh <- openFile "input.txt" ReadMode
7   outh <- openFile "output.txt" WriteMode
8   inpStr <- hGetContents inh
9   hPutStr outh (map toUpper inpStr)
10  hClose inh
11  hClose outh
```

在使用 `hGetContents` 时，我们甚至不需要从输入文件中消费所有的数据。每当 Haskell 系统确定 `hGetContents` 所返回的整个字符可以被垃圾回收时 – 意味着它不会再被使用 – 文件将自动被关闭。同样的原则也适用于从文件中读取的数据。每当给定的数据不再需要时，Haskell 环境就会释放存储该数据的内存。严格来说我们完全都不需要调用 `hClose`。然而这是一个好的习惯，因为之后对程序的改动会使得 `hClose` 变得很重要。

Warning

在使用 `hGetContents` 时，必须记住即使在之后的程序可能不再显式的引用 `Handle`，但还是要保持 `Handle` 不被关闭，直到 `hGetContents` 的结果被消费掉了。不这么做的话就可能会导致丢失部分或者全部文件数据。这是因为 Haskell 是惰性的，所以通常可以假设只有在输出了涉及输入的计算结果之后才消费了输入。

readFile 与 writeFile

Haskell 程序员经常使用 `hGetContents` 作为筛选。从一个文件读取数据，处理数据，接着将数据写到别的地方。这非常的通用所以有了一些快捷的方式：`readFile` 以及 `writeFile` 就是这样的函数，以字符串的方式处理文件。它们处理了所有关于打开文件，关闭文件，读取数据，以及写入数据的细节。`readFile` 在内部使用了 `hGetContents`。

通过 `ghci` 查看这些函数的类型：

```
1 ghci> :type readFile
2 readFile :: FilePath -> IO String
3 ghci> :type writeFile
4 writeFile :: FilePath -> String -> IO ()
```

使用 `readFile` 与 `writeFile` 改造之前的代码：

```
1 import Data.Char (toUpper)
2
3 main :: IO ()
4 main = do
5     inpStr <- readFile "input.txt"
6     writeFile "output.txt" $ map toUpper inpStr
```

关于 Lazy Output

现在我们能理解惰性输入是如何在 Haskell 中工作的。那么惰性输出又是怎么样呢？

我们知道 Haskell 中所有计算都是在其值被需要时才会进行。由于函数例如 `writeFile` 以及 `putStr` 会输出所有传给它们的 `String`，那么整个 `String` 必须被计算。因此可以保证 `putStr` 的参数将被完整求值。

那么输入惰性的意义呢？上述例子中，对 `putStr` 或 `writeFile` 的调用是否会强制将整个输入字符串立刻加载到内存中，仅仅只是为了输出？

回答是不。`putStr`（以及其它类似的所有输出函数）在数据可同时，输出数据。它们同样也不需要保留已经输出了的数据，因此只要程序中没有其它东西需要它，内存就可以立刻释放。在某种意义上，可以将 `readFile` 与 `writeFile` 之间的 `String` 视为连接两者的管道。数据进入一端，以某种方式转换后，从另一端流出。

interact

通过 `interact` 函数可以进一步简化我们的程序：

```
1 import Data.Char (toUpper)
2
3 main :: IO ()
4 main = interact $ map toUpper
```

仅用了一行！测试：

```
1 $ runghc toupper-lazy4.hs < input.txt > output.txt
```

或者想要打印到屏幕上：

```
1 $ runghc toupper-lazy4.hs < input.txt
```

如果想要看到 Haskell 的输出确实是在接收到数据块后立刻输出到数据块，可以不加任何参数运行 `runghc toupper-lazy4.hs`，这样就可以看到在敲击输入回车后，所有字符都会被立刻以大写的形式输出。

我们同样还可以通过 `interact` 编写在输出前添加新行的程序：

```
1 import Data.Char (toUpper)
2
3 main = interact $ map toUpper . (++) "Your data, in uppercase, is:\n\n"
```

由于我们在 `(++)` 后调用的 `map`，那么新行也会变为大写，修改一下：

```
1 import Data.Char (toUpper)
2
3 main = interact $ (++) "Your data, in uppercase, is:\n\n" . map toUpper
```

这就将新行移到 `map` 外去了。

interact 加上过滤

`interact` 的另一个常用方法就是过滤。比如打印带有字符“a”的每行：

```
1 main = interact $ unlines . filter (elem 'a') . lines
```

通过 `ghci` 查看上述的三个新函数：

```
1 ghci> :type lines
2 lines :: String -> [String]
3 ghci> :type unlines
4 unlines :: [String] -> String
5 ghci> :type elem
6 elem :: (Eq a) => a -> [a] -> Bool
```

测试：

```
1 $ runghc filter.hs < input.txt
2 I like Haskell
3 Haskell is great
```

IO 单子

Actions

大多数语言不会区分纯函数和非纯函数。Haskell 的函数有其数学意义：它们是存粹的计算，且不会被任何外界的东西改变。另外，计算可以在任何时刻进行。

那么就需要另一些工具来处理 I/O 了。这个工具在 Haskell 中叫做 *actions*。Actions 类似于函数，在定义的时候不会做任何事情，而是在执行任务时才会被唤醒。I/O actions 被定义在 **IO** 单子中。我们会在第十四章中讲解单子。看看一些类型：

```
1 ghci> :type putStrLn
2 putStrLn :: String -> IO ()
3 ghci> :type getLine
4 getLine :: IO String
```

`putStrLn` 的类型跟其它函数无异。函数接受一个参数并返回一个 `IO ()`。这个 `IO ()` 就是 action。我们可以存储以及传递 actions 至纯代码中，尽管这并不常用。一个 action 不会做任何事情直到它被唤起。看看例子：

```
1 str2action :: String -> IO ()
2 str2action input = putStrLn $ "Data: " ++ input
3
4 list2actions :: [String] -> [IO ()]
5 list2actions = map str2action
6
7 numbers :: [Int]
8 numbers = [1 .. 10]
9
10 strings :: [String]
11 strings = map show numbers
12
13 actions :: [IO ()]
14 actions = list2actions strings
15
16 printitall :: IO ()
17 printitall = runall actions
18
19 -- Take a list of actions, and execute each of them in turn.
20 runall :: [IO ()] -> IO ()
21 runall [] = return ()
22 runall (fst : remaining) = do
23     fst
24     runall remaining
25
26 main :: IO ()
27 main = do
28     str2action "Start of the program"
29     printitall
```

```
30 str2action "Done!"
```

`str2action` 这个函数接受一个参数并返回一个 `IO ()`。在 `main` 的结尾，可以在另一个 action 中直接使用它，并打印出一行。或者可以存储 – 而不是执行 – 这个 action 到纯代码中。在 `list2actions` 中 – 使用了 `map` 应用在 `str2action` 上并返回一个列表的 actions，正如我们在纯代码中的那样。可以看到通过 `printitall` 的所有内容都是使用纯的工具构建的。

尽管定义了 `printitall`，它不会被执行直到它的 action 在某处被计算。注意在 `main` 中是如何将 `str2action` 作为一个 I/O action 被执行的，而之前我们在 I/O 单子外使用它，并将结果转为一个列表。

可以这么理解：每个声明，除了 `let`，在一个 `do` 代码块中必须产生一个将要执行的 I/O action。

对 `printitall` 的调用最终执行所有这些操作。实际上由于 Haskell 是惰性的，所以直到这里才会生成 actions。

测试：

```
1 % runhaskell actions.hs
2 Data: Start of the program
3 Data: 1
4 Data: 2
5 Data: 3
6 Data: 4
7 Data: 5
8 Data: 6
9 Data: 7
10 Data: 8
11 Data: 9
12 Data: 10
13 Data: Done!
```

实际上我们可以将其写成更紧凑的形式。以下是重构：

```
1 str2message :: String -> String
2 str2message input = "Data: " ++ input
3
4 str2action :: String -> IO ()
5 str2action = putStrLn . str2message
6
7 numbers :: [Int]
8 numbers = [1 .. 10]
9
10 main :: IO ()
11 main = do
12     str2action "Start of the program"
13     mapM_ (str2action . show) numbers
14     str2action "Done!"
```

注意 `str2action` 中使用了标准函数组合操作符。在 `main` 中，调用了 `mapM_`，该函数类似于 `map` 接受一个函数与一个列表。而提供给 `mapM_` 的函数则是一个应用至列表中每个元素的 I/O action。`mapM_` 会抛出函数的结果，不过如果有需要，也可以使用 `mapM` 返回 I/O 结果的列表。看一下它们的类型：

```
1 ghci> :type mapM
2 mapM :: (Monad m) => (a -> m b) -> [a] -> m [b]
3 ghci> :type mapM_
4 mapM_ :: (Monad m) => (a -> m b) -> [a] -> m ()
```

Tip

实际上这些函数并不仅仅作用于 I/O；它们可用于任何 `Monad`。

为什么有了 `map` 还需要 `mapM` 呢？因为 `map` 是一个返回列表的纯函数，它不能直接执行 actions，而 `mapM` 则是 IO 单子中的工具，可用于实际执行 actions。

回到 `main`，`mapM_` 应用了 `(str2action . show)` 在每个 `numbers` 中的元素上。`show` 将每个数值转为 `String`，而 `str2action` 转换每个 `String` 为一个 action。`mapM_` 结合这些独立的 actions 成为一个大的 action 并打印出来。

Sequencing

`do` 代码块其实是合并所有 actions 的简便注解。在不使用 `do` 时还有两个操作符可以使用：`>>` 与 `>>=`。

```
1 ghci> :type (>>)
2 (>>) :: (Monad m) => m a -> m b -> m b
3 ghci> :type (>>=)
4 (>>=) :: (Monad m) => m a -> (a -> m b) -> m b
```

`>>` 操作符将两个 actions 串联起来：先执行第一个 action，再执行第二个 action。整个计算的结果则是第二个 action 的结果，而第一个 action 的结果则被丢弃。这类似于 `do` 代码块中的一行。可以用 `putStrLn "line 1" >> putStrLn "line 2"` 来测试一下。它将会打印出两行，将第一个 `putStrLn` 的结果丢弃，将第二个的结果作为返回。

`>>=` 操作符运行一个 action，然后将其结果传给一个返回 action 的函数。第二个 action 也将运行，整个表达式的结果还是第二个 action 的结果。例如，可以用 `getLine >>= putStrLn` 测试，它从键盘读取一行然后再将其输出。

现在不用 `do` 代码块来重写一下之前的例子：

```
1 main :: IO ()
2 main = do
3     putStrLn "Greetings! What is your name?"
```

```

4 inpStr <- getLine
5 putStrLn $ "Welcome to Haskell, " ++ inpStr ++ "!"

```

不使用 `do` :

```

1 main :: IO ()
2 main =
3   putStrLn "Greetings! What is your name?"
4   >> getLine
5   >>= (\inpStr -> putStrLn $ "Welcome to Haskell, " ++ inpStr ++ "!")

```

在定义 `do` 代码块时, Haskell 编译器内部会执行类似的转换。

Return 的本质

在本章开头我们提到了 `return` 可能并不像是它长得那样。很多语言的关键字 `return` 是立刻终结函数的执行并返回一个值给到调用者。

Haskell 中的 `return` 函数则大相径庭。Haskell 中的 `return` 将数据包装进单子中。在谈及 I/O 时, `return` 用于将纯数据放入 IO 单子中。

为什么要这么做呢? 记住, 任何结果依赖于 I/O 的东西都必须在 IO 单子内。因此, 如果我们正在编写一个执行 I/O 的函数, 那么一个纯计算, 我们需要使用 `return` 来使这个纯计算成为函数的正确返回值。例如:

```

1 import Data.Char (toUpper)
2
3 isGreen :: IO Bool
4 isGreen = do
5   putStrLn "Is green your favorite color?"
6   inpStr <- getLine
7   return $ (toUpper . head $ inpStr) == 'Y'

```

我们有一个返回 `Bool` 的纯计算。计算传递给 `return`, 将其放入 IO 单子内。由于它是 `do` 代码块中的最后一个值, 它变成了 `isGreen` 的返回值, 但这并不因为我们使用了 `return` 函数。

下面是同样一个程序, 只不过将纯计算剥离成为一个独立的函数。这样可以隔离纯函数, 同样使得意图更加的清晰。

```

1 import Data.Char (toUpper)
2
3 isYes :: String -> Bool
4 isYes inpStr = (toUpper . head $ inpStr) == 'Y'
5
6 isGreen :: IO Bool
7 isGreen = do
8   putStrLn "Is green your favorite color?"
9   isYes <$> getLine

```

与原文不同之处在于使用了 `<$>`，即 `fmap` 函数，替换掉了之前的 `inpStr <- getLine` 以及 `return`。

最后是一个做作的例子用于展示 `return` 并不需要位于 `do` 代码块的结尾。

```
1 returnTest :: IO ()
2 returnTest =
3     do one <- return 1
4         let two = 2
5         putStrLn $ show (one + two)
```

Haskell 真的是命令式的吗？

这些 `do` 代码块看起来像是命令式的语言。毕竟大多数情况下我们都是在按顺序运行命令。

但是 Haskell 本质上仍然是一种惰性语言。虽然有时有必要对 I/O 操作进行排序，但这是使用了 Haskell 中的工具来完成的。Haskell 还通过 IO 单子实现了 I/O 与语言其他部分的良好分离。

Lazy I/O 的副作用

本章提到的 `hGetContents`，其返回的 `String` 可以用在纯代码中。

我们需要更具体的了解副作用是什么。当我们说 Haskell 没有副作用时，这到底是什么意思？

某种层度上，副作用总是可能的。一个没有优化的循环，即使是用纯代码写的，也有可能导致系统的内存耗尽然后导致程序崩溃。或者它可能导致数据被交换到磁盘。

当我们提到没有副作用，意思其实是 Haskell 的纯代码不会运行能触发副作用的命令。纯函数不会修改一个全局变量，I/O 请求，或者运行一个命令关闭系统。

当你有一个 `hGetContents` 而来的字符串被传递至一个纯函数时，函数不会知道 `String` 是从磁盘文件而来的。它只会表现得跟通常那样，但处理该 `String` 可能会导致环境发出 I/O 命令。纯函数不会这么做；它们是纯函数处理的结果，就像将内存交换到磁盘的例子一样。

在某些情况下，你可能需要更多的控制 I/O 发生的确切时间。也许是正在交互式的从用户那里读取数据，或者通过管道从另一个程序读取数据，并需要直接与用户通信。这些情况下，`hGetContents` 可能不合适。

缓存

缓存模式

Haskell 有三种不同的缓存模式：`BufferMode` 类型：`NoBuffering`，`LineBuffering` 以及 `BlockBuffering`。

`NoBuffering` 正如其名 – 无缓存。通过例如 `hGetLine` 等函数读取的数据将每次从操作系统重读取一个字符。写入的数据将立刻写入，也经常一次写入一个字符。通常 `NoBuffering` 的性能很差，不适合通用用途。

`LineBuffering` 使输出缓冲区在输出换行符或缓存过大时被写入。在输入时，它通常会尝试读取任何可用的数据块，直到它第一次看到换行符。当从终端读取数据时，每次按下回车键后应该立刻返回数据。这通常是合理的默认值。

`BlockBuffering` 使 Haskell 在可能的情况下以固定大小块来读写数据。在批量处理大量数据时，这是性能最好的方法，即使这些数据是以行记录的。然而它不能用于交互式程序，因为它会阻塞输入，直到读取到完整的块。`BlockBuffering` 接受 `Maybe` 类型的参数：如果 `Nothing`，它将使用定义好的缓存大小；或者你可以使用 `Just 4096` 之类的设置将缓存设置为 4096 字节。

默认和缓存模式取决于操作系统和 Haskell 的实现。可以通过调用 `hGetBuffering` 向系统询问当前的缓存模式。当前的缓存模式可以通过 `hSetBuffering` 设置，它接受 `Handle` 和 `BufferMode`。例如 `hSetBuffering stdin (BlockBuffering Nothing)`。

刷新缓存

对于任何类型的缓存，有时我们会希望强制 Haskell 写出保存在缓存中的任何数据。有些时候这会自动发生：比如说调用 `hClose` 时。而调用 `hFlush` 将强制立刻写入任何挂起的数据。当 `Handle` 是一个网络 socket 且想要立刻传输数据时，或者是想要将磁盘上的数据提供给可能并发读取它的其它程序时，这会很有用。

读取命令行参数

`System.Environment.getArgs` 以 `IO [String]` 返回所有参数。类似 C 里的 `argv`，始于 `argv[1]`。程序名 (C 中的 `argv[0]`) 则可以通过 `System.Environment.getProgName` 得到。

`System.Console.GetOpt` 模块提供了一些解析命令行选项的工具。如果我们的程序有一些复杂的选项，该模块则很有帮助。

环境变量

读取环境变量可以用 `System.Environment` 的两个函数：`getEnv` 或 `getEnvironment`。前者查找特定变量，变量不存在时抛出异常；后者将所有环境变量以 `[(String, String)]` 输出，可以使用例如 `lookup` 函数来找到所需的变量。

8 Efficient file processing, regular expressions, and file name matching

高效文件处理

下面是一个简单的微基准测试，读取一个全是数字的文本文件，打印它们的总和：

```
1 main :: IO ()
2 main = do
3     contents <- getContents
4     print $ sumFile contents
5     where
6         sumFile = sum . map read . words
```

尽管 `String` 是用于读写文件的默认类型，它并不高效，因此像这样一个简单的程序性能是很差的。

`String` 代表着 `Char` 值的列表；列表中每个元素的内存分配都是独立的，需要额外的开销。这些因素会影响必须读写文本或二进制数据程序的内存消耗和性能。像这样的简单基准测试，即使是用解释性语言（如 Python）编写的程序，性能也比使用 `String` 的 Haskell 代码高出一个数量级。

`bytestring` 库提供了一个快速并且又低开销的 `String` 类型替代品。通过 `bytestring` 编写的代码通常可以媲美或超过 C 语言的性能和内存消耗，同时还保持了 Haskell 的表现力和简洁性。

该库提供了两个模块。每个定义的函数几乎都是 `String` 对应函数的直接替代品：

- `Data.ByteString` 模块定义了一个严格 *strict* 类型的 `ByteString`。其以二进制或文本数据的数组形式来表示字符串。
- `Data.ByteString.Lazy` 模块提供了惰性 *lazy* 类型的 `ByteString`。其以块 *chunks* 数组来表示字符串，最大大小为 64KB。

两个 `ByteString` 类型在特定环境都有更好的表现。对于流式处理一个大数（几百 MB 至 TB），惰性 `ByteString` 类型表现的最好。它的块大小被调整为适合现代 CPU 的 L1 缓存，且垃圾回收可以快速丢弃不再使用的流数据块。

而严格的 `ByteString` 类型在不太关心内存占用或需要随机访问数据的环境下性能更好。

二进制 I/O 与 qualified imports

现在来开发一个函数来说明 `ByteString` 的一些 API。我们将决定一个文件是否为 ELF 对象：这种格式多用于现代 Unix 类型系统的可执行文件。

查看文件的首四个字节，检查它们是否匹配一个字节列表：

```

1 import Data.ByteString.Lazy qualified as L
2
3 hasElfMagic :: L.ByteString -> Bool
4 hasElfMagic content = L.take 4 content == elfMagic
5 where
6     elfMagic = L.pack [0x7f, 0x45, 0x4c, 0x46]

```

`ByteString` 的惰性模块以及严格模块意图解决二进制 I/O。Haskell 用 `Word8` 来展示字节的数据类型；可以通过 `Data.Word` 来引入它。

上述例子的 `L.pack` 函数接受一个 `Word8` 列表，打包它们成为一个惰性的 `ByteString`。（`L.unpack` 函数作用相反。）而 `hasElfMagic` 函数则是比较 `ByteString` 的前四个字符。

以下是将其运用在一个文件上的例子：

```

1 isElfFile :: FilePath -> IO Bool
2 isElfFile path = do
3     content <- L.readFile path
4     return $ hasElfMagic content

```

`L.readFile` 函数是 `readFile` 的惰性 `ByteString` 版本，即按需读取。它效率很高，每次读取最大 64KB。惰性 `ByteString` 很适合我们的任务：因为我们至多读取文件的前四个字节，因此我们可以安全的使用该函数在任何大小的文件上。

Text I/O

`bytestring` 库还提供了两个模块用于 text I/O 功能，`Data.ByteString.Char8` 以及 `Data.ByteString.Lazy.Char8`。它们将单独的字符串元素导出成 `Char` 而不是 `Word8`。

Warning

上述模块中的函数进作用于字节大小的 `Char` 值，也就是仅适用于 ASCII 以及某些欧洲字符集。超过 255 的字符会被截断。

面向字符的 `bytestring` 模块为文本处理提供了一些有用的函数。以下是包含了月度的股票价格的文件：

```

1 ghci> putStr =<< readFile "prices.csv"
2 Date,Open,High,Low,Close,Volume,Adj Close
3 2008-08-01,20.09,20.12,19.53,19.80,19777000,19.80
4 2008-06-30,21.12,21.20,20.60,20.66,17173500,20.66
5 2008-05-30,27.07,27.10,26.63,26.76,17754100,26.76
6 2008-04-30,27.17,27.78,26.76,27.41,30597400,27.41

```

如何找到最高价呢？收盘价在第四列，用都好分隔。下面是得到收盘价的函数：

```

1 closing = readPrice . (!! 4) . L.split ','

```

该函数是 point-free 风格，需要从右往左阅读。`L.split` 将一个惰性的 `ByteString` 分隔开来，分隔发生在每一次找到匹配值时。`(!!)` 操作符则是获取列表中第 `k` 个元素。`readPrice` 函数将一个代表分数的字符串转换为数值：

```
1 readPrice :: L.ByteString -> Maybe Int
2 readPrice str = case L.readInt str of
3   Nothing -> Nothing
4   Just (dollars, rest) ->
5     case L.readInt (L.tail rest) of
6       Nothing -> Nothing
7       Just (cents, more) ->
8         Just (dollars * 100 + cents)
```

上面使用了 `L.readInt` 函数，即解析整数，返回整数以及字符串剩余部分。
找到最高收盘价的函数：

```
1 highestClose = maximum . (Nothing :) . map closing . L.lines
2
3 highestCloseFrom path = do
4   contents <- L.readFile path
5   print $ highestClose contents
```

这里使用了一个小技巧。当我们提供一个空列表给 `maximum` 函数会抛出异常：

```
1 ghci> maximum [3,6,2,9]
2 9
3 ghci> maximum []
4 *** Exception: Prelude.maximum: empty list
```

测试：

```
1 ghci> :load HighestClose
2 [1 of 1] Compiling Main             ( HighestClose.hs, interpreted )
3 Ok, modules loaded: Main.
4 ghci> highestCloseFrom "prices.csv"
5 Loading package array-0.1.0.0 ... linking ... done.
6 Loading package bytestring-0.9.0.1 ... linking ... done.
7 Just 2741
```

因为从逻辑中分离了 I/O，测试空数据时不需要创建一个空文件：

```
1 ghci> highestClose L.empty
2 Nothing
```

文件名匹配

很多面向系统的编程语言都提供了根据模式来匹配文件的功能。尽管 Haskell 的标准库有不错的系统编程工具，它并没有提供这类的匹配函数。

这类模式通常被称为 `glob` 模式，通配符模式或者 `shell` 样式模式。它们有一些简单的规则：

- 将字符串与模式进行匹配，从字符串的开头开始，到末尾结束。
- 大多字面字符都匹配自己。例如，模式中的文本 `foo` 将匹配输入字符串中的 `foo`，且仅匹配 `foo`。
- `*`（星号）字符表示“匹配任何内容”；它将匹配任何文本，包括空字符串。
- `?`（问号）字符匹配任何单个字符。模式 `pic?? .jpg` 将匹配 `picaa.jpg` 或 `pic01.jpg` 等名称。
- `[`（开方括号）字符开始一个字符类，以 `]` 结束。它的意思是“匹配这个类中的任何字符”。字符类可以在 `[` 后面加一个 `!` 来表示“匹配不属于这个类的任何字符”。

作为一种简写，一个字符后面跟着一个 `-`（破折号），后面跟着另一个字符，表示一个范围：“匹配此集合中的任何字符”

虽然 Haskell 的标准库中没有提供 `glob` 模式的匹配，但是它提供了一个非常好的正则表达式库。

Haskell 中的正则表达式

首先让我们加载 `Text.Regex.Posix` 库

```
1 ghci> :module +Text.Regex.Posix
```

根据官方 Wiki 提到的 `regex-posix` 速度非常的慢，在生产代码中需要换成别的库。

我们这里只需要正则表达式匹配的函数，一个中缀操作符 `(=~)` 即可（从 Perl 中借用）。要克服的第一个障碍是 Haskell 的 `regex` 库大量使用了多态。因此 `(=~)` 操作符的类型签名很难理解，因此这里暂时不做解释。

`=~` 对它的两个参数和返回类型使用了 `typeclasses`。第一个参数（即 `=~` 左侧）是要匹配的文本；第二个参数（右侧）则是要匹配的正则表达式。我们可以传 `String` 或 `ByteString` 作为参数。

结果的多种类型

`=~` 操作符的结果类型是多态的，因此 Haskell 编译器需要知道其类型。在真实代码中，它有可能根据使用的场景被推导出正确的类型。但是 `ghci` 没法这么做，它没有足够的信息来进行推导。

当 `ghci` 不能推导 `target` 类型时，我们需要进行显式指定：

```

1 ghci> "my left foot" =~ "foo" :: Bool
2 Loading package array-0.1.0.0 ... linking ... done.
3 Loading package containers-0.1.0.1 ... linking ... done.
4 Loading package bytestring-0.9.0.1 ... linking ... done.
5 Loading package mtl-1.1.0.0 ... linking ... done.
6 Loading package regex-base-0.93.1 ... linking ... done.
7 Loading package regex-posix-0.93.1 ... linking ... done.
8 True
9 ghci> "your right hand" =~ "bar" :: Bool
10 False
11 ghci> "your right hand" =~ "(hand|foot)" :: Bool
12 True

```

正则表达式库的核心是一个名为 `RegexContext` 的 typeclass，用于描述 `target` 类型的行为；基础库为我们定义了很多实例。`Bool` 类型就定义了该 typeclass 的实例，因此我们能得到可用的结果。另一个实例就是 `Int`：

```

1 ghci> "a star called henry" =~ "planet" :: Int
2 0
3 ghci> "honorificabilitudinitatibus" =~ "[aeiou]" :: Int
4 13

```

如果寻求的是 `String` 结果，我们将得到第一个匹配的子字符串，或是不匹配时得到一个空字符串。

```

1 ghci> "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" :: String
2 "ii"
3 ghci> "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" :: String
4 ""

```

另外一个可作为结果的类型是 `[String]`，其返回所有匹配字符串的列表：

```

1 ghci> "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" :: [String]
2
3 <interactive>:1:0:
4   No instance for (RegexContext Regex [Char] [String])
5     arising from a use of `=~' at <interactive>:1:0-50
6   Possible fix:
7     add an instance declaration for
8     (RegexContext Regex [Char] [String])
9   In the expression:
10     "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" :: [String]
11   In the definition of `it':
12     it = "I, B. Ionsonii, uurit a lift'd batch" =~ "(u|ii)" ::
13         [String]
14 ghci> "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" :: [String]
15
16 <interactive>:1:0:
17   No instance for (RegexContext Regex [Char] [String])
18     arising from a use of `=~' at <interactive>:1:0-54

```

```

19     Possible fix:
20         add an instance declaration for
21         (RegexContext Regex [Char] [String])
22     In the expression:
23         "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" :: [String]
24     In the definition of `it`:
25         it = "hi ludi, F. Baconis nati, tuiti orbi" =~ "Shakespeare" ::
26             [String]

```

Tip

注意字符串返回

如果期望结果是一个 `String`，那么就要小心了。因为 `(=~)` 返回一个空字符串来表示“没有匹配”，如果空字符串也可能是 `regex` 的有效匹配，那么就会难以识别。在这种情况下就应该是有一个不同的返回类型，例如 `[String]`。

上述都是“简单的”返回类型，这远没有结束。继续往下推进前，让我们定义一个在之后例子中都会用到的匹配：

```

1 ghci> let pat = "(foo[a-z]*bar|quux)"

```

我们可以获得关于匹配发生的上下文的大量信息。如果期望是 `(String,String,String)` 元组，则会在第一个匹配之前获得文本，匹配成功的文本，以及紧随其后的文本。

```

1 ghci> "before foodiebar after" =~ pat :: (String,String,String)
2 ("before ", "foodiebar", " after")

```

如果匹配失败，那么整个文本则会在元组中以“之前”元素作为返回，而其它两个元素为空：

```

1 ghci> "no match here" =~ pat :: (String,String,String)
2 ("no match here", "", "")

```

请求一个四元素元组则会给我们第四个元素，它是匹配模式中所有组的列表：

```

1 ghci> "before foodiebar after" =~ pat :: (String,String,String,[String])
2 ("before ", "foodiebar", " after", ["foodiebar"])

```

我们还可以获取匹配的数值信息。一对 `Int` 给出的第一个匹配的起始偏移量以及长度。如果我们请求这些配对的列表，我们将得到所有匹配的信息。

```

1 ghci> "before foodiebar after" =~ pat :: (Int,Int)
2 (7,9)
3 ghci> "i foobarbar a quux" =~ pat :: [(Int,Int)]
4
5 <interactive>:1:0:
6   No instance for (RegexContext Regex [Char] [(Int, Int)])
7   arising from a use of `=~' at <interactive>:1:0-26
8   Possible fix:

```



```

9      add an instance declaration for
10      (RegexContext Regex [Char] [(Int, Int)])
11      In the expression: "i foobarbar a quux" =~ pat :: [(Int, Int)]
12      In the definition of `it`:
13      it = "i foobarbar a quux" =~ pat :: [(Int, Int)]

```

如果请求单个元组，则用值 `-1` 作为元组的第一个元素（匹配偏移量）表示匹配失败；如果请求元组列表，则用空列表表示匹配失败：

```

1  ghci> "eleemosynary" =~ pat :: (Int,Int)
2  (-1,0)
3  ghci> "mondegreen" =~ pat :: [(Int,Int)]
4
5  <interactive>:1:0:
6      No instance for (RegexContext Regex [Char] [(Int, Int)])
7      arising from a use of `=~' at <interactive>:1:0-18
8      Possible fix:
9      add an instance declaration for
10      (RegexContext Regex [Char] [(Int, Int)])
11      In the expression: "mondegreen" =~ pat :: [(Int, Int)]
12      In the definition of `it`: it = "mondegreen" =~ pat :: [(Int, Int)]

```

更多的关于正则表达式

混合与匹配字符串类型

之前提到 `=~` 操作符使用 `typeclasses` 作为其参数类型以及返回类型。可以使用 `String` 或严格 `ByteString` 的值来进行正则表达式计算。

```

1  ghci> :module +Data.ByteString.Char8
2  ghci> :type pack "foo"
3  pack "foo" :: ByteString

```

可以尝试不同组合的 `String` 与 `ByteString`：

```

1  ghci> pack "foo" =~ "bar" :: Bool
2  False
3  ghci> "foo" =~ pack "bar" :: Int
4  0
5  ghci> pack "foo" =~ pack "o" :: [(Int, Int)]
6
7  <interactive>:1:0:
8      No instance for (RegexContext Regex ByteString [(Int, Int)])
9      arising from a use of `=~' at <interactive>:1:0-21
10     Possible fix:
11     add an instance declaration for
12     (RegexContext Regex ByteString [(Int, Int)])
13     In the expression: pack "foo" =~ pack "o" :: [(Int, Int)]

```

```

14     In the definition of `it':
15     it = pack "foo" =~ pack "o" :: [(Int, Int)]

```

不过需要注意的是，如果期望匹配结果中包含字符串值，那么所匹配的文本必须是同一类型的字符串：

```

1  ghci> pack "good food" =~ ".ood" :: [ByteString]
2
3  <interactive>:1:0:
4      No instance for (RegexContext Regex ByteString [ByteString])
5      arising from a use of `=~' at <interactive>:1:0-25
6      Possible fix:
7      add an instance declaration for
8      (RegexContext Regex ByteString [ByteString])
9      In the expression: pack "good food" =~ ".ood" :: [ByteString]
10     In the definition of `it':
11     it = pack "good food" =~ ".ood" :: [ByteString]

```

上述例子中，使用 `pack` 将 `String` 转为了 `ByteString`。类型检查器可以接受是因为 `ByteString` 在结果类型中展示出来了。但是如果尝试获取一个 `String`，则不会成功：

```

1  ghci> "good food" =~ ".ood" :: [ByteString]
2
3  <interactive>:1:0:
4      No instance for (RegexContext Regex [Char] [ByteString])
5      arising from a use of `=~' at <interactive>:1:0-20
6      Possible fix:
7      add an instance declaration for
8      (RegexContext Regex [Char] [ByteString])
9      In the expression: "good food" =~ ".ood" :: [ByteString]
10     In the definition of `it':
11     it = "good food" =~ ".ood" :: [ByteString]

```

我们可以通过将左侧的字符串类型与结果再次匹配来简单的修复这个问题：

```

1  ghci> "good food" =~ ".ood" :: [String]
2
3  <interactive>:1:0:
4      No instance for (RegexContext Regex [Char] [String])
5      arising from a use of `=~' at <interactive>:1:0-20
6      Possible fix:
7      add an instance declaration for
8      (RegexContext Regex [Char] [String])
9      In the expression: "good food" =~ ".ood" :: [String]
10     In the definition of `it': it = "good food" =~ ".ood" :: [String]

```

其它需要注意的事项

查看 Haskell 库文档时，可以看到几个与 `regex` 相关的模块。`Text.Regex.Base` 模块定义了所有其他 `regex` 模块所遵循的通用 API。可以同时安装 `regex` API 的多个实现。在编写本文时，GHC 与一个实现 `Text.Regex.Posix` 捆绑在一起。顾名思义，这个包提供 POSIX `regex` 语义。

Tip

Perl 与 POSIX 正则表达式

如果是从 Perl、Python 或 Java 等语言来学习 Haskell 的，并且在其中一种语言中使用过正则表达式，那么应该意识到由 `Text.Regex.Posix` 模块处理的 POSIX `regex` 与 Perl 风格的 `regex` 在某些重要方面是不同的。以下是一些比较明显的区别。

Perl `regex` 引擎在匹配选项时执行左偏匹配，而 POSIX 引擎选择最贪婪的匹配。这意味着给定一个正则表达式 `(foo|fo*)` 和一个文本字符串 `foooooo`，perl 风格的引擎将给出 `foo` 的匹配（最左边的匹配），而 POSIX 引擎将匹配整个字符串（最贪婪的匹配）。

POSIX `regex` 的语法不如 perl 风格的 `regex` 统一。它们还缺乏 perl 风格的 `regex` 提供的许多功能，例如零宽度断言和对贪婪匹配的控制。

将一个 glob 模式转为一个正则表达式

我们已经见识到了很多种将文本域正则表达式匹配的方法，让我们将注意力转回到 glob 模式。我们希望编写一个接受一个 glob 模式的函数，将其表示形式返回为正则表达式。glob 模式和正则表达式两者都是 `text` 字符串，因此函数的类型就很明确了：

```
1 module GlobRegex
2   ( globToRegex,
3   )
4 where
5
6 import Text.Regex.Posix ((=))
7
8 globToRegex :: String -> String
```

我们生成的正则表达式必须是固定的，使得它可以从同到尾的匹配字符串。

```
1 globToRegex cs = '^' : globToRegex' cs ++ "$"
```

由于 `String` 就是 `[Char]` 的别名，这里 `:` 操作符将一个值（这里是 `^` 字符）放到一个列表的头部，而这个列表则是还未出现的 `globToRegex` 函数所返回的。

Tip

在定义一个值之前使用它

Haskell 使用一个值或者函数时，并不需要它们被声明或者在一个源文件中定义。一个定义出现在第一个被使用的地方之后是很正常的。Haskell 编译器并不关心这一层面上的顺序。这使我们可以灵活的按照逻辑来构建代码，而不是遵循编译器编写者的顺序。

Haskell 模块的编写者通常会使用这个灵活性来让“更加重要”的代码前置于源文件中。这正式我们编写 `globToRegex` 函数以及其帮助函数的方式。

有了正则表达式以后 `globToRegex` 函数则用作于大部分的翻译工作。以下使用 Haskell 的模式匹配来进行编码：

```
1 globToRegex' :: String -> String
2 globToRegex' "" = ""
3 globToRegex' ('*' : cs) = ".*" ++ globToRegex' cs
4 globToRegex' ('?' : cs) = '.' : globToRegex' cs
5 globToRegex' ('[' : '!' : c : cs) = "[" ++ c : charClass cs
6 globToRegex' ('[' : c : cs) = '[' : c : charClass cs
7 globToRegex' ('[' : _) = error "unterminated character class"
8 globToRegex' (c : cs) = escape c ++ globToRegex' cs
```

第一个子句规定了如果达到了 glob 模式的末尾（到那时会看到空字符串），返回 `$`，即正则表达式中的“匹配到了末尾”。接下来的是的子句则是一系列从 glob 语义切换到正则表达语义的模式。最后一个子句处理其它的每个字符，转义优先。

`escape` 函数确保正则引擎不会解释某些特定字符：

```
1 escape :: Char -> String
2 escape c
3   | c `elem` regexChars = '\\' : [c]
4   | otherwise = [c]
5 where
6   regexChars = "\\+()~$.{}|!"
```

`charClass` 帮助函数用于检测字符是否正确的终止，不修改输入直到遇到 `]`：

```
1 charClass :: String -> String
2 charClass (']' : cs) = ']' : globToRegex' cs
3 charClass (c : cs) = c : charClass cs
4 charClass [] = error "unterminated character class"
```

我们已经定义好了 `globToRegex` 与其帮助函数们，现在加载进 `ghci` 中测试一下：

```
1 ghci> :set -package regex-posix
2 package flags have changed, resetting and loading new packages...
3 ghci> :l GlobRegex.hs
4 [1 of 1] Compiling GlobRegex      ( GlobRegex.hs, interpreted )
```

```

5 Ok, one module loaded.
6 ghci> globToRegex "f??.c"
7 "^f\\.\\.\\.c$"

```

看起来像是一个合理的正则，那么用它来匹配字符串呢？

```

1 ghci> "foo.c" =~ globToRegex "f??.c" :: Bool
2 True
3 ghci> "test.c" =~ globToRegex "t[ea]s*" :: Bool
4 True
5 ghci> "teste.txt" =~ globToRegex "t[ea]s*" :: Bool
6 True

```

正常工作！我们还可以为 `fnmatch` 创建一个临时定义：

```

1 ghci> let fnmatch pat name = name =~ globToRegex pat :: Bool
2 ghci> :type fnmatch
3 fnmatch
4   :: rgx-bs-0.94.0.2-01560e7d:Text.Regex.Base.RegexLike.RegexLike
5     Text.Regex.Posix.Wrap.Regex source1 =>
6     String -> source1 -> Bool
7 ghci> fnmatch "d*" "myname"
8 False

```

`fnmatch` 并不符合“Haskell 本性”，通常 Haskell 风格的函数都需要描述性的“驼峰”名称，例如“file name matches”为 `fileNameMatches`，那么我们的库可以这样命名：

```

1 matchesGlob :: FilePath -> String -> Bool
2 name `matchesGlob` pat = name =~ globToRegex pat

```

一个重要的旁白：编写惰性函数

在一个命令式的语言中，`globToRegex'` 函数通常会表述为一个循环。例如 Python 的标准 `fnmatch` 模块中包含了一个名为 `translate` 的函数，其功能与我们的 `globToRegex` 函数一致。它的实现就是循环。

如果你有别的函数式编程的经验例如 Scheme 或 ML，那么肯定有“模仿循环的方法就是通过尾递归”这个概念。

再来看一下 `globToRegex'` 函数，并不是一个尾递归函数。可以测试一下最后的子句（其它子句结构都类似）。

```

1 globToRegex' (c : cs) = escape c ++ globToRegex' cs

```

它递归的应用了自身，且递归的结果作为一个参数在 `(++)` 函数上。由于递归的应用并非作为函数的最后处理部分，那么 `globToRegex'` 并不是一个尾递归。

那么为什么这个函数并非尾递归呢？答案就在 Haskell 的非严格计算策略。在提及这个话题之前，让我们快速的了解为什么在传统语言中，我们需要避免这样的尾递归定义。以下是 `(++)` 操作符的一个简单定义。它属于递归，但不是尾递归。

```

1  (++) :: [a] -> [a] -> [a]
2
3  (x:xs) ++ ys = x : (xs ++ ys)
4  []      ++ ys = ys

```

在严格执行语言中，如果我们计算 `"foo" ++ "bar"`，整个列表会被构建出来，接着再返回。非严格计算则会延迟直到真正需要被用到的时候。

如果需要表达式 `"foo" + "bar"` 的一个元素时，函数定义的第一个模式将被匹配，接着返回表达式 `x : (xs ++ ys)`。因为 `(:)` 构造函数是非严格的，那么 `xs ++ ys` 的计算就能被递延：按需生产更多的元素。当我们生产更多的结果时，我们将不再使用 `x`，因此垃圾回收器将会进行回收。由于是按需生产元素，且不持有已经完成的部分，编译器就可以以常量的空间来计算我们的代码。

使用我们的模式匹配器

通过一个函数用于匹配 glob 模式是很好的，但是我们希望将其用于实际应用。在 Unix 系的系统中 `glob` 函数返回所有匹配到的文件名称以及路径。让我们在 Haskell 中也构建一个类似的函数：

```

1  module Glob (namesMatching) where

```

`System.Directory` 模块提供了用于处理路径以及内容的标准函数。

```

1  import System.Directory (doesDirectoryExist, doesFileExist, getCurrentDirectory,
    getDirectoryContents)

```

而 `System.FilePath` 模块抽象了操作系统路径名称转换的细节，`(</>)` 函数将两个路径连接在一起：

```

1  ghci> :m +System.FilePath
2  ghci> "foo" </> "bar"
3  "foo/bar"

```

`dropTrailingPathSeparator` 函数正如其名，去除末尾的路径分隔：

```

1  ghci> dropTrailingPathSeparator "foo/"
2  "foo"

```

`splitFileName` 则分离了路径与文件：

```

1  ghci> splitFileName "foo/bar/Quux.hs"
2  ("foo/bar/", "Quux.hs")

```

`System.Directory` 模块结合 `System.FilePath` 的使用，我们可以编写一个同时作用于 Unix 系与 Windows 的 `namesMatching` 函数：

```

1  import System.FilePath (dropTrailingPathSeparator, splitFileName, (</>))

```

在这个模块中，我们将模仿一个“for”循环；首次在 Haskell 中尝试错误处理；并使用写好的 `matchesGlob` 函数。

```
1 import Control.Exception (handle)
2 import Control.Monad (forM)
3 import GlobRegex (matchesGlob)
```

由于路径与文件存在于拥有副作用的“真实世界”，我们的 globbing 函数的返回值则必须带上 `IO`。

如果传入的字符串没有包含模式字符，那么就简单的检查给定的名称是否存在于文件系统重。（注意，我们在这里使用 Haskell 的函数守护语法来编写一个漂亮整洁的定义。使用“if”也是可以的，不过就没有这么美观了）

```
1 isPattern :: String -> Bool
2 isPattern = any (`elem` "[*?")
3
4 namesMatching pat
5   | not (isPattern pat) = do
6     exists <- doesNameExist pat
7     return ([pat | exists])
```

函数 `doesNameExist` 会在稍后进行定义。

那么如果字符串是一个 glob 模式呢？继续我们的函数定义：

```
1 | otherwise = do
2   case splitFileName pat of
3     ("", baseName) -> do
4       curDir <- getCurrentDirectory
5       listMatches curDir baseName
6     (dirName, baseName) -> do
7       dirs <-
8         if isPattern dirName
9           then namesMatching (dropTrailingPathSeparator dirName)
10          else return [dirName]
11       let listDir =
12         if isPattern baseName
13           then listMatches
14           else listPlain
15       pathNames <- forM dirs $ \dir -> do
16         baseNames <- listDir dir baseName
17         return (map (dir </>) baseNames)
18       return (concat pathNames)
```

我们使用了 `splitFileName` 将字符串分离成了一对“所有东西除了最终名称”以及“最终名称”。如果首个元素为空，则在当前路径寻找一个模式；否则检查路径名称是否包含模式，没有包含则创建一个包含路径名称的单例列表，包含则列举所有匹配的路径。

Tip

需要注意的事情

`System.FilePath` 模块可以变得有点棘手。上述的例子中 `splitFileName` 函数留下了末尾的斜杠

```
1 ghci> :module +System.FilePath
2 ghci> splitFileName "foo/bar"
3 ("foo/", "bar")
4
```

如果我们忘了（或者不了解）要移除斜杠，我们则会无限在 `namesMatching` 中递归，因为下列行为

```
1 ghci> splitFileName "foo/"
2 ("foo/", "")
3
```

你可以猜想一下之后将会发生什么！

最后，收集了所有匹配的路径，得到了一个列表的列表，再将它们打平成一个单独的列表。

这里陌生的 `forM` 函数其行为类似于一个“for”循环：它映射它的第二个参数（一个 action）在第一个参数（一个列表）上，并返回一个列表的结果。

还剩下一些需要处理的问题。首先是 `doesNameExist` 函数。`System.Directory` 模块并不能检查一个名称是否存在与文件系统中，它强制要求我们决定检查的是文件还是路径。这个 API 很丑陋，因此我们需要将两者合二为一。首先检查是否为文件，因为文件相较于路径更为普遍。

```
1 doesNameExist :: FilePath -> IO Bool
2 doesNameExist name = do
3   fileExists <- doesFileExist name
4   if fileExists
5     then return True
6     else doesDirectoryExist name
```

还有另外两个函数需要定义，它们的返回都是一个路径中的所有名称。`listMatches` 函数返回的是一个路径中匹配了给定 glob 模式的所有文件。

```
1 listMatches :: FilePath -> String -> IO [String]
2 listMatches dirName pat = do
3   dirName' <-
4     if null dirName
5       then getCurrentDirectory
6       else return dirName
7   handle errorHandler $ do
8     names <- getDirectoryContents dirName'
```



```

9     let names' =
10         if isHidden pat
11         then filter isHidden names
12         else filter (not . isHidden) names
13     return (filter (`matchesGlob` pat) names')
14 where
15     errorHandler :: SomeException -> IO [String]
16     errorHandler _ = return []
17
18 isHidden :: [Char] -> Bool
19 isHidden ('.' : _) = True
20 isHidden _ = False

```

与原文不同之处在于 `handle errorHandler` 这里需要显式的为 `errorHandler` 标注类型，否则原文的 `handle (const (return []))` 将会在编译时抛出类型不明确的异常，详见StackOverflow。

`listPlain` 函数的返回要么是一个空的列表要么是一个单例列表，这取决于传递的名称是否存在：

```

1 listPlain :: FilePath -> String -> IO [String]
2 listPlain dirName baseName = do
3     exists <-
4         if null baseName
5         then doesDirectoryExist dirName
6         else doesNameExist (dirName </> baseName)
7     return ([baseName | exists])

```

通过 API 设计来处理错误

如果传递给我们的 `globToRegex` 的是一个错误的模式，并不会是一场灾难。这有可能使用户输入了错误的模式，这种情况下我们需要报告有意义的错误信息。

```

1 type GlobError = String
2
3 globToRegex :: String -> Either GlobError String

```

运行我们的代码

`namesMatching` 函数并不能很好地单独工作，我们需要将其与其它函数结合。

首先定义一个 `renameWith` 函数，并非只是简单的为一个文件重命名，而是将一个函数应用至文件名称，用函数返回的值来重命名文件。

```

1 import System.FilePath (replaceExtensions)
2 import System.Directory (doesFileExist, renameDirectory, renameFile)
3 import Glob (namesMatching)

```

```

4
5 renameWith :: (FilePath -> FilePath) -> FilePath -> IO FilePath
6 renameWith f path = do
7   let path' = f path
8   rename path path'
9   return path'

```

这里还是需要一个帮助函数来处理文件与路径：

```

1 rename :: FilePath -> FilePath -> IO ()
2 rename old new = do
3   isFile <- doesFileExist old
4   let f = if isFile then renameFile else renameDirectory
5   f old new

```

`System.FilePath` 模块提供了很多操作文件名的函数，它们可以很好地与 `renameWith` 与 `namesMatching` 函数结合，使得我们可以快速的使用它们来创造复杂行为的函数。例如修改 C++ 源文件的后缀。

```

1 cc2cpp :: IO [FilePath]
2 cc2cpp = mapM (renameWith (flip replaceExtension ".cpp")) =<< namesMatching "*.cc"

```

`flip` 函数接受另一个函数作为参数，交换该函数的入参。`=<<` 函数将其右侧 action 的结果喂给左侧的 action。

9 I/O case study: a library for searching the filesystem

略

简单开始：递归展示一个路径

在设计我们的库之前，让我们先解决几个小问题。我们的第一个问题就是递归的列出一个路径的所有文件与子路径。

```

1  module RecursiveContents (getRecursiveContents) where
2
3  import Control.Monad (forM)
4  import System.Directory (doesDirectoryExist, getDirectoryContents)
5  import System.FilePath ((</>))
6
7  getRecursiveContents :: FilePath -> IO [FilePath]
8  getRecursiveContents topdir = do
9      names <- getDirectoryContents topdir
10     let properNames = filter (`notElem` [".", ".."]) names
11     paths <- forM properNames $ \name -> do
12         let path = topdir </> name
13         isDirectory <- doesDirectoryExist path
14         if isDirectory
15             then getRecursiveContents path
16             else return [path]
17     return (concat paths)

```

`filter` 表达式确保一个路径列表中不会包含特殊的路径名称，例如 `.` 或者 `..`。如果没有过滤它们则会有无限的递归。

这里又遇到了上一章的 `forM`；它等同于 `mapM` 只不过参数调转了：

```

1  ghci> :m +Control.Monad
2  ghci> :t mapM
3  mapM :: (Traversable t, Monad m) => (a -> m b) -> t a -> m (t b)
4  ghci> :t forM
5  forM :: (Traversable t, Monad m) => t a -> (a -> m b) -> m (t b)

```

与原文（如下）不同：

```

1  ghci> :m +Control.Monad
2  ghci> :type mapM
3  mapM :: (Monad m) => (a -> m b) -> [a] -> m [b]
4  ghci> :type forM
5  forM :: (Monad m) => [a] -> (a -> m b) -> m [b]

```

函数体中会检查当前输入是否为一个路径：如果是，则递归调用 `getRecursiveContents` 来列举该路径；否则，它返回一个单例列表，即当前输入的名称。（别忘了 `return` 函数在 Haskell 有独特的意义：它通过单子类型构造函数来包装一个值。）

另一个值得指出的点在于变量 `isDirectory` 的使用。一个命令式的语言例如 Python，通常会表述为 `if os.path.isdir(path)`，然而 `doesDirectoryExist` 函数是一个 *action*；它的返回类型是 `IO Bool` 而不是 `Bool`。又因为一个 `if` 表达式需要的是 `Bool` 类型，我们需要使用 `<-` 来获取被 `IO` 所包裹着的 *action* 的 `Bool` 返回，这样才能在 `if` 中使用解包过后的 `Bool`。

每个 `loop` 的便利都会返回一个名称列表，因此这里 `forM` 的结果是 `IO [[FilePath]]`。我们使用 `concat` 将其打平称为一个列表。

重访匿名与命名函数

在之前的匿名 (`lambda`) 函数的章节中，我们列举了一些不要使用匿名函数的原因，然而在这里我们使用了一个匿名函数作为 `loop` 的本体。这就是 Haskell 中最常见的匿名函数使用方法。

我们已经从 `forM` 与 `mapM` 的类型中得知它们接受函数作为参数。大多数的 `loop` 本体是一个在程序中出现一次的代码块。因为我们只会在一处使用这个 `loop` 本体，那么为什么要给它一个名称呢？

当然了，有时我们需要在若干地方部署同样的代码。这种情况下与其复制黏贴一个匿名函数，不如给已存在的函数一个名称。

为什么同时提供了 `mapM` 与 `forM` ？

同时存在两个除了参数顺序不同功能却一样的函数看起来有点奇怪，实际上它们适用于不同的场景。

考虑上述例子，使用匿名函数作为函数体。如果我们使用 `mapM` 而不是 `forM`，那么则需要将变量 `properNames` 放置在函数体之后。为了让代码能被正确的解析，我们还需要将整个匿名函数体用圆括号包裹起来，或者是一个命名函数来代替。

相反，如果 `loop` 本体就是一个命名函数，且需要遍历的列表是由一个复杂的表达式计算而得的时候，那么就有更好的理由使用 `mapM` 了。

一个简单的查询函数

我们可以是用 `getRecursiveContents` 函数作为一个简单的文件查询的基础：

```
1 import RecursiveContents (getRecursiveContents)
2
3 simpleFind :: (FilePath -> Bool) -> FilePath -> IO [FilePath]
4 simpleFind p path = do
5     names <- getRecursiveContents path
6     return $ filter p names
```

这个函数接受一个谓词，用于筛选 `getRecursiveContents` 所返回的名称。每个传递到谓词的名称都是一个完整的路径，那么我们该如何执行一个类似“找到所有后缀为 `.c` 的文件”这样的常规操作呢？

`System.FilePath` 模块包含了大量的宝贵的函数可以帮助我们操作文件名。上述案例可以用 `takeExtension` 。

```
1 ghci> :m +System.FilePath
2 ghci> :t takeExtension
3 takeExtension :: FilePath -> String
4 ghci> takeExtension "foo/bar.c"
5 ".c"
6 ghci> takeExtension "quux"
7 ""
```

这样我们就可以编写一个接受路径并提取后缀再与 `.c` 进行比较的函数了：

```
1 ghci> :l SimpleFinder
2 [1 of 2] Compiling RecursiveContents ( RecursiveContents.hs, interpreted )
3 [2 of 2] Compiling Main ( SimpleFinder.hs, interpreted )
4 Ok, two modules loaded.
5 ghci> :t simpleFind (\p -> takeExtension p == ".c")
6 simpleFind (\p -> takeExtension p == ".c")
7 :: FilePath -> IO [FilePath]
```

虽然 `simpleFind` 可以工作，不过还有一些瑕疵。首先是谓词并没有很强的表达力。它只能查看一个路径名称，而不能知晓输入的是一个文件还是路径。这就意味着尝试使用 `simpleFind` 将列出 `.c` 结尾的目录已经具有相同扩展名的文件。

其次就是 `simpleFind` 没有提供该如何遍历文件系统的控制。为什么这点很重要，可以考虑一下在拥有子版本管理的树形文件系统中，进行源文件的搜索。子版本在每个它管理的路径下都会维护一个私有的 `.svn` 路径；它们每一个包含了若干我们不需要考虑的子文件夹与文件。虽然我们可以简单的过滤掉任何包含 `.svn` 的情况，不过更有效的方式则是在第一时间就避免这些文件夹的遍历。

最后就是 `simpleFind` 是严格执行的，因为它包含了一系列在 `IO` 单子中所执行的 actions。如果我们有上百万的文件需要遍历，我们则需要等待很久并且还会得到一个巨大的包含了百万名称的结果。这对于资源管理与响应速度而言都非常的不好。我们更加倾向于一个惰性流式的结果。

谓词：从贫穷到富裕，同时保持 pure

我们的谓词只能查看文件名称。这就包含了大量有趣的行为：例如当我们想要展示的文件大于给定的大小时。

一个比较简单的做法就是修改 `IO`：谓词的类型不再是 `FilePath -> Bool`，而改为 `FilePath -> IO Bool`。这使得我们可以用任意 I/O 作为谓词函数的一部分。尽管这看起来

很吸引人，但它也有潜在的问题：这样的谓词可能会有任意的副作用，因为返回类型为 `I/O a` 的函数可以拥有任意的副作用。

让我们利用类型系统来编写更加可以预测，更是 bug 的代码：我们将通过避免“IO”污染来保持谓词的纯粹性。这样可以确保它们不会产生任何讨厌的副作用。同时我们还会喂给它们更多的信息，这样它们就能获得我们想要的表达能力，而不会成为潜在的危险。

Haskell 的 `System.Directory` 模块提供了一组有用的文件元数据。

```
1 ghci> :m +System.Directory
```

- 我们可以使用 `doesFileExist` 与 `doesDirectoryExist` 来判断一个输入是否是文件还是路径。近些年暂时还没有便捷的方法来查看其它的文件类型，例如命名通道，软硬链接等。

```
1 ghci> :t doesFileExist
2 doesFileExist :: FilePath -> IO Bool
3 ghci> doesFileExist "."
4 False
5 ghci> :t doesDirectoryExist
6 doesDirectoryExist :: FilePath -> IO Bool
7 ghci> doesDirectoryExist "."
8 True
9
```

- `getPermissions` 函数允许我们得知对于一个文件或者路径的操作是否合法

```
1 ghci> :t getPermissions
2 getPermissions :: FilePath -> IO Permissions
3 ghci> :i Permissions
4 type Permissions :: *
5 data Permissions
6     = directory-1.3.6.2:System.Directory.Internal.Common.Permissions {readable ::
7     Bool,
8     writable ::
9     Bool,
10    executable
11    :: Bool,
12    searchable
13    :: Bool}
14
15 -- Defined in 'directory-1.3.6.2:System.Directory.Internal.Common'
16 instance [safe] Eq Permissions
17 -- Defined in 'directory-1.3.6.2:System.Directory.Internal.Common'
18 instance [safe] Ord Permissions
19 -- Defined in 'directory-1.3.6.2:System.Directory.Internal.Common'
20 instance [safe] Show Permissions
21 -- Defined in 'directory-1.3.6.2:System.Directory.Internal.Common'
22 instance [safe] Read Permissions
```

```

18      -- Defined in 'directory-1.3.6.2:System.Directory.Internal.Common'
19      ghci> getPermissions "."
20      Permissions {readable = True, writable = True, executable = False, searchable =
      True}
21      ghci> :t searchable
22      searchable :: Permissions -> Bool
23      ghci> searchable it
24

```

- 最后则是 `getModificationTime` 可以告诉我们输入是何时被修改的：

```

1      ghci> :t getModificationTime
2      getModificationTime
3      :: FilePath
4      -> IO time-1.11.1.1:Data.Time.Clock.Internal.UTCTime.UTCTime
5      ghci> getModificationTime "."
6      2023-10-14 13:58:10.895659105 UTC
7

```

那么对于新的谓词而言有多少数据是需要知道的呢？由于可以通过 `Permissions` 来查看一个输入是否为文件还是路径，因此我们不再需要 `doesFileExist` 与 `doesDirectoryExist`。这样我们有了四个需要查看的数据：

```

1  import Control.Exception (bracket, handle)
2  import Control.Monad (filterM)
3  import Data.Time.Clock (UTCTime)
4  import RecursiveContents (getRecursiveContents)
5  import System.Directory (Permissions (..), getModificationTime, getPermissions)
6  import System.FilePath (takeExtensions)
7  import System.IO (IOMode (..), hClose, hFileSize, openFile)
8
9  type Predicate =
10     FilePath ->
11     Permissions ->
12     Maybe Integer ->
13     UTCTime ->
14     Bool

```

与原文不同之处在于导入部分，由于 `import System.Time (ClockTime(..))` 已经废弃了，根据 `getModificationTime` 可知需要使用 `import Data.Time.Clock (UTCTime)`。

我们的 `Predicate` 类型就是一个有着四个参数函数的别名。

注意谓词的返回类型是 `Bool` 而不是 `IO Bool`：谓词是纯的，并不运行 I/O。有了这个类型，编写一个更具表达性的查询函数就好办了。

```

1  getFileSize :: FilePath -> IO (Maybe Integer)
2  getFileSize = undefined
3

```

```

4  betterFind :: Predicate -> FilePath -> IO [FilePath]
5  betterFind p path = getRecursiveContents path >>= filterM check
6      where
7          check name = do
8              perms <- getPermissions name
9              size <- getFileSize name
10             modified <- getModificationTime name
11             return $ p name perms size modified

```

我们暂时先跳过 `getFileSize` 函数，稍后会进行讲解。

我们不能使用 `filter` 来调用谓词 `p`，因为 `p` 的纯性意味着它不能执行收集所需元数据所需的 I/O。

这时就需要一个我们并不熟悉的 `filterM` 函数了，它的行为类似于普通的 `filter` 函数，不过是在 `IO` 单子中对谓词进行计算，即允许谓词执行 I/O。

```

1  ghci> :m +Control.Monad
2  ghci> :t filterM
3  filterM :: Applicative m => (a -> m Bool) -> [a] -> m [a]

```

安全的调整文件大小

虽然 `System.Directory` 没有提供文件大小的查询，不过我们可以用 `System.IO` 来做。它包含了一个名为 `hFileSize` 的函数，可以返回一个文件的字节大小。

```

1  simpleFileSize :: FilePath -> IO Integer
2  simpleFileSize path = do
3      h <- openFile path ReadMode
4      size <- hFileSize h
5      hClose h
6      return size

```

虽然这个函数可以工作，但是它还不能为我们所用。在 `betterFind` 中，我们可以无条件的对任何地址输入调用 `getFileSize`；如果输入的不是一个文件它会返回 `Nothing`，否则返回 `Just`。实际上该函数的输入不是一个文件或者不能打开文件（权限不够）时会抛出异常。

以下是一个安全版本：

```

1  saferFileSize :: FilePath -> IO (Maybe Integer)
2  saferFileSize path = handle errorHandler $ do
3      h <- openFile path ReadMode
4      size <- hFileSize h
5      hClose h
6      return $ Just size
7      where
8          errorHandler :: SomeException -> IO (Maybe Integer)
9          errorHandler _ = return Nothing

```


这里与原文不同在于匿名函数 `handle (_ -> return Nothing)` 改成了 `errorHandler`，并标注了类型 `errorHandler :: SomeException -> IO (Maybe Integer)`。（与第八章的处理方式一致）

函数体内部几乎是一致的，除了 `handle` 子句。

这里的异常处理是只要有异常发生就返回 `Nothing`，其余的改动就是返回值被包裹了 `Just`。

`saferFileSize` 函数现在拥有了正确的类型签名，也不会再抛出任何异常了，不过它仍然不完整。这里只处理了 `openFile` 的异常，但是 `hFileSize` 仍然会抛出异常，比如说命名管道。这样的异常会被 `handle` 捕获，但是我们的 `hClose` 永远不会被执行。

Haskell 的实现会在文件句柄不再使用时自动关闭，但是这要在垃圾回收时才会发生，而接下来垃圾回收的时间却不能保证。

文件句柄是稀有资源，这份稀有性是由操作系统决定的，比如 Linux 系统中一个进程默认同时打开文件的最大数量是 1024 个。

这不难想象一个场景，在调用了使用了 `saferFileSize` 的 `betterFind` 函数因为穷尽了最大文件开启数，而垃圾回收还没开始时，程序崩溃了。

这是一种特别危险的 bug：它会在若干条件组合在一起时变得难以追踪。在 `betterFind` 访问了足够多数量的非文件时句柄没有被关闭而达到了进程最大文件开启数的上限，然后在垃圾回收还未发生之前又打开了另外的文件。

更糟糕的是，任何后续错误都是由程序中无法访问数据而引起，并且还没有垃圾回收。出现这样的 bug 依赖于程序结构，文件系统内容，以及当前程序的运行时还未触发垃圾回收。

这样的问题在开发的过程中很容易被忽视，然后后续发生时会让诊断变得异常困难。

获取-使用-释放

我们需要 `openFile` 成功时 `hClose` 能总是被调用。`Control.Exception` 模块为此提供了 `bracket` 函数。

```
1 ghci> :m +Control.Exception
2 ghci> :type bracket
3 bracket :: IO a -> (a -> IO b) -> (a -> IO c) -> IO c
```

`bracket` 函数接受三个 actions 参数。第一个 action 用于获取一个资源，第二个则是释放资源，第三个则是介于前面两者之间，让我们称其为“使用”action。如果“获取”action 成功了，“释放”action 总是会被调用。这确保了资源总是能被释放。“使用”和“释放”actions 分别传给了“获取”action。

当“使用”action 在执行时发生了异常，`bracket` 会调用“释放”action 然后再重新抛出异常。如果“使用”action 成功了，`bracket` 会调用“释放”action 然后再返回“使用”action 所返回的值。

我们现在可以编写一个完全安全的函数了：它不会抛出异常；也不会产生句柄垃圾而导致程序崩溃。

```

1  getFileSize :: FilePath -> IO (Maybe Integer)
2  getFileSize path = handle errorHandler $
3      bracket (openFile path ReadMode) hClose $ \h -> do
4          size <- hFileSize h
5          return (Just size)
6  where
7      errorHandler :: SomeException -> IO (Maybe Integer)
8      errorHandler _ = return Nothing

```

让我们看一下这里的 `bracket`，首先是打开文件，返回打开文件的句柄，接着是关闭句柄，最后是对句柄调用 `hFileSize` 并将结果包装进 `Just`。

我们需要同时使用 `bracket` 与 `handle` 才能达成目的。前者能保证垃圾文件句柄不会堆积，后者确保异常被处理。

用于谓词的领域特定语言

让我们尝试着写一个谓词，我们的谓词将会检查一个 C++ 源文件是否超过了 128KB。

```

1  myTest path _ (Just size) _ = takeExtension path == ".cpp" && size > 131072
2  myTest _ _ _ _ = False

```

这看起来并不怎么样。谓词有四个参数，总是忽略其中两个，定义另外两个。我们当然可以做的更好，让我们编写一些让谓词变得更简洁的代码。

有时这类库被认为是内嵌领域特定语言：我们使用编程语言的工具（因此是内嵌的）来编写代码用以优雅的解决特类问题（因此是领域特定的）。

第一步首先是编写一个返回某个参数的函数。下面是从传入至 `Predicate` 的参数中提取 `path`：

```

1  pathP path _ _ _ = path

```

如果不提供一个类型签名，Haskell 则会为该函数推导出一个非常通用的类型。这在未来可能会带来错误而难以解析，因此我们给 `pathP` 一个类型。

```

1  type InfoP a =
2      FilePath -> -- path to directory entry
3      Permissions -> -- permissions
4      Maybe Integer -> -- file size (Nothing if not file)
5      UTCTime -> -- last modified
6      a
7
8  pathP :: InfoP FilePath
9  pathP path _ _ _ = path

```

我们创建了一个类型同义词使得可以将其用于其它类似结构函数。该类型同义词接受一个类型参数，这样我们就可以指定不同的结果类型。

```
1 sizeP :: InfoP Integer
2 sizeP _ _ (Just size) _ = size
3 sizeP _ _ Nothing _ = -1
```

实际上，本章较早之前所定义的 `Predicate` 类型与 `InfoP Bool` 是一致的。（因此我们可以去除 `Predicate` 类型。）

那么 `pathP` 与 `sizeP` 怎么用呢？通过一点点粘合，可以将它们放入谓词中使用（`P` 后缀即意味着“predicate”）。

```
1 equalP :: (Eq a) => InfoP a -> a -> InfoP Bool
2 -- equalP f k = \w x y z -> f w x y z == k
3 equalP f k w x y z = f w x y z == k
```

`equalP` 的类型签名值得注意。它接受一个 `InfoP a`，同时兼容 `pathP` 与 `sizeP`。它接受一个 `a`，返回一个 `InfoP Bool`，即 `Predicate` 的同义词。换言之，`equalP` 构建一个谓词。

`equalP` 函数通过返回一个匿名函数来工作。该匿名函数接受一个谓词作为参数，将其传递给 `f`，接着将其返回与 `k` 作比较。

由于 Haskell 的所有函数都是柯里化的，这么编写 `equalP` 实际上并不必要。我们可以省略匿名函数接着依赖柯里化来完成剩下的部分，现在让我们来编写一个相同效果的函数。

```
1 equalP' :: (Eq a) => InfoP a -> a -> InfoP Bool
2 equalP' f k w x y z = f w x y z == k
```

在继续探索之前，让我们将我们的模块加载至 `ghci`。

```
1 ghci> :l BetterPredicate.hs
2 [1 of 2] Compiling RecursiveContents ( RecursiveContents.hs, interpreted )
3 [2 of 2] Compiling Main ( BetterPredicate.hs, interpreted )
4 Ok, two modules loaded.
```

让我们看看由这些函数所构建的谓词是否生效。

```
1 ghci> :t betterFind (sizeP `equalP` 1024)
2 betterFind (sizeP `equalP` 1024) :: FilePath -> IO [FilePath]
```

注意我们并没有真正的调用 `betterFind`，不过却能确认表达式的类型。我们现在拥有了一个更具表现力的方式来展示所有相同大小的文件。

通过 lifting 避免重复

除开 `equalP`，我们还可以编写其它的二元函数。我们倾向于不为每个定义都编写一个完整的定义，因为这看起来冗余而没有必要。

为此让我们使用 Haskell 的抽象能力。我们将 `equalP` 的定义而不是直接调用 `(==)`，将其作为参数传递给希望调用的二元函数中。

```
1 liftP :: (a -> b -> c) -> InfoP a -> b -> InfoP c
2 liftP q f k w x y z = f w x y z `q` k
3
4 greaterP, lesserP :: (Ord a) => InfoP a -> a -> InfoP Bool
5 greaterP = liftP (>)
6 lesserP = liftP (<)
```

这里接受如 `(>)` 这样的函数，转换它为另一个作用于不同上下文的函数，如 `greaterP`，这意味着将其提升 *lifting* 至该上下文。这解释了函数名中带有 `lift`。Lifting 让我们复用代码并且减少重复的模式。在本书的后续章节中，我们将大量的使用它。

我们在提升一个函数时，通常会分别将其原始版本以及新版本称为 *unlifted* 与 *lifted*。

通过这种方式，我们的 `q`（需要提升的函数）作为 `liftP` 的第一个参数是经过深思熟虑的。这使得可以编写如 `greaterP` 以及 `lesserP` 这样简洁的定义。相较于其它的语言，偏应用使得寻找“最佳”的参数顺序成为了 Haskell 中 API 设计中更为重要的环节。在没有偏应用的语言中，参数的顺序在于使用习惯与方便性。而在 Haskell 中将参数设计在错误的位置，则会丢失偏应用所带来的简洁性。

我们可以通过组合子来复现一下这样的简洁性。例如，`forM` 在 2007 年之前并没有被加入至 `Control.Monad` 模块中，更早之前人们则会使用 `flip mapM` 来替代。

```
1 ghci> :m +Control.Monad
2 ghci> :t mapM
3 mapM :: (Traversable t, Monad m) => (a -> m b) -> t a -> m (t b)
4 ghci> :t forM
5 forM :: (Traversable t, Monad m) => t a -> (a -> m b) -> m (t b)
6 ghci> :t flip mapM
7 flip mapM
8 :: (Traversable t, Monad m) => t a -> (a -> m b) -> m (t b)
```

与原文不同之处在于上述函数变得更加泛化了，从之前的只对列表生效变为了对所有满足 `Traversable` typeclass 的类型生效。

将谓词们粘合在一起

如果我们希望组合谓词们，那么就可以遵循之前所做的那样：

```
1 simpleAndP :: InfoP Bool -> InfoP Bool -> InfoP Bool
2 simpleAndP f g w x y z = f w x y z && g w x y z
```

现在我们了解了 *lifting*，那么减少之前必须为布尔值操作符所编写的代码量就变得很自然了。

```
1 liftP2 :: (a -> b -> c) -> InfoP a -> InfoP b -> InfoP c
```

```

2 liftP2 q f g w x y z = f w x y z `q` g w x y z
3
4 andP2 = liftP (&&)
5 orP2 = liftP (||)

```

注意 `liftP2` 跟之前的 `liftP` 很像。实际上前者更通用，因为我们可以根据 `liftP2` 来编写 `liftP`：

```

1 constP :: a -> InfoP a
2 constP k _ _ _ = k
3
4 liftP' q f k w x y z = f w x y z `q` constP k w x y z

```

Tip

组合子 Combinators

Haskell 中，我们将接受函数作为参数并返回新函数的函数成为组合子。

现在我们有了一些帮助函数，可以回到较早之前定义的 `myTest` 函数了。

```

1 myTest :: (Ord a, Num a) => FilePath -> p1 -> Maybe a -> p2 -> Bool
2 myTest path _ (Just size) _ = takeExtension path == ".cpp" && size > 131072
3 myTest _ _ _ _ = False

```

那么该函数使用新的组合子会长什么样呢？

```

1 liftPath :: (FilePath -> a) -> InfoP a
2 liftPath f w _ _ = f w
3
4 myTest2 = (liftPath takeExtension `equalP` ".cpp") `andP` (sizeP `greaterP` 131072)

```

我们添加了一个最终的组合子，`liftPath`，因为操作文件名是一个常用的操作。

定义与使用新的操作符

我们还可以通过定义新的中缀操作符来让我们的领域特定语言更进一步。

```

1 -- explicit annotation
2 (==?) :: InfoP String -> String -> InfoP Bool
3 (==?) = equalP
4
5 (&&?) :: InfoP Bool -> InfoP Bool -> InfoP Bool
6 (&&?) = andP
7
8 -- explicit annotation
9 (>?) :: InfoP Integer -> Integer -> InfoP Bool
10 (>?) = greaterP
11
12 myTest3 = (liftPath takeExtension ==? ".cpp") &&? (sizeP >? 131072)

```

注意与原文不同之处在于 `(==?)` 以及 `(>?)` 加上了显式类型签名, 因为在没有 `myTest3` 时, 编译器无法正确的进行类型推导而导致编译错误。

我们选择如 `(==?)` 这样的名称作为提升函数, 这样可以在视觉上与原始函数对应。

上面定义中的圆括号是必要的, 因为我们没有告诉 Haskell 新操作符的优先级或结合性。语言规定没有固定声明的操作符应被视为 `infixl 9`, 即在最高优先级从左到右求值。

我们可以通过为新操作符编写固定声明, 第一步是找出未提升的操作符使得可以模拟它们。

```
1 ghci> :i ==
2 type Eq :: * -> Constraint
3 class Eq a where
4   (==) :: a -> a -> Bool
5   ...
6   -- Defined in 'GHC.Classes'
7 infix 4 ==
8 ghci> :i &&
9 (&&) :: Bool -> Bool -> Bool -- Defined in 'GHC.Classes'
10 infixr 3 &&
11 ghci> :i >
12 type Ord :: * -> Constraint
13 class Eq a => Ord a where
14   ...
15   (>) :: a -> a -> Bool
16   ...
17   -- Defined in 'GHC.Classes'
18 infix 4 >
```

有了它们以后, 我们可以编写一个不带圆括号的表达式, 解析后与 `myTest3` 一致:

```
1 infix 4 ==?
2 infixr 3 &&?
3 infix 4 >?
4
5 myTest4 = liftPath takeExtension ==? ".cpp" &&? sizeP >? 131072
```

控制遍历

在遍历文件系统时, 我们希望得到更多在方向上的控制。一个简单的方法便是允许传递给定路径的子路径列表至函数, 并返回另一个列表。该列表允许元素被移除, 或者可以以其他方式排序, 或是两者皆可。最简单的控制函数是 `id`, 返回的值与输入值保持一致。

为了多样性, 我们将改变一些展示的方式。相较于使用一个解释性函数类型 `InfoP a`, 我们将使用一个普通的代数数据类型来展示同样的信息。

```
1 import Data.Time (UTCTime)
2 import System.Directory (Permissions)
3
4 data Info = Info
```

```

5   { infoPath :: FilePath,
6     infoPerms :: Maybe Permissions,
7     infoSize  :: Maybe Integer,
8     infoModTime :: Maybe UTCTime
9   }
10  deriving (Eq, Ord, Show)
11
12  getInfo :: FilePath -> IO info

```

我们使用 record 语义来“自由的”访问函数，例如 `infoPath`。`traverse` 函数的类型很简单，正如之前推测的那样。为了获取文件或者路径的 `Info`，我们可以调用 `getInfo` action。

```

1  traverse' :: ([Info] -> [Info]) -> FilePath -> IO [Info]

```

与原文不同之处在于 `traverse` 已经在 `Prelude` 中定义了，因此需要另一个名称。

`traverse` 的定义如下：

```

1  traverse' :: ([Info] -> [Info]) -> FilePath -> IO [Info]
2  traverse' order path = do
3    names <- getUsefulContents path
4    contents <- mapM getInfo (path : map (path </>) names)
5    liftM concat $ forM (order contents) $ \info -> do
6      if isDirectory info && infoPath info /= path
7        then traverse' order (infoPath info)
8        else return [info]
9
10  getUsefulContents :: FilePath -> IO [String]
11  getUsefulContents path = do
12    names <- getDirectoryContents path
13    return (filter (`notElem` [".", ".."]) names)
14
15  isDirectory :: Info -> Bool
16  isDirectory = maybe False searchable . infoPerms

```

首先是变量 `contents` 的赋值，从右至左来阅读代码。我们已经知道了 `names` 是一个路径列表，需要确保当前路径被前置放置在列表中的每个元素中，同时还要包含当前路径其本身。接着使用 `mapM` 应用 `getInfo` 函数在上述的返回值（路径列表）上。

再次从右往左阅读代码，我们可以看到该行的最后一个元素是一个匿名函数的定义。给定该匿名函数一个 `Info` 值，该函数要么递归的访问一个路径（有额外的检查可以确保我们不会再次访问 `path`），要么返回一个单利列表（用于匹配 `traverse'` 的返回类型）。

用户提供的遍历控制函数 `order` 处理 `contents` 的结果为一个 `Info` 值列表，然后再被 `forM` 将上述匿名函数应用至这个列表的每个元素上。

最后是 `liftM` 函数，接受一个普通函数 `concat`，提升该函数至 `IO` 单子。换言之，`liftM` 将 `forM` 的结果（类型为 `IO [[Info]]`）从 `IO` 单子中提取出来，将 `concat` 应用在提取的结果上（返回值类型为 `[Info]`，即我们所需要的），然后再将返回结果放入 `IO` 单子中。

最后，不要忘记定义我们的 `getInfo` 函数：

```
1 getInfo :: FilePath -> IO Info
2 getInfo path = do
3     perms <- maybeIO (getPermissions path)
4     size <- maybeIO (withFile path ReadMode hFileSize)
5     modified <- maybeIO (getModificationTime path)
6     return (Info path perms size modified)
7
8 maybeIO :: IO a -> IO (Maybe a)
9 maybeIO act = handle (\(SomeException _) -> return Nothing) $ Just `liftM` act
```

`size <- maybeIO (bracket (openFile path ReadMode) hClose hFileSize)` 可以被 `withFile` 简化成 `size <- maybeIO (withFile path ReadMode hFileSize)`，其次就是 `maybeIO` 的匿名函数的入参显式声明类型 `(\ (SomeException _) -> return Nothing)`。

这里唯一需要注意的是一个有用的组合子 `maybeIO`，其将一个可能会抛出异常的 `IO` action，以 `Maybe` 包装结果。

密集，可读性与学习过程

Haskell 中像 `traverse'` 这样密集的代码是不常见的。代码的表现力是很重要的，它需要相关的少量代码可以被流畅的阅读以及编写。

对比一下下面相同作用的密集代码，这更像是一个经验较少的 Haskell 程序员所写的：

```
1 traverseVerbose order path = do
2     names <- getDirectoryContents path
3     let usefulNames = filter (`notElem` [".", ".."]) names
4     contents <- mapM getEntryName (" " : usefulNames)
5     recursiveContents <- mapM recurse (order contents)
6     return (concat recursiveContents)
7     where
8         getEntryName name = getInfo (path </> name)
9         isDirectory info = maybe False searchable (infoPerms info)
10        recurse info = do
11            if isDirectory info && infoPath info /= path
12            then traverseVerbose order (infoPath info)
13            else return [info]
```

与原文不同之处在于 `isDirectory` 的 `case ... of` 被 `maybe` 替换了。

编写可维护的 Haskell 代码的关键是在密集与可读性中寻找平衡。

另一个角度看待遍历

尽管 `traverse'` 函数相较于 `betterFind` 函数，给与了我们更多的控制，但它还有一个重要的不足：我们可以在路径上进行递归，却不能过滤其它的名称，直到我们构建了整个列表的

名称树。如果遍历的路径包含了 100,000 个文件，而我们只关心其中三个，我们将分配 100,000 个元素的列表，然后才有机会修剪改列表保留三个所需的元素。

一种解决方案就是提供一个筛选函数作为 `traverse'` 的参数，这样可以将其应用在我们生成的名称列表上。这使得分配的列表只包含所需的元素。

然而这种方案同样有一个弱点：假设我们知道只需三个元素，而这三个元素在 100,000 个元素的前排，这种情况下我们无需再访问剩余的 99,997 个元素。这绝不是一个人为的例子：例如，邮箱的文件夹里的邮件，一个路径代表了邮箱所包含的千分之十的文件是非常常见的。

我们可以从另一个角度来解决上述两种方案的弱点：如果我们将文件系统的遍历视为一个路径层次的折叠 *fold* 呢？

我们熟知的 fold 有 `foldr` 以及 `foldl'`，非常简洁的阐述了遍历一个列表并累积一个结果。延展折叠路径列表的概念并不难，不过我们所希望的是添加一个控制到折叠上。我们将这个控制表述为一个代数数据类型。

```

1  import ControlledVisit (Info)
2
3  data Iterate seed
4    = Done {unwrap :: seed}
5    | Skip {unwrap :: seed}
6    | Continue {unwrap :: seed}
7    deriving (Show)
8
9  type Iterator seed = seed -> Info -> Iterate seed

```

`Iterator` 类型给与我们所想要折叠的函数了一个便捷的别名。它接受一个 `seed` 以及一个用于表示路径的 `Info` 值，返回一个新的 `seed` 以及一个折叠函数的指示，该指示代表着 `Iterate` 类型的构造函数。

- 如果指示为 `Done`，那么遍历应该立即停止。由 `Done` 所包装的值应该作为结果被返回。
- 如果指示为 `Skip`，同时当前 `Info` 代表着一个路径，遍历将不会在该路径上递归。
- 其余情况下的遍历为 `Continue`，使用包装的值作为 fold 函数的下一个调用。

我们的折叠逻辑上是一个左折叠，因为折叠第一个输入，同时每一步的 `seed` 是上一步所返回的结果。

```

1  foldTree :: Iterator a -> a -> FilePath -> IO a
2  foldTree iter initSeed path = do
3    endSeed <- fold initSeed path
4    return $ unwrap endSeed
5  where
6    fold seed subpath = getUsefulContents subpath >>= walk seed
7    walk seed (name : names) = do
8      let path' = path </> name

```

```

9      info <- getInfo path'
10     case iter seed info of
11       done@(Done _) -> return done
12       Skip seed' -> walk seed' names
13       Continue seed'
14         | isDirectory info -> do
15           next <- fold seed' path'
16           case next of
17             done@(Done _) -> return done
18             seed'' -> walk (unwrap seed'') names
19         | otherwise -> walk seed' names
20     walk seed _ = return (Continue seed)

```

上述代码中有几个有趣的点。首先作用域的使用避免了额外的参数传递。顶层的 `foldTree` 函数仅是 `fold` 的一个包装，从 `fold` 的构造函数中提取最终结果。

由于 `fold` 是一个本地函数，我们无需传递 `foldTree` 的 `iter` 变量给它；它本身就可以访问外层的作用域。同样的，`walk` 也可以看到外层作用域的 `path`。

另一个值得关注的点就是 `walk` 是一个尾递归循环，而非早前定义的函数中被 `forM` 所调用的匿名函数。这使得可以在需要的时候停止循环，也就是当迭代器返回 `Done` 时退出。

尽管 `fold` 调用 `walk`，`walk` 调用 `fold` 递归的遍历子路径。每个函数返回一个被包装在 `Iterate` 的 `seed`：当 `fold` 被 `walk` 调用再返回，`walk` 检测其结果并决定是否继续还是退出（`Done`）。这种方式下，调用者提供的迭代器返回 `Done` 时会立刻终止两个函数之间的递归调用。

那么一个迭代器在实际使用上是什么样的呢？下面是一个复杂的例子，用于查看至多三个 bitmap 图像，且不会递归到 Subversion 元数据目录中。

```

1  atMostThreePictures :: Iterator [FilePath]
2  atMostThreePictures paths info
3    | length paths == 3 =
4      Done paths
5    | isDirectory info && takeFileName path == ".svn" =
6      Skip paths
7    | otherwise =
8      Continue paths
9  where
10     extension = map toLower $ takeExtension path
11     path = infoPath info

```

调用 `foldTree atMostThreePictures []`，返回值类型为 `IO [FilePath]`。

当然了，迭代器也不需要很复杂。下面是一个统计路径数的例子：

```

1  countDirectories count info =
2  Continue (if isDirectory info then count + 1 else count)

```

这两门传递给 `foldTree` 的初始 `seed` 需要是数字零。

有用的编码指导

略

10 Code case study: parsing a binary data format

本章我们将要探讨一个通用任务：解析一个二进制文件。使用该任务有两个目的，第一是借助解析来探讨程序的规划，重构，以及“模板化代码移除”。我们将展示如何简化重复的代码，并为我们在第 14 章 Monads 中做准备。

我们将使用的文件格式源自 netpbm 套件，这是一个古老而可敬的用于处理位图图像的程序和文件格式集合。这些文件格式具有广泛的时候以及容易解析的双重优点。更重要的是为了方便起见，netpbm 文件没有被压缩。

灰度文件

netpbm 的灰度文件格式为 PGM (“portable grey map”)。它有两种格式构；“plain”(或“P2”)格式是由 ASCII 编码的，而更常用的“raw”(“P5”)则是二进制格式。

两种格式的文件都有 header，即一个用于描述格式的“魔力”字符串。对于一个 plain 文件而言，字符串为 **P2**；raw 文件则是 **P5**。紧随字符串后的则是一个空格然后再是三个数字：图片的宽度，长度，以及最大灰度。这些数字皆为 ASCII 小数，由空格分隔。

接下来的就是图片数据。raw 文件中是二进制的字符串，plain 文件中则是由单空格字符分隔的 ASCII 小数构成。

raw 文件可以包含一系列的图片，一张接着一张，每张由 header 开头；plain 文件仅包含一张图片。

解析一个原始 PGM 文件

我们的第一个解析函数仅需考虑 raw PGM 文件，同时该解析函数是一个纯函数。该函数并不对数据获取负责，仅仅用作于解析。这在 Haskell 中是一个常规操作。通过分离数据读取，我们获得了更加灵活的操作空间。

我们将使用 **ByteString** 类型来存储 greymap 数据。由于 PGM 文件的头部是 ASCII 文本，而本体是二进制的，我们需要同时引用文本以及二进制的 **ByteString** 模块。

```
1 import qualified Data.ByteString.Lazy as L
2 import qualified Data.ByteString.Lazy.Char8 as L8
3 import Data.Char (isSpace)
```

我们将使用一个直接的数据类型来表示 PGM 图片。

```
1 data Greymap = Greymap
2 { greyWidth :: Int,
3   greyHeight :: Int,
4   greyMax :: Int,
5   greyData :: L.ByteString
6 }
```

```
7 deriving (Eq)
```

通常而言，Haskell 的 `Show` 实例应该生成一个字符串作为展示，同时我们可以通过 `read` 将其从字符串中转回。然而对于 `bitmap` 图像文件而言，这会潜在的生产出巨大的文本字符串，例如对一个图片执行 `show`。因此我们不会让编译器自动派生一个 `Show` 实例：我们编写自己的实现，并将其简化。

```
1 instance Show Greymap where
2   show (Greymap w h m _) = "Greymap " ++ show w ++ "x" ++ show h ++ " " ++ show m
```

因为我们的 `Show` 实例是特意避开了打印 `bitmap` 数据，因此我们无法编写一个 `Read` 实例，毕竟我们不能通过 `show` 的结果来重新构建一个合法的 `Greymap`。

接下来是我们解析函数的类型：

```
1 parseP5 :: L.ByteString -> Maybe (Greymap, L.ByteString)
```

该函数接受一个 `ByteString`，如果解析成功则返回一个单独解析好的 `Greymap`，以及剩下还需要解析的字符串。

我们的解析函数需要需要花费一点时间。首先是确保输入的文件是 raw PGM；接着解析文件头剩下的数值；然后再是解析 `bitmap` 数据。下面是一个浅显的表达方式：

```
1 matchHeader :: L.ByteString -> L.ByteString -> Maybe L.ByteString
2 matchHeader = undefined
3
4 getNat :: L.ByteString -> Maybe (Int, L.ByteString)
5 getNat = undefined
6
7 getBytes :: Int -> L.ByteString -> Maybe (L.ByteString, L.ByteString)
8 getBytes = undefined
9
10 -- parse function
11 parseP5 :: L.ByteString -> Maybe (Greymap, L.ByteString)
12 parseP5 s =
13   case matchHeader (L8.pack "PS") s of
14     Nothing -> Nothing
15     Just s1 ->
16       case getNat s1 of
17         Nothing -> Nothing
18         Just (width, s2) ->
19           case getNat (L8.dropWhile isSpace s2) of
20             Nothing -> Nothing
21             Just (height, s3) ->
22               case getNat (L8.dropWhile isSpace s3) of
23                 Nothing -> Nothing
24                 Just (maxGrey, s4)
25                   | maxGrey > 255 -> Nothing
26                   | otherwise ->
27                     case getBytes 1 s4 of
```

```

28         Nothing -> Nothing
29         Just (_, s5) ->
30             case getBytes (width * height) s5 of
31                 Nothing -> Nothing
32                 Just (bitmap, s6) ->
33                     Just (Greymap width height maxGrey bitmap, s6)

```

这非常字面化的代码将所有的解析放在了一个长的阶梯型的 `case` 表达式上。每个函数在消费完其所需的字符串后，都会返回剩下的 `ByteString`。接着解构每个结果，如果解析失败则返回 `Nothing`。下面是解析过程中所用到的函数：

```

1  matchHeader :: L.ByteString -> L.ByteString -> Maybe L.ByteString
2  matchHeader prefix str
3      | prefix `L8.isPrefixOf` str = Just (L8.dropWhile isSpace (L.drop (L.length prefix) str))
4      | otherwise = Nothing
5
6  getNat :: L.ByteString -> Maybe (Int, L.ByteString)
7  getNat s = case L8.readInt s of
8      Nothing -> Nothing
9      Just (num, rest)
10         | num <= 0 -> Nothing
11         | otherwise -> Just (fromIntegral num, rest)
12
13  getBytes :: Int -> L.ByteString -> Maybe (L.ByteString, L.ByteString)
14  getBytes n str =
15      let count = fromIntegral n
16          both@(prefix, _) = L.splitAt count str
17      in if L.length prefix < count
18         then Nothing
19         else Just both

```

移除样板代码

虽然 `parseP5` 函数能工作，其形态并不令人满意。我们不停地重复 `Maybe` 值的构建与解构，且仅在匹配 `Just` 时继续。所有的这些相似的 `case` 表达式就像是“模板代码”那样。

我们可以看到有两个模式。首先是很多应用的函数都有相同的类型，接受 `ByteString` 作为其最后一个参数，并返回 `Maybe`。其次 `parseP5` 函数中的每一步“阶梯”都会解构一个 `Maybe` 值，要么失败或是传递未解包的结果给一个函数。

我们可以轻易地用一个函数捕获第二个模式。

```

1  (>>?) :: Maybe a -> (a -> Maybe b) -> Maybe b
2  Nothing >>? _ = Nothing
3  Just v >>? f = f v

```

`>>?` 函数的作用非常简单：它接受一个值作为其右参，以及一个函数作为其左参。如果值非 `Nothing`，那么便将函数应用在 `Just` 值上。我们定义了作为一个操作符的函数，以此可

以链接所有函数在一起。最后我们并未提供给 `(>>?)` 一个优先级声明, 那么默认为 `infixl 9` (左结合, 最强的操作符优先级)。换言之, `a >>? b >>? c` 将会从左至右进行计算, 类似于 `(a >>? b) >>? c`。

有了这个链接函数, 我们可以再尝试一下编写我们的解析函数。

```
1 parseP5_take2 :: L.ByteString -> Maybe (Greymap, L.ByteString)
2 parseP5_take2 s =
3   matchHeader (L8.pack "P5") s
4     >>? \s ->
5       skipSpace ((), s)
6         >>? (getNat . snd)
7           >>? skipSpace
8             >>? \(width, s) ->
9               getNat s
10                >>? skipSpace
11                  >>? \(height, s) ->
12                    getNat s
13                      >>? \(maxGrey, s) ->
14                        getBytes 1 s
15                          >>? (getBytes (width * height) . snd)
16                            >>? \(bitmap, s) -> Just (Greymap width height maxGrey bitmap, s)
17
18 skipSpace :: (a, L.ByteString) -> Maybe (a, L.ByteString)
19 skipSpace (a, s) = Just (a, L8.dropWhile isSpace s)
```

理解这个函数的关键在于链条的思考。每个 `(>>?)` 的左侧总是一个 `Maybe` 值; 右侧则是一个返回 `Maybe` 值的函数。那么表达式的两侧类型都是 `Maybe`, 即满足下一个 `(>>?)` 表达式。

另一个增加可读性的工作就是添加了 `skipSpace` 函数。

隐式状态

我们的代码仍然在显式的传递一对一对的值, 第一个元素作为解析过程中的结果, 第二个元素则是当前剩余的 `ByteString`。当我们想要扩展代码时, 例如追踪消费了多少字节数量以便报告解析异常位置的时候, 我们需要修改八个不同的地方, 仅仅是为了传递一个三元组。

即使在少量代码的情况下, 这样的需求都会使得代码难以修改。这个问题存在于使用模式匹配从每个元组中提取值时: 我们默认了在代码中直接使用元组。

让我们讲解一下新代码为何是不灵活的。首先修改解析器使用的状态类型。

```
1 data ParseState = ParseState
2 { string :: L.ByteString,
3   offset :: Int64
4 }
5 deriving (Show)
```

在我们新的代数数据类型中，我们同时拥有了追踪剩余字符串以及当前偏移量的能力。最重要的变化还是使用了 `record` 语义：现在可以避免对状态的模式匹配了，而是使用访问函数 `string` 以及 `offset`。

我们给与需要解析的状态一个名称。当我们为某物命名时，它可以变得更有意义。例如，我们可以将解析视为一种函数：其消费一个解析状态，并同时生产一个新的解析状态，以及一些额外的信息。我们可以直接将其视为一个 Haskell 类型。

```
1 simpleParse :: ParseState -> (a, ParseState)
2 simpleParse = undefined
```

为了更好的帮助用户，我们可以在解析失败时报告异常信息。这仅仅需要我们解析器一点小小的改动。

```
1 betterParse :: ParseState -> Either String (a, ParseState)
2 betterParse = undefined
```

之前显式的使用状态元组时，要扩展解析器的时候我们很快就发现问题了。为了避免重复，我们将使用 `newtype` 声明来隐去解析类型的细节。

```
1 newtype Parse a = Parse
2 { runParse :: ParseState -> Either String (a, ParseState)
3 }
```

记住 `newtype` 定义就是一个编译时的包装函数，因此它没有任何的运行时负荷。当我们使用该函数时，我们将应用 `runParser` 访问函数。

如果我们不从模块中导出 `Parse` 值，我们可以确保他人不会意外的创建一个解析器，也避免了通过模式匹配来查看内部实现。

唯一性解析器

让我们尝试定义一个简单的解析器，`identity` 解析器。它的功能便是将任何传递进其的参数转换为解析器的结果。这种情况下，它更像是 `id` 函数。

```
1 identity :: a -> Parse a
2 identity a = Parse (\s -> Right (a, s))
```

该函数不会改变解析状态，并使用其参数作为解析的结果。我们将函数体包装成 `Parse` 类型来满足类型检查器。那么我们该如何使用该包装后的函数来进行解析呢？

首先我们通过 `runParse` 函数去除 `Parse` 的包装，以此获取其中的函数。接着构造一个 `ParseState`，并以此运行我们的解析函数。最后则是将解析的结果从最终的 `ParseState` 中分离。

```
1 parse :: Parse a -> L.ByteString -> Either String a
2 parse parser initState =
3   case runParse parser (ParseState initState 0) of
```



```

4     Left err -> Left err
5     Right (result, _) -> Right result

```

由于 `identity` 解析器与 `parse` 函数都不会测试状态，我们甚至不需要创建一个字符串输入来进行测试。

```

1 ghci> :l Parse.hs
2 [1 of 1] Compiling Main             ( Parse.hs, interpreted )
3 Ok, one module loaded.
4 ghci> :t parse (identity 1) undefined
5 parse (identity 1) undefined :: Num a => Either String a
6 ghci> parse (identity 1) undefined
7 Right 1
8 ghci> parse (identity "foo") undefined
9 Right "foo"

```

解析器甚至不会检查其不感兴趣的输入，之后我们将见识到其有用之处。

Record 语义，更新，以及模式匹配

Record 语义不仅仅只有访问函数有用：我们用它来拷贝以及修改现有值的一部分。使用中，概念如下：

```

1 modifyOffset :: ParseState -> Int64 -> ParseState
2 modifyOffset initState newOffset = initState {offset = newOffset}

```

这会创建一个新的与 `initState` 一样的 `ParseState` 值，不过其 `offset` 字段被设置成了指定的 `newOffset`。

```

1 ghci> :l Parse.hs
2 [1 of 1] Compiling Main             ( Parse.hs, interpreted )
3 Ok, one module loaded.
4 ghci> let before = ParseState (L8.pack "foo") 0
5 ghci> let after = modifyOffset before 3
6 ghci> before
7 ParseState {string = "foo", offset = 0}
8 ghci> after
9 ParseState {string = "foo", offset = 3}

```

我们可以在花括号内部设置任意数量的字段，通过逗号进行分隔。

一个更有趣的解析器

现在让我们聚焦到编写一个更有意义的解析器上。我们暂时没有太大的野心：我们需要的是解析一个单字节。

```

1 parseByte :: Parse Word8
2 parseByte =
3     getState ==> \initState ->

```

```

4     case L.uncons (string initState) of
5       Nothing ->
6         bail "no more input"
7       Just (byte, remainder) ->
8         putState newState ==> \_ ->
9           identity byte
10      where
11        newState = initState {string = remainder, offset = newOffset}
12        newOffset = offset initState + 1

```

定义中出现了若干新函数。

`L8.uncons` 函数从一个 `ByteString` 中接受首个元素。

```

1 ghci> L8.uncons (L8.pack "foo")
2 Just ('f',"oo")
3 ghci> L8.uncons L8.empty
4 Nothing

```

`getState` 函数获取当前的解析状态，而 `putState` 则是用于替换状态；`bail` 函数则是终结解析并报告异常；`(==>)` 函数将解析器链接起来。我们将稍后对每个函数进行解析。

Tip

Hanging lambdas

`parseByte` 的定义拥有一个视觉上的风格是我们之前没有讨论过的。它包含的匿名函数为参数以及 `->` 作为一行的结尾，而函数体起始于下一行。

这种方式的匿名函数并没有一个官方的名称，所以让我们称其“hanging lambda”。它主要的作用是为函数体留下更多的空间，同样可以将函数及其后续部分的关系在视觉上进行区分。例如通常而言，首个函数的结果会被当做参数传递给下一个函数。

获取与修改解析状态

`parseByte` 函数并不将解析状态作为一个参数，而是调用 `getState` 来获取一个状态的拷贝，`putState` 则是将当前状态替换。

```

1 getState :: Parse ParseState
2 getState = Parse (\s -> Right (s, s))
3
4 putState :: ParseState -> Parse ()
5 putState s = Parse (\_ -> Right ((), s))

```

当读取这些函数，记住元组的左元素是 `Parse` 的结果，而右元素则是当前解析状态。这使得接下来的函数变得更加方便使用。

`getState` 函数提取当前解析中的状态，使得调用者可以访问字符串。`putState` 函数替换当前解析中的状态。该状态通过 `==>` 链在下一个函数中可见。

这些函数让我们显式的移动状态在仅需要其所需的函数中。很多函数并不需要知道当前的状态，因此它们永远也不会调用 `getState` 或者 `putState`。这相较于之前手动使用元组的解析器，这让我们可以编写出更简洁的代码，

我们将解析状态的细节打包进 `ParseState` 类型中，通过访问函数而不是模式匹配来进行工作。现在的解析状态是被隐式的传递，这为我们带来了极大的便利。如果我们希望添加更多的信息在解析状态中，我们仅需要修改 `ParseState` 的定义，以及任何需要这些新的信息的函数。相较于早期编写的解析器，所有的状态都被模式匹配所暴露，现在的代码则更加模块化：所有被影响的代码为需要新信息的代码。

报告解析异常

我们小心的定义 `Parse` 类型来兼容所有可能的异常。`==>` 组合子检查一个解析异常，并在出现异常时停止。然而我们仍未介绍 `bail` 函数，其用于报告解析异常。

```
1 bail :: String -> Parse a
2 bail err = Parse $ \s ->
3   Left $
4     "byte offset " ++ show (offset s) ++ ": " ++ err
```

在我们调用 `bail` 之后，`(==>)` 将会成功的在 `Left` 构造子上进行模式匹配包装错误信息，接着唤起链条中下一个解析函数。这将导致错误信息通过链条反向传递至上一级调用者。

链起所有解析器

`(==>)` 函数类似之前的 `(>>?)` 函数：它用于“粘合”使得所有函数链接在一起。

```
1 (==>) :: Parse a -> (a -> Parse b) -> Parse b
2 firstParser ==> secondParser = Parse chainedParser
3 where
4   chainedParser initState =
5     case runParse firstParser initState of
6       Left errorMessage ->
7         Left errorMessage
8       Right (firstResult, newState) ->
9         runParse (secondParser firstResult) newState
```

`(==>)` 的函数体很有意思。回忆一下 `Parse` 类型是一个被包装装的函数。因为 `(==>)` 将两个 `Parse` 值链接生产出第三个值，它必须返回一个包装后的函数。

该函数不需要真正的“做”什么：它仅仅是创建一个闭包 *closure* 来记住 `firstParser` 与 `secondParser` 的值。

Tip

闭包其实就是一个带有环境的函数，其边界变量为其可见区域。闭包在 Haskell 中很常见。例如 `(+5)` 就是一个闭包。该闭包的实现必须将值 `5` 作为操作符 `(+)` 的第二个参数，因此无论函数传递的值是什么，其返回将会加五。

闭包不会被解包并执行，直到应用了 `parse` 函数。此时它将被应用至一个 `ParseState` 上，首先是应用 `firstParser` 并检验其结果。如果解析失败，闭包同样也会失败；否则它将传递解析的结果，以及新的 `ParseState` 至 `secondParser`。

这真的是一个美妙的东西：我们高效的将 `ParseState` 隐式的传递至 `Parse` 链。（我们在稍后的章节中继续学习。）

函子简介

我们现在已经很熟悉 `map` 函数了，它将一个函数应用在一个列表中的每个元素上，返回很可能是另一种类型的列表。

```
1 ghci> map (+1) [1,2,3]
2 [2,3,4]
3 ghci> map show [1,2,3]
4 ["1","2","3"]
5 ghci> :t map show
6 map show :: Show a => [a] -> [String]
```

`map` -类似的行为在其它实例中也很有用。例如，假设有一个二元树。

```
1 data Tree a
2 = Node (Tree a) (Tree a)
3 | Leaf a
4 deriving (Show)
```

如果我们想要将一个字符串树转换为一个包含了这些字符串长度的树，我们可以这样做：

```
1 treeLengths (Leaf s) = Leaf (length s)
2 treeLengths (Node l r) = Node (treeLengths l) (treeLengths r)
```

现在可以将目光转向更为通用的函数：

```
1 treeMap :: (a -> b) -> Tree a -> Tree b
2 treeMap f (Leaf a) = Leaf (f a)
3 treeMap f (Node l r) = Node (treeMap f l) (treeMap f r)
```

正如我们希望的那样，`treeLengths` 与 `treeMap length` 可以得到同样的结果。

```
1 ghci> :l TreeMap.hs
2 [1 of 1] Compiling Main             ( TreeMap.hs, interpreted )
3 Ok, one module loaded.
```

```

4 ghci> let tree = Node (Leaf "foo") (Node (Leaf "x") (Leaf "quux"))
5 ghci> treeLengths tree
6 Node (Leaf 3) (Node (Leaf 1) (Leaf 4))
7 ghci> treeMap length tree
8 Node (Leaf 3) (Node (Leaf 1) (Leaf 4))
9 ghci> treeMap (odd . length) tree
10 Node (Leaf True) (Node (Leaf True) (Leaf False))

```

Haskell 提供了一个著名的 typeclass 来泛化 `treeMap`。该 typeclass 就是 `Functor`，其定义为一个函数，`fmap`。

我们可以认为 `fmap` 类似于之前小节中所提到的提升 *lifting* 函数。它接受类型为 `a -> b` 的函数，将其提升至应用在容器上的 `f a -> f b` 函数，这里的 `f` 即容器类型。

如果我们将 `f` 替换为 `Tree` 类型，那么 `fmap` 的类型应与 `treeMap` 一致，那么实际上我们可以用 `treeMap` 作为 `Tree` 的 `fmap` 实现。

```

1 instance Functor Tree where
2   fmap = treeMap

```

`Functor` 的定义强制了一些明显的限制在 `fmap` 上。例如我们仅能实现 `Functor` 实例在只有一个类型参数的类型上。

例如我们不可以将 `fmap` 实现在 `Either a b` 或 `(a, b)` 上，因为它们有两个类型参数。同样的，我们不能实现在 `Bool` 或 `Int` 上，因为它们没有类型参数。

另外，我们不能在类型约束上添加任何约束。这是什么意思呢？为了更好的进行解释，首先看一下普通的 `data` 定义以及其 `Functor` 实例。

```

1 data Foo a = Foo a
2
3 instance Functor Foo where
4   fmap f (Foo a) = Foo (f a)

```

当我们定义一个新的类型，我们可以添加一个类型约束在 `data` 关键字之后。

```

1 data Eq a => Bar a = Bar a
2
3 instance Functor Bar where
4   fmap f (Bar a) = Bar (f a)

```

报错如下：

```

1 • No instance for (Eq a) arising from a use of ‘Bar’
2   Possible fix:
3     add (Eq a) to the context of
4     the type signature for:
5     fmap :: forall a b. (a -> b) -> Bar a -> Bar b
6 • In the pattern: Bar a
7   In an equation for ‘fmap’: fmap f (Bar a) = Bar (f a)
8   In the instance declaration for ‘Functor Bar’ typecheck(-Wdeferred-type-errors)

```

类型定义上的约束并不好

在一个类型定义上添加一个约束这种行为基本上永远不会是一个好主意。它造成的影响便是强迫用户在每个函数上也都添加类型约束。假设我们需要一个栈数据结构，使得我们可以查询其元素是否遵从某种排序。下面是该数据类型的定义：

```
1 data (Ord a) => OrdStack a
2   = Bottom
3   | Item a (OrdStack a)
4   deriving (Show)
```

如果我们希望编写一个函数用于检查栈结构是否增加，我们显然需要一个 `Ord` 约束来执行一对元素的比较。

```
1 isIncreasing :: (Ord a) => OrdStack a -> Bool
2 isIncreasing (Item a rest@(Item b _))
3   | a < b = isIncreasing rest
4   | otherwise = False
5 isIncreasing _ = True
```

然而因为我们在类型定义上加上了类型约束，该约束实际上影响了完全不需要的地方：我们需要添加 `Ord` 约束至 `push`，其本身并不关心堆上的顺序。

```
1 push :: (Ord a) => a -> OrdStack a -> OrdStack a
2 push a s = Item a s
```

这个时候尝试移除上述函数的 `Ord` 约束，`push` 则会在类型检查时失败。

这也是为什么较早之前我们尝试着为 `Bar` 编写一个 `Functor` 实例时失败了：它需要 `Eq` 约束在 `fmap` 上。

现在我们有 `Haskell` 中在类型定义上添加类型约束是不好的认知，那么更合理的做法呢？答案就是简单的移除在类型定义上的类型约束，且放置在真正需要约束的函数上。

`fmap` 的中缀使用

很多时候我们会看到 `fmap` 作为操作符一样调用：

```
1 ghci> (1+) `fmap` [1,2,3] ++ [4,5,6]
2 [2,3,4,4,5,6]
```

如果希望作为操作符那样使用 `fmap`，那么 `Control.Applicative` 模块中有一个操作符 `(<$>)`，它是 `fmap` 的别名。名称中 `$` 展现了一种相似性，即将一个函数应用至其参数（通过 `($)` 操作符）以及将一个函数提升至一个函子。

灵活的实例

我们可能希望能为 `Either Int b` 实现一个 `Functor` 实例，因为它仅有一个类型参数。

```

1 instance Functor (Either Int) where
2   fmap _ (Left n) = Left n
3   fmap f (Right r) = Right (f r)

```

然而 Haskell 98 的类型系统不能保证，这样的实例约束的检查能终止。非终止的约束检查可能会让编译器陷入无限循环，因此禁用了这样形式的实例。

```

1 • Illegal instance declaration for ‘Functor (Either Int)’
2   (All instance types must be of the form (T a1 ... an)
3   where a1 ... an are *distinct type variables*,
4   and each type variable appears at most once in the instance head.
5   Use FlexibleInstances if you want to disable this.)
6 • In the instance declaration for ‘Functor (Either Int)’ typecheck

```

原书使用一下代码解决编译出错的问题：

```

1 {-# LANGUAGE FlexibleInstances #-}
2
3 instance Functor (Either Int) where
4   fmap _ (Left n) = Left n
5   fmap f (Right r) = Right (f r)

```

然而实际上新的 Haskell 仍然不能通过编译，报错如下：

```

1 • Overlapping instances for Functor (Either Int)
2   arising from a use of ‘GHC.Base.$dm<$’
3   Matching instances:
4     instance Functor (Either a) -- Defined in ‘Data.Either’
5     instance Functor (Either Int)
6       -- Defined at <...>/EitherIntFlexible.hs:7:10
7 • In the expression: GHC.Base.$dm<$ @ (Either Int)
8   In an equation for ‘<$’ : (<$) = GHC.Base.$dm<$ @ (Either Int)
9   In the instance declaration for ‘Functor (Either Int)’ typecheck(-Wdeferred-type-errors)

```

实际上，Haskell 的标准库早已更新，详见上一本书的笔记《Learn Me a Haskell》中第八章的“函子 typeclass”部分。Haskell 的标准实现如下：

```

1 instance Functor (Either a) where
2   fmap f (Right x) = Right (f x)
3   fmap f (Left x) = Left x

```

因为有了 Haskell 的标准实现，我们可以直接进行以下操作：

```

1 ghci> fmap (== "cheeseburger") (Left 1 :: Either Int String)
2 Left 1
3 ghci> fmap (== "cheeseburger") (Right "fries" :: Either Int String)
4 Right False

```

更多有关函子的思考

我们已经做了一些有关函子该如何工作的隐式假设。那么将它们显式的展示出来并思考它们的规则就更有帮助了，因为这让我们可以将函子视为一致的，良好行为的对象。我们仅需记住两个简单的规则。

第一条规则就是函子必须是保持恒等 *identity* 的，也就是说将 `fmap id` 应用至一值时，返回的总是相同的值。

```
1 ghci> fmap id (Node (Leaf "a") (Leaf "b"))
2 Node (Leaf "a") (Leaf "b")
```

第二条规则则是函子必须是可组合的 *composable*，也就是说将两个 `fmap` 组合起来使用与分别两次 `fmap` 使用的结果是一致的。

```
1 ghci> (fmap even . fmap length) (Just "twelve")
2 Just True
3 ghci> fmap (even . length) (Just "twelve")
4 Just True
```

另一种看待这两天规则的方式是，函子必须保持形状 *shape*。集合的结构不应该受到函子的影响；只有它所包含的值应该改变。

```
1 ghci> fmap odd (Just 1)
2 Just True
3 ghci> fmap odd Nothing
4 Nothing
```

如果你在编写一个 `Functor` 实例，那么牢记这些规则并且进行了测试是很有帮助的，因为编译器并不能上述列举的规则。另一方面，如果仅仅使用函子，那么规则是“自然而然的”无需记住它们。

为解析编写函子实例

对于已经调研过得这些类型，我们对 `fmap` 所预期的行为就显而易见了。相较于 `Parse` 因为复杂度的缘故会较难理解。一个合理的猜想就是我们所 `fmap` 的函数应该被应用在一个解析器的当前结果上，同时不对解析状态做任何修改。

```
1 instance Functor Parse where
2   fmap f parser =
3     parser ==> \result ->
4       identity (f result)
```

这个定义很容易阅读，让我们用几个快速的测试来检查我们是否遵循了函子的规则。

首先检查的是 `identity` 是否被遵循。首先是一个应该失败的解析：从一个空的字符串中解析一个字节（别忘了 `(<$>)` 就是 `fmap`）。


```

1 ghci> :l Parse.hs
2 [1 of 2] Compiling Main             ( Parse.hs, interpreted )
3 Ok, one module loaded.
4 ghci> parse parseByte L.empty
5 Left "byte offset 0: no more input"
6 ghci> parse (id <$> parseByte) L.empty
7 Left "byte offset 0: no more input"

```

现在测试一下成功的案例：

```

1 ghci> let input = L8.pack "foo"
2 ghci> L.head input
3 102
4 ghci> parse parseByte input
5 Right 102
6 ghci> parse (id <$> parseByte) input
7 Right 102

```

通过上述结果同样可知我们的函子实例遵循了第二条规则，即保持形状。失败保持失败，成功保持成功。

最后就是组合起来也是保存原有形状的。

```

1 ghci> :m +Data.Char
2 ghci> parse ((chr . fromIntegral) <$> parseByte) input
3 Right 'f'
4 ghci> parse (chr <$> fromIntegral <$> parseByte) input
5 Right 'f'

```

为解析使用函子

所有关于函子的讲解都是为了一个目的：它们可以让我们编写简洁有力的代码。回忆一下之前介绍过的 `parseByte` 函数。我们总是想要处理 ASCII 字符而不是 `Word8` 值。

我们可以模仿 `parseByte` 那样编写一个 `parseChar` 函数，不过现在可以通过 `Parse` 的函子特性来避免这些重复的代码。函子接受一个解析的结果并为该结果应用一个函数，因此我们需要一个函数可以将 `Word8` 转换成一个 `Char`。

```

1 w2c :: Word8 -> Char
2 w2c = chr . fromIntegral
3
4 parseChar :: Parse Char
5 parseChar = w2c <$> parseByte

```

可以使用函子来编写一个简洁的“peek”函数。它将在输入的字符串结束时返回 `Nothing`；否则返回下一个字符，同时不消耗它（即检查时不会影响当前的解析状态）。

```

1 peekByte :: Parse (Maybe Word8)
2 peekByte = fmap fst . L.uncons . string <$> getState

```

同样的方法可以为 `parseChar` 定义一个更加简洁的 `peekChar` 版本。

```
1 peekChar :: Parse (Maybe Char)
2 peekChar = fmap w2c <$> peekByte
```

注意 `peekByte` 与 `peekChar` 都调用了 `fmap`，即 `<$>`。这是因为 `Parse (Maybe a)` 类型是一个函子中的函子。因此我们需要将一个函数提升两次来“走进”里面的函子。

最后让我们编写另一个泛用的组合子，也就是类似于 `takeWhile` 版本的 `Parse`：消费输入直到谓词返回 `True`。

```
1 parseWhile :: (Word8 -> Bool) -> Parse [Word8]
2 parseWhile p =
3   (fmap p <$> peekByte) ==> \mp ->
4     if mp == Just True
5     then
6       parseByte ==> \b ->
7         (b :) <$> parseWhile p
8     else identity []
```

下面是一个不使用函子的繁琐版本，不过更加直接：

```
1 parseWhileVerbose p =
2   peekByte ==> \mc ->
3     case mc of
4       Nothing -> identity []
5       Just c
6         | p c ->
7           parseByte ==> \b ->
8             parseWhileVerbose p ==> \bs ->
9               identity (b : bs)
10        | otherwise ->
11          identity []
```

这个繁琐的版本在不熟悉函子时虽然看似易读，但是在 Haskell 中使用函子更为常见，它的表达更加简洁，因此应该同时成为读与写的本能。

重写 PGM 解析器

没有新的解析代码时，原始的 PGM 解析函数会是什么样子呢？

```
1 parseRawPGM =
2   parseWhileWith w2c notWhite ==> \header ->
3     skipSpaces
4     ==>& assert (header == "P5") "invalid raw header"
5     ==>& parseNat
6     ==> \width ->
7       skipSpaces
8       ==>& parseNat
9       ==> \height ->
```

```

10     skipSpaces
11     ==>& parseNat
12     ==> \maxGrey ->
13         parseByte
14         ==>& parseBytes (width * height)
15         ==> \bitmap ->
16             identity (Greymap width height maxGrey bitmap)
17 where
18     notWhite = (`notElem` " \r\n\t")

```

该定义又使用了以下的帮助函数：

```

1  parseWhileWith :: (Word8 -> a) -> (a -> Bool) -> Parse [a]
2  parseWhileWith f p = fmap f <$> parseWhile (p . f)
3
4  parseNat :: Parse Int
5  parseNat =
6      parseWhileWith w2c isDigit ==> \digits ->
7          if null digits
8              then bail "no more input"
9          else
10             let n = read digits
11                 in if n < 0
12                     then bail "integer overflow"
13                     else identity n
14
15  (==>&) :: Parse a -> Parse b -> Parse b
16  p ==>& f = p ==> const f
17
18  skipSpaces :: Parse ()
19  skipSpaces = parseWhileWith w2c isSpace ==>& identity ()
20
21  assert :: Bool -> String -> Parse ()
22  assert True _ = identity ()
23  assert False err = bail err

```

`(==>&)` 组合子像 `(==>)` 那样将解析器链接起来，不过右侧的解析器会无视左侧所返回的结果。`assert` 函数允许我们检查一个属性，当该属性为 `False` 时会终止解析并返回一个有用的错误信息。

最后就是检查与修改解析状态。

```

1  parseBytes :: Int -> Parse L.ByteString
2  parseBytes n =
3      getState ==> \st ->
4          let n' = fromIntegral n
5              (h, t) = L.splitAt n' (string st)
6              st' = st {offset = offset st + L.length h, string = t}
7              in putState st'
8              ==>& assert (L.length h == n') "end of input"

```

9 `==>& identity h`

未来的方向

略

11 Testing and quality assurance

构建真实系统意味着需要考虑质量控制，鲁棒性以及正确性。使用正确的质量保证机制，良好的代码类似于一个各项功能完备的精确的仪器。

Haskell 中有着若干工具帮助我们构建这样精准的系统。最显著的当然是语言自带的类型系统，它允许静态的强制执行复杂的不变量 – 使得编写的代码无法违背其约束。除此之外，纯粹性以及多态鼓励模块化，可重构和可测试的代码风格。

在维护代码的正确表达性上测试担任了很重要的角色。Haskell 中主要的测试机制是传统的单元测试（通过 HUnit 库），以及其更强大的派生品：基于类型的“属性”测试，即 QuickCheck 一个开源的 Haskell 测试框架。基于属性的测试鼓励更高层次的测试方法，是以抽象不变函数的形式测试，需要满足测试库所生成真实的测试数据。相较于手写测试用例的不足，通过这样测试的代码通常能覆盖所有的边缘用例。

本章我们学习如何使用 QuickCheck 来构建代码的不变性，并测试上一章节编写的代码。另外还会学习如何使用 GHC 代码覆盖工具：HPC。

QuickCheck：基于类型的测试

为了展示基于属性的测试是如何工作的，我们用一个简单的场景开始：用户编写了一个特殊的排序函数，想要测试其行为。

首先是导入 QuickCheck 库，正如其他所需的包那样（通过以下命令在 `.cabal` 存在的项目下安装包 `cabal install --lib QuickCheck`）：

```
1 import Test.QuickCheck
2 import Data.List
```

以及希望被测试的函数 – 一个自定义的排序：

```
1 qsort :: (Ord a) => [a] -> [a]
2 qsort [] = []
3 qsort (x : xs) = qsort lhs ++ [x] ++ qsort rhs
4 where
5     lhs = filter (< x) xs
6     rhs = filter (>= x) xs
```

这是一个典型的 Haskell 排序实现：学习函数式编程的优雅，不过没那么高效（这并不是一个替换排序）。现在我们希望检查该函数是否遵循基本的排序规则。一个不错的不变性测试就是幂等性 – 运行一个函数两次的返回是否一致。这个不变性的测试不难：

```
1 prop_idempotent xs = qsort (qsort xs) == qsort xs
```

我们将使用 QuickCheck 来转换带有 `prop_` 的前缀测试属性，用于区别普通的代码。这个幂等性的属性被简单的编写为 Haskell 函数来表明任何输入的数据被排序后是否具有相等性。我们通过手动测试若干案例的方式来检查它是否合理：

```

1 ghci> :l QC-basics.hs
2 [1 of 2] Compiling Main             ( QC-basics.hs, interpreted )
3 Ok, one module loaded.
4 ghci> prop_idempotent []
5 True
6 ghci> prop_idempotent [1,1,1,1]
7 True
8 ghci> prop_idempotent [1..100]
9 True
10 ghci> prop_idempotent [1,5,2,1,2,0,9]
11 True

```

看起来不错，不过手动输入数据还是太麻烦了，同时也违背了函数式编程的高效性：让机器做这事儿！QuickCheck 库带有 Haskell 所有数据类型的数据生成器可以用于自动化手动输入这个过程。通过类型系统所生成的（临时）随机数据，QuickCheck 使用 `Arbitrary` typeclass 来表示的统一接口。QuickCheck 通常会隐藏数据生成的过程，不过我们仍然可以通过手动的方式运行生成器来获取 QuickCheck 所生成的数据分布。例如，生成一个随机布尔值列表。

```

1 ghci> generate arbitrary :: IO [Bool]
2 [False,False,True,False,True,False,False]

```

注：新的 QuickCheck 的 `generate` 的函数签名是 `generate :: Gen a -> IO a`。

QuickCheck 生成的测试数据就像是这样，并通过 `quickCheck` 函数传递给所选择的属性。属性其本身的类型决定了数据生成器的使用。`quickCheck` 接着检查属性是否满足所有生产的数据。

```

1 ghci> :t quickCheck
2 quickCheck :: Testable prop => prop -> IO ()
3 ghci> quickCheck (prop)
4 product           properFraction           propertyForAllShrinkShow
5 prop_idempotent   property
6 ghci> quickCheck (prop_idempotent :: [Integer] -> Bool)
7 +++ OK, passed 100 tests.

```

对于生成的 100 个不同的列表，属性都可以适配。在开发测试时，看到每个测试用例中真实生成的数据是很有意义的。为了达到这个效果，可以使用 `verboseCheck` 函数来查看更详细的输出。

测试属性

好的库包含了一系列正交的基础，它们彼此之间都有合理的关联。使用 QuickCheck 来指定函数间的关系，通过开发属性间合理交互的函数来帮助我们找到好的库接口。QuickCheck 这时所扮演的是 API 的“lint”工具 – 其提供了确保我们库 API 合理性的机器支持。

列表排序函数当然拥有若干有意思的属性，它们与其它列表操作紧密关联。例如：排序列表中第一个元素总是最小的元素。我们可以在 Haskell 中指定这一特性，通过 `List` 库的

`minimum` 函数:

```
1 prop_minimum xs = head (qsort xs) == minimum xs
```

不过会抛出异常:

```
1 ghci> quickCheck (prop_minimum :: [Integer] -> Bool)
2 *** Failed! (after 1 test):
3 Exception:
4 Prelude.head: empty list
5 CallStack (from HasCallStack):
6   error, called at libraries/base/GHC/List.hs:1646:3 in base:GHC.List
7   errorEmptyList, called at libraries/base/GHC/List.hs:85:11 in base:GHC.List
8   badHead, called at libraries/base/GHC/List.hs:81:28 in base:GHC.List
9   head, called at QC-basics.hs:21:19 in main:Main
10 []
```

也就是这个属性只满足非空列表。感谢 QuickCheck 有着一整套属性使得我们可以更加精确的指定我们的不变，剔除不希望考虑到的值。对于非空列表，我们真实想表达的是：如果列表是非空的，那么排序列表中的第一个元素是最小的。根据 `(==>)` 隐式函数，在运行属性前剔除非空的数据:

```
1 prop_minimum' xs = not (null xs) ==> head (qsort xs) == minimum xs
```

通过分离空列表，现在可以确认属性实际上成立:

```
1 ghci> quickCheck (prop_minimum' :: [Integer] -> Property)
2 +++ OK, passed 100 tests; 18 discarded.
```

注意我们将属性的类型从简单的 `Bool` 结果变为了更泛化的 `Property` 类型（属性自身现在是一个函数，在测试前筛选非空列表，而不是简单的一个布尔常数）。

我们现在可以通过其它应该满足的不变来完成排序函数的所有基础属性了：输出是有序的（每个元素都应该小于等于其前者）；输出是输入的组合（通过列表差异函数 `(\\)`；最后被排序的元素是最大的元素；如果找到了两个不同列表的最小元素，那么将两个列表合并后再排序，该元素还是第一个元素。这些属性可以表示为：

```
1 prop_ordered xs = ordered (qsort xs)
2 where
3   ordered [] = True
4   ordered [x] = True
5   ordered (x : y : xs) = x <= y && ordered (y : xs)
6
7 prop_permutation xs = permutation xs (qsort xs)
8 where
9   permutation xs ys = null (xs \\ ys) && null (ys \\ xs)
10
11 prop_maximum xs = not (null xs) ==> last (qsort xs) == maximum xs
12
13 prop_append xs ys =
```

```

14 not (null xs)
15   => not (null ys)
16   => head (qsort (xs ++ ys))
17   == min (minimum xs) (minimum ys)

```

测试模型

另一个技巧就是测试模型的实现。可以将列表排序的实现绑定标准库中的排序函数，同时如果它们表现的一致，就可以得知自定义的排序的正确性了。

```

1 prop_sort_model xs = sort xs == qsort xs

```

这一类基于模型的测试异常强大。通常开发人员会有一个参考现实或原型，虽然不够高效，但却是正确的。这可以保存下来，并用于确保优化后的产品代码符合这套参考。通过构建一套大型的基于模型的测试，并定期运行它们（例如在每次提交时），我们可以轻松的确确保代码的准确性。大型 Haskell 项目通常绑定了与项目本身大小相当的属性套件，在每次更改时都要测试数千个不变量，保证代码符合规范，并确保能按要求运行。

测试用例：指定一个漂亮打印机

在 Haskell 所构建的大型系统中，为单独的函数测试它们的自然属性属于基础的构建模块。接下来我们将要学习更复杂的场景：为之前章节所编写的漂亮打印库构建测试套装。

生成测试数据

回忆一下漂亮打印是基于 `Doc` 所构建的程序，它是一个代表着良好格式化后的文档的代数数据结构。

```

1 data Doc
2   = Empty
3   | Char Char
4   | Text String
5   | Line
6   | Concat Doc Doc
7   | Union Doc Doc
8   deriving (Show, Eq)

```

库本身是实现的一系列函数用于构建与转换该文档类型的值，最后将完成的文档渲染成字符串。

QuickCheck 鼓励采用一种测试方法，开发人员指定的不变量可以满足任何数据类型。为了测试漂亮打印库，我们还需要输入数据源。利用 QuickCheck 中 `Arbitrary` 类所提供的组合字套件来构建随机数据。该类提供了一个函数，`arbitrary`，用于生成每种类型的数据，通过它我们可以为自定义的数据类型定义数据生成器。


```

1 class Arbitrary a where
2   arbitrary :: Gen a

```

注：该 `typeclass` 由 `QuickCheck` 库提供。

需要注意的是生成器运行在一个 `Gen` 环境中，由类型表示。这是一个简单的状态传递单子，用于隐藏贯穿代码的随机数生成器状态。我们会在稍后的章节中讲解单子，现在我们只需要知道 `Gen` 定义为单子，可以使用 `do` 语义来编写可访问隐式随机数源的生成器。为了编写自定义类型的生成器，我们使用库中定义的一组函数来引入新的随机值，并将它们粘合在一起用以构建我们需要的数据结构类型。关键函数的类型为：

```

1 elements :: [a] -> Gen a
2 choose :: Random a => (a, a) -> Gen a
3 oneof :: [Gen a] -> Gen a

```

注：以上三个函数皆由 `QuickCheck` 库提供。

函数 `elements`，接受一个列表，返回一个生成器，该生成器将随机返回该列表中的值。`choose` 与 `oneof` 将在稍后用到。通过它们我们可以为一个简单的类型编写生成器。例如：

```

1 data Ternary
2   = Yes
3   | No
4   | Unknown
5   deriving (Eq, Show)

```

为其实现 `Arbitrary` `typeclass`：

```

1 instance Arbitrary Ternary where
2   arbitrary = elements [Yes, No, Unknown]

```

另外一种数据生成的方案就是先生成一种 `Haskell` 的基础类型，再转换成期望的实际类型。通过 `choose` 来选择 0 至 2 的随机整数，然后再映射成 `ternary` 值：

```

1 instance Arbitrary Ternary where
2   arbitrary = do
3     n <- choose (0, 2) :: Gen Int
4     return $ case n of
5       0 -> Yes
6       1 -> No
7       _ -> Unknown

```

对于简单的 `sum` 类型，这种方法可以很好的工作，因为整数可以很好的映射到数据类型的构造函数上。对于 `product` 类型（例如结构或元组），我们需要为每个成员分别生成乘积（递归嵌套类型），再将成员进行合并。例如生成一对随机值：

```

1 instance (Arbitrary a, Arbitrary b) => Arbitrary (a, b) where
2   arbitrary = do
3     x <- arbitrary
4     y <- arbitrary
5     return (x, y)

```

注：上述代码同样由 QuickCheck 库提供。

现在来为所有 `Doc` 类型的变体来编写一个生成器。我们需要将问题进行分解，首先是为每个类型生成随机构造函数，接着根据结果生成每个字段的组件。这里最复杂的部分就是 `union` 以及 `concatenation` 变体了。

首先，我们需要编写一个实例来生成随机字符 – QuickCheck 并没有默认的字符实例，因为有大量不同的文本编码，我们希望一个字符测试。我们无需考虑文档中的文本内容，因此简单的字母以及标点符号的生成器就足够了（更丰富的生成器可以基于该生成器进行拓展）：

```
1 newtype DocChar = DocChar Char
2
3 fromDocChar :: DocChar -> Char
4 fromDocChar (DocChar x) = x
5
6 instance Arbitrary DocChar where
7   arbitrary = elements (DocChar <$> ['A' .. 'Z'] ++ ['a' .. 'z'] ++ " ~!@#%~&*()")
```

注：与原文不同在于使用了 `newtype`，因为 QuickCheck 已经为 `Char` 实现了 `Arbitrary`。这里 `(<$>)` 的计算优先级是 4，而 `(++)` 的优先级是 5，因此不需要额外的括号。

有了这些现在就可以编写文档的实例了，即通过枚举构造函数，并填充其字段。我们选择随机整数来表示所需要生成的文档变体，再根据结果进行分配。生成 `concat` 与 `union` 文档节点时，只需要递归 `arbitrary`，将类型推导交给 `Arbitrary` 的实例本身：

```
1 instance Arbitrary Doc where
2   arbitrary = do
3     n <- choose (1, 6) :: Gen Int
4     case n of
5       1 -> return Empty
6       2 -> do
7         (x :: DocChar) <- arbitrary
8         return (Char $ fromDocChar x)
9       3 -> Text <$> arbitrary
10      4 -> return Line
11      5 -> do
12        x <- arbitrary
13        Concat x <$> arbitrary
14      6 -> do
15        x <- arbitrary
16        Union x <$> arbitrary
```

注：与原文不同之处在于处理 `Char` 的步骤，先是显式声明了 `x` 为 `DocChar`，然后再使用 `fromDocChar` 取出 `Char` 值。

上述代码非常的直接，我们可以通过 `oneof` 函数对其进行重写，该函数我们较早之前见过，其作用是在列表中挑选一个后生成生成器（我们也可以使用单子化的组合子，用 `liftM` 来避免命名生成器的各种中间结果）：

```

1 instance Arbitrary Doc where
2   arbitrary =
3     oneof
4       [ return Empty,
5         fmap (Char . fromDocChar . DocChar) arbitrary,
6         fmap Text arbitrary,
7         return Line,
8         liftM2 Concat arbitrary arbitrary,
9         liftM2 Union arbitrary arbitrary
10      ]

```

注：与原文不同之处在于列表第二个元素 `Char` 的处理方式上，通过 newtype `DocChar` 的 `arbitrary` 函数再提取出 `Char` 来实现调用非默认的 `Char` `arbitrary`。

上述两种对 `Doc` 实现 `Arbitrary` 实例的方式，后者更为简洁，仅仅是在生成器列表中进行选择，不过二种表达方式讲述的都是同一件事。我们可以通过生成一个随机文档列表来进行测试

```

1 ghci> generate arbitrary :: IO [Doc]
2 [Concat (Union (Text "1\_FS\83431\nir^\DC1IH\t\GS\10583r\"_ -t\1007943(\1091024\180519\ETX
   \53558y\998555") Line) (Union (Text "s4hxR(\1108184") Empty),Empty,Empty,Char '&',Char 'D',
   Empty,Line,Empty,Line,Char '>',Te
3 xt "99986\25864;'}\CAN",Concat Line Empty,Text "\NAK",Char '&',Union Empty (Concat (Char
   '\998638') (Union
4 (Text "\t\28481\1025790\95127\1078739x;D") (Concat (Char 'Q') Line))),Empty,Concat Line (
   Text "Gwj~v\1032477\EOT\58278k},^\1016570\59993:Y>\1074673q;A \7956\127844"),Empty,Concat
   Empty (Text "\199725\SI1(<\176480\133060,A\FSu\EM>\DC1\SYN\1102841\RS\STX\DELim\n
   \"|\1106202De\n#"),Char 'J',Union Empty Line,Concat (Union
5 (Concat Line (Char 'E')) (Char '\NUL')) (Concat (Text "78\152486\194828?vy\157411\1*\DC1\
   NAK\NULrd\SYN:\t
6 ") Line),Text "")]

```

注：与原文不同，`generate` 的函数签名变了。

结果是各种文档变体混合的列表。之后在每次测试运行时会生成上百个结果。现在可以为文档函数编写一些更泛化的属性了。

测试文档构建

基于文档的两个基础函数是 `null` 文档常数（一个 `nullary` 函数），`empty`，以及 `append` 函数。它们的类型是：

```

1 empty :: Doc
2 (<>) :: Doc -> Doc -> Doc

```

总之，它们有一个很好的属性：将空列表 `append` 或 `prepend` 到第二个列表上，第二个列表保持不变。我们可以将这个不变量声明为一个属性：

```

1 prop_empty_id x =

```

```

2 empty <> x == x
3   && x <> empty == x

```

注：原有 Doc 代码在第五章的 `Prettify.hs` 文件中（改动后位于 `code/Ch11/Prettify2.hs` 文件下）。与第五章不一样的地方：为 `Doc` 实现了 `Semigroup`，因此可以对 `Doc` 使用操作符 `(<>)`。

为了确保它是正确的，测试一下：

```

1 ghci> quickCheck prop_empty_id
2 00, passed 100 tests.

```

将 `quickCheck` 改为 `verboseCheck` 可以看到生成出来的测试文档。

其它的 API 测试：

```

1 prop_char c = char c == Char c
2
3 prop_text s = text s == if null s then Empty else Text s
4
5 prop_line = line == Line
6
7 prop_double d = double d == text (show d)

```

以列表作模型

高阶函数是重用程序的基本粘合剂，漂亮打印库也不例外 – 自定义的 `fold` 函数用于内部实现文档拼接以及文档块之间的交错分隔符。文档的 `fold` 接受一个文档列表，通过提供的组合函数将它们粘合起来：

```

1 fold :: (Doc -> Doc -> Doc) -> [Doc] -> Doc
2 fold f = foldr f empty

```

那么为 `fold` 编写测试就很简单了。

```

1 prop_hcat xs = hcat xs == glue xs
2   where
3     glue [] = empty
4     glue (d : ds) = d <> glue ds

```

同理对于 `punctuate`，我们可以通过列表散布（interspersions）的方式来插入标点符号：

```

1 import Data.List (intersperse)
2
3 prop_punctuate s xs = punctuate s xs == intersperse s xs

```

然而对其进行测试时，我们会得到错误：

```

1 ghci> quickCheck prop_punctuate
2 *** Failed! Falsified (after 6 tests and 1 shrink):
3 Char ' '
4 [Union (Text "\ESC") (Text "R0"), Union Empty (Char '\1097718')]

```

漂亮打印库优化掉了多余的空文档，这是模型实现无法做到的，因此我们需要增强模型以匹配实际情况。首先，可以在整个文档列表中散布标点符号文本，然后用一个小的循环来清理分散在其中的 `Empty` 文档，像是这样：

```
1 prop_punctuate' s xs = punctuate s xs == combine (intersperse s xs)
2   where
3     combine [] = []
4     combine [x] = [x]
5     combine (x : Empty : ys) = x : combine ys
6     combine (Empty : y : ys) = y : combine ys
7     combine (x : y : ys) = x `Concat` y : combine ys
```

通过 CHCi 来确认结果。

```
1 ghci> :l QC.hs
2 [1 of 4] Compiling Prettify2      ( Prettify2.hs, interpreted )
3 [2 of 4] Compiling Arbitrary    ( Arbitrary.hs, interpreted )
4 [3 of 4] Compiling Main        ( QC.hs, interpreted )
5 Ok, three modules loaded.
6 ghci> quickCheck prop_punctuate'
7 +++ OK, passed 100 tests.
```

将它们合在一起

我们可以将所有的测试放到一个单独文件中，然后使用 QuickCheck 的派生函数来运行。基本的批处理驱动程序通常已经足够好了。

```
1 {-# LANGUAGE TemplateHaskell #-}
2 import Test.QuickCheck.All
3
4 -- All the prop tests before `return []`
5
6 return []
7 runTests = $quickCheckAll
```

注：根据StackOverflow的答案，QuickCheck 在 2.0 之后就没了 `Test.QuickCheck.Batch`，因此需要在 `QC.hs` 文件的头尾加上上述代码。而新的 `Run.hs` 代码如下：

```
1 import QC
2
3 main = do
4   runTests
5   return ()
```

通过 GHCi 测试：

```
1 ghci> :l Run.hs
2 [1 of 5] Compiling Prettify2      ( Prettify2.hs, interpreted )
3 [2 of 5] Compiling Arbitrary    ( Arbitrary.hs, interpreted )
```

```

4  [3 of 5] Compiling QC                ( QC.hs, interpreted )
5  [4 of 5] Compiling Main              ( Run.hs, interpreted )
6  Ok, four modules loaded.
7  ghci> main
8  === prop_empty_id from ./QC.hs:14 ===
9  +++ OK, passed 100 tests.
10
11 === prop_char from ./QC.hs:18 ===
12 +++ OK, passed 100 tests.
13
14 === prop_text from ./QC.hs:20 ===
15 +++ OK, passed 100 tests.
16
17 === prop_line from ./QC.hs:22 ===
18 +++ OK, passed 1 test.
19
20 === prop_double from ./QC.hs:24 ===
21 +++ OK, passed 100 tests.
22
23 === prop_hcat from ./QC.hs:28 ===
24 +++ OK, passed 100 tests.
25
26 === prop_punctuate from ./QC.hs:33 ===
27 *** Failed! Falsified (after 4 tests):
28 Char 'H'
29 [Concat (Char '1') (Concat (Char '\119225') (Concat (Char 'r') Empty)),Text ""]
30
31 === prop_punctuate' from ./QC.hs:35 ===
32 +++ OK, passed 100 tests.

```

我们在这里将代码构建为一个独立的测试脚本，实例和属性在它们自己的文件中，做到了与库的源代码分离。这是一个典型的库项目，测试与库本身分开，并通过模块系统导入库。然后可以编译和执行测试脚本：

```

1  ghc --make Run.hs
2  Loaded package environment from <xxx>
3  [1 of 5] Compiling Prettify2          ( Prettify2.hs, Prettify2.o, Prettify2.dyn_o )
4  [2 of 5] Compiling Arbitrary          ( Arbitrary.hs, Arbitrary.o, Arbitrary.dyn_o )
5  [3 of 5] Compiling QC                ( QC.hs, QC.o, QC.dyn_o )
6  [4 of 5] Compiling Main              ( Run.hs, Run.o )
7  [5 of 5] Linking Run

```

接着运行：

```

1  ./Run
2  === prop_empty_id from ./QC.hs:14 ===
3  +++ OK, passed 100 tests.
4
5  === prop_char from ./QC.hs:18 ===
6  +++ OK, passed 100 tests.

```

```

7
8   === prop_text from ./QC.hs:20 ===
9   +++ OK, passed 100 tests.
10
11  === prop_line from ./QC.hs:22 ===
12  +++ OK, passed 1 test.
13
14  === prop_double from ./QC.hs:24 ===
15  +++ OK, passed 100 tests.
16
17  === prop_hcat from ./QC.hs:28 ===
18  +++ OK, passed 100 tests.
19
20  === prop_punctuate from ./QC.hs:33 ===
21  *** Failed! Falsified (after 4 tests):
22  Union (Char '\1089590') (Char 't')
23  [Line,Text "\129350\167965\EOT"]
24
25  === prop_punctuate' from ./QC.hs:35 ===
26  +++ OK, passed 100 tests.

```

通过 HPC 测量测试覆盖率

HPC (Haskell Program Coverage) 是编译器的扩展，用于观察给定程序运行期间实际执行的代码部分。这在测试的上下文中很有用，因为它可以让我们精确的观察哪些函数、分支以及表达式被求值。HPC 附带了一个简单的工具来生成有用的程序覆盖图，这可以放大测试套件中的弱点。

我们需要在编译时添加 `-fhpc` 标记：

```

1  ghc -fhpc Run.hs --make
2  Loaded package environment from /Users/jacobxie/.ghc/aarch64-darwin-9.4.7/environments/
   default
3  [1 of 5] Compiling Prettify2          ( Prettify2.hs, Prettify2.o, Prettify2.dyn_o ) [HPC flags
   changed]
4  [2 of 5] Compiling Arbitrary          ( Arbitrary.hs, Arbitrary.o, Arbitrary.dyn_o ) [HPC flags
   changed]
5  [3 of 5] Compiling QC              ( QC.hs, QC.o, QC.dyn_o ) [Source file changed]
6  [4 of 5] Compiling Main              ( Run.hs, Run.o ) [HPC flags changed]
7  [5 of 5] Linking Run [Objects changed]

```

在测试运行期间，程序的跟踪记录被写入至当前目录下的 `.tix` 与 `.mix` 文件中。然后命令行工具 `hpc` 使用这些文件来显示有关的各种统计信息。首先，可以使用 `hpc` 报告标志获得运行期间测试代码的摘要。我们将排除测试程序本身（使用 `--exclude` 标志），以便只关注库的代码：

```

1  hpc report Run --exclude=Main --exclude=QC
2  26% expressions used (54/202)

```

```
3 0% boolean coverage (0/3)
4   0% guards (0/3), 3 unevaluated
5   100% 'if' conditions (0/0)
6   100% qualifiers (0/0)
7 21% alternatives used (8/37)
8   0% local declarations used (0/4)
9 44% top-level declarations used (16/36)
```

最后一行可以看到 42% 的定义在测试运行期间被评估。第一次尝试还算不错。随着我们从库中测试越来越多的函数，这个数字将会上升。文本版本的测试摘要对于快速总结很有用，弹药真正了解发生了什么，那么就可以使用 `markup` 标记：

```
1 hpc markup Run --exclude=Main --exclude=QC
2 Writing: Arbitrary.hs.html
3 Writing: Prettify2.hs.html
4 Writing: hpc_index.html
5 Writing: hpc_index_fun.html
6 Writing: hpc_index_alt.html
7 Writing: hpc_index_exp.html
```

略

12 Barcode recognition

本章我们将会使用第十章所编写的图像解析库来构建一个识别条形码的应用。

关于条形码

绝大多数包装好的和批量生产的消费品上面都会有条形码。尽管在各个专门的领域都有若干的条形码系统，但消费品通常使用 UPC-A 或 EAN-13。前者是在美国开发的，而后者则起源于欧洲。

EAN-13 开发晚于 UPC-A，它是 UPC-A 的超集。（实际上，UPC-A 已经在 2005 年宣布废弃了，尽管在美国它仍然被大量的使用。）任何可以识别 EAN-13 的软件或硬件也会自动处理 UPC-A 条形码。

正如其名，EAN-13 描述了一个 13 位数的序列，它被划分成四个组：

- 开头的两位数描述了数字系统。它可以表明制造商的国家，或是其它的一些分类，例如 ISBN（书籍身份）。
- 接下来的五位数则是生产商的 ID，通过国家的授权进行指定。
- 再接下来的五位数则是产品 ID，通过制造商指定。（小型的制造商可能有更长的制造商 ID 以及更短的产品 ID，但是它们加在一起仍然是十位数。）
- 最后一位则是校验数，允许一个扫描器验证它扫描的位数字符串。

EAN-13 编码

在我们关心 EAN-13 条形码的解码之前，我们需要理解它们是如何被编码的。EAN-13 所使用的系统有点复杂。让我们首先从校验数开始，也就是字符串的最后一个位数。

```

1 checkDigit :: (Integral a) => [a] -> a
2 checkDigit ds = 10 - (sum products `mod` 10)
3   where
4     products = mapEveryOther (* 3) (reverse ds)
5
6 mapEveryOther :: (a -> a) -> [a] -> [a]
7 mapEveryOther f = zipWith ($) (cycle [f, id])

```

这是一种通过代码比口头描述更容易理解的算法。从字符串的右侧开始计算，每个连续的数字要么乘以 3，要么不动（`cycle` 函数会无限的重复它的输入列表）。校验数则是它们的和（以 10 取模）与数字 10 之间的差。

条形码是一系列固定宽度的条码，黑色代表二元“1”位，白色代表“0”位。因此，相同数字的运行看前来像一个更粗的条码。

条形码中的位序列如下所示：

- 前导保护序列，编码为 101。
- 一组六位数字，每个有七位宽。
- 另一个保护序列，编码为 01010。
- 一组六位以上的数字。
- 尾部的保护序列，编码为 101。

左边和右边组中的数字有单独的编码。左边用奇偶校验位编码，奇偶校验位编码条形码的第 13 位。

数组介绍

在继续之前，以下是本章所需的模块导入：

```
1 import Control.Applicative ((<$>))
2 import Control.Monad (forM_)
3 import Data.Array (Array (..), bounds, elems, indices, ixmap, listArray, (!))
4 import Data.ByteString.Lazy.Char8 qualified as L
5 import Data.Char (digitToInt)
6 import Data.Ix (Ix (..))
7 import Data.List (foldl', group, sort, sortBy, tails)
8 import Data.Map qualified as M
9 import Data.Maybe (catMaybes, listToMaybe)
10 import Data.Ratio (Ratio)
11 import Data.Word (Word8)
12 import Parse -- from Chapter 10
13 import System.Environment (getArgs)
```

注：Parse 库从第十章而来。

条形码的编码过程很大程度上是表格驱动的，即使用位模式的小表来决定如何对每个数字进行编码。Haskell 基本的数据类型中列表和元组都不适合用于表格，其包含的元素需要能被随机访问。列表需要线性的遍历才能访问第 k 个元素。元组没有这个问题，但是 Haskell 的类型系统使得很难编写一个函数来接受元组和元素偏移量，并返回元组中偏移量处的元素。

常用的数据类型可供常数耗时的随机访问就是数组了。Haskell 提供了若干数字类型。这里我们将编码表格表达为字符串数组。

最简单的数组类型位于 `Data.Array` 模块中，也正是我们现在所用的。与其它 Haskell 通常的类型一样，这些数组也都是不可变的。不可变数组只有在创建时才能用值填充一次，其它时候是不能修改其内容的。（标准库同样提供了其它的数组类型，有些是可变的，只不过现在我们暂时还用不上。）

```

1  leftOddList =
2    [ "0001101",
3      "0011001",
4      "0010011",
5      "0111101",
6      "0100011",
7      "0110001",
8      "0101111",
9      "0111011",
10     "0110111",
11     "0001011"
12  ]
13
14  rightList = map complement <$> leftOddList
15  where
16    complement '0' = '1'
17    complement '1' = '0'
18
19  leftEvenList = map reverse rightList
20
21  parityList =
22    [ "111111",
23      "110100",
24      "110010",
25      "110001",
26      "101100",
27      "100110",
28      "100011",
29      "101010",
30      "101001",
31      "100101"
32  ]
33
34  listToArray :: [a] -> Array Int a
35  listToArray xs = listArray (0, l - 1) xs
36  where
37    l = length xs
38
39  leftOddCodes, leftEvenCodes, rightCodes, parityCodes :: Array Int String
40  leftOddCodes = listToArray leftOddList
41  leftEvenCodes = listToArray leftEvenList
42  rightCodes = listToArray rightList
43  parityCodes = listToArray parityList

```

`Data.Array` 模块的 `listArray` 函数将列表转为数组。第一个参数是数组所需的边界；第二个参数则是列表。

`Array` 的一个不寻常特征是，它的类型在它包含的数据和索引类型上都是参数化的。一维 `String` 数组的类型是 `Array Int String`，而二维则是 `Array (Int,Int) String`。

```

1 ghci> :m +Data.Array
2 ghci> :t listArray
3 listArray :: Ix i => (i, i) -> [e] -> Array i e

```

构造一个数组：

```

1 ghci> listArray (0,2) "foo"
2 array (0,2) [(0,'f'),(1,'o'),(2,'o')]

```

注意我们指定了数组的上下界。这些界是包含边界的，即从 0 至 2 的数组包含了 0, 1 以及 2。

```

1 ghci> listArray (0,3) [True,False,False,True,False]
2 array (0,3) [(0,True),(1,False),(2,False),(3,True)]
3 ghci> listArray (0,10) "too short"
4 array (0,10) [(0,'t'),(1,'o'),(2,'o'),(3,' '), (4,'s'),(5,'h'),(6,'o'),(7,'r'),(8,'t'),(9,**
Exception: (Array.!): undefined array element

```

当数组被构建后，可以使用 `(!)` 操作符根据索引来访问元素。

```

1 ghci> let a = listArray (0,14) ['a'..]
2 ghci> a ! 2
3 'c'
4 ghci> a ! 100
5 *** Exception: Ix{Integer}.index: Index (100) out of range ((0,14))

```

由于数组的构造函数允许指定数组的边界，我们无需使用类似于 C 语言那样基于零的边界。

```

1 ghci> let a = listArray (-9,5) ['a'..]
2 ghci> a ! (-2)
3 'h'

```

索引类型可以是任意 `Ix` 类型的成员。例如 `Char` 作为索引：

```

1 ghci> let a = listArray ('a','h') [97..]
2 ghci> a ! 'e'
3 101

```

创建一个高阶维度数组，我们可以使用 `Ix` 实例的元组作为索引类型。以下是一个三维数组：

```

1 ghci> let a = listArray ((0,0,0),(9,9,9)) [0..]
2 ghci> a ! (4,3,7)
3 437

```

数组与惰性

用于构造数组的列表，其元素数必须大于等于数组所需的元素。如果没有提供足够的元素则会在运行时抛出异常。该异常将何时出现取决于数组的具体实现。

这里我们使用的是一个非严格的数组类型。

```

1 ghci> let a = listArray (0,5) "bar"
2 ghci> a ! 2
3 'r'
4 ghci> a ! 4
5 *** Exception: (Array.!): undefined array element

```

Haskell 同样提供了严格数组，其行为就不同了。我们将在后续小节中讨论两者的优劣。

折叠数组

`bounds` 函数返回一个描述了创建数组时的边界元组。`indices` 函数返回列表的每个索引。我们可以用其定义一些有用的折叠，因为 `Data.Array` 模块并没有定义任何折叠函数。

```

1 -- Strict left fold, similar to foldl' on lists
2 foldA :: (Ix k) => (a -> b -> a) -> a -> Array k b -> a
3 foldA f s a = go s (indices a)
4 where
5     go s (j : js) = let s' = f s (a ! j) in s' `seq` go s' js
6     go s _ = s
7
8 -- Strict left fold using the first element of the array as its starting value, similar to
9   foldl on lists
10 foldA1 :: (Ix k) => (a -> a -> a) -> Array k a -> a
11 foldA1 f a = foldA f (a ! fst (bounds a)) a

```

你可能会疑惑为什么数组模块并没有提供如此有用的折叠函数。这是因为一维数组与列表之间有一些显著的差异。例如自然而然，折叠有两个方向：从左向右，从右向左。除此之外，每次只能折叠一个元素。

而对于二维的数组来说这根本行不通。首先，这时的折叠可以有多种方案。我们可能还是想每次只折叠单个元素，但是现在有了列折叠或行折叠的可能性。最重要的是，对于一次一个元素的折叠，不再只有两个序列需要遍历。

换言之，对于二维数组，就存在若干可能的折叠方式组合，因此没有太多令人信服的理由为标准库选择少量排列。这个问题在高维情况下会更加复杂，因此最好让开发人员编写适合自身应用需求的折叠。

修改数组元素

对于不变数组而言，虽然存在“修改”函数，实际上并不常用。例如，`accum` 函数接受一个数组，以及一个 `(index, value)` 对的列表，返回一个新的被替换后的数组。

由于数组是不变的，修改一个元素就需要拷贝整个数组，这实际上是相当昂贵的。

另一个数组类型，`Data.Array.Diff` 模块中的 `DiffArray`，尝试在少量修改时减少消耗，但是在编写这本书的时候，它在实际运用中还是太慢了。

Tip

不要放弃希望

在 Haskell 中实际上高效的修改一个数组是可能的，即通过 `ST` 单子。我们将在第 26 章对其进行学习。

对 EAN-13 条形码进行编码

尽管我们的目标是解码条形码，不过拥有一个编码器作为引用是很有帮助的。例如通过 `decode . encode` 来检查输出是否正确，确保我们代码的正确性。

```

1  encodeEAN13 :: String -> String
2  encodeEAN13 = concat . encodeDigits . map digitToInt
3
4  -- this function computes the check digit; don't pass one in.
5  encodeDigits :: [Int] -> [String]
6  encodeDigits s@(first : rest) =
7      outerGuard : lefties ++ centerGuard : righties ++ [outerGuard]
8      where
9          (left, right) = splitAt 5 rest
10         lefties = zipWith leftEncode (parityCodes ! first) left
11         righties = map rightEncode (right ++ [checkDigit s])
12
13 leftEncode :: Char -> Int -> String
14 leftEncode '1' = (leftOddCodes !)
15 leftEncode '0' = (leftEvenCodes !)
16
17 rightEncode :: Int -> String
18 rightEncode = (rightCodes !)
19
20 outerGuard = "101"
21
22 centerGuard = "01010"

```

字符串的编码长度是十二位数字，通过 `encodeDigits` 来添加第十三个校验位数字。

条形码由两组六位数字进行编码，由守护序列位于两组其中以及两侧。那么其它数字呢？

左侧组的每个数字通过奇数或偶数编码，奇偶的选择取决于字符串中的第一位数字。如果第一位为零，那么左侧组则是由偶数进行编码；为一，则是由奇数进行编码。这种方式的编码很优雅，它使得 EAN-13 条形码变得向后兼容旧的 UPC-A 标准。

解码器的约束

在讨论解码钱，需要为条形码可以工作的类别设定一些约束。

手机相机或者网络相机输出的是 JPEG 图像，但是编写一个 JPEG 解码器需要占用若干章节。因此我们通过处理 netpbm 文件格式简化解码问题。我们将使用第十章所开发的解析组合子。

分治法

我们的任务是将一个有效的条形码从相机图片中提取出来。我们可以将这个大问题切分成几个有序的子问题，它们每个都是独立可追溯的。

- 将颜色数据转换成可处理的数据。
- 从图像中采样一个扫描线，并提取一组猜测数据，用以确定这一行中的编码数字可能会是什么。
- 从猜测数据中创建一个有效的解码列表。

更多的子问题将会在后续进行分割。

将彩色图像转换成易于处理的东西

条形码实际上是黑白条的序列，我们希望编写一个简单的解析器，一个简单的表示是单色图像，其中每个像素不是黑色就是白色。

解析一个彩色图像

Netpbm 彩色图像格式仅仅比第十章所提到的灰度图像格式更复杂一点。文件头的识别字符串是“P6”，剩下的头文件与灰度格式一致。图像的文件体内，每个像素都代表三个字节，它们是红，绿，蓝。

我们将图像数据表示为二维的像素数组。这里使用数组是为了更好的学习它。对于应用而言，当然可以使用列表的列表。这里使用数组的唯一优点就是：可以高效的提取一行数据。

```
1 type Pixel = Word8
2
3 type RGB = (Pixel, Pixel, Pixel)
4
5 type Pixmap = Array (Int, Int) RGB
```

我们提供了一些类型同义词使得类型签名更可读。

由于 Haskell 在布局数组方面给予了相当大的自由，我们必须选择一种来表示。我们无需显式的存储图像的维度，因为可以使用 `bounds` 函数来提取它们。

实际上解析器非常简短，感谢我们在第十章所开发的组合子。

```

1  parseRawPPM :: Parse Pixmap
2  parseRawPPM =
3    parseWhileWith w2c (/= '\n') ==> \header ->
4      skipSpaces
5      ==>& assert (header == "P6") "invalid raw header"
6      ==>& parseNat
7      ==> \width ->
8        skipSpaces
9        ==>& parseNat
10       ==> \height ->
11         skipSpaces
12         ==>& parseNat
13         ==> \maxValue ->
14           assert (maxValue == 255) "max value out of spec"
15           ==>& parseByte
16           ==>& parseTimes (width * height) parseRGB
17           ==> \pxs ->
18             identity (listArray ((0, 0), (width - 1, height - 1)) pxs)
19
20  parseRGB :: Parse RGB
21  parseRGB =
22    parseByte ==> \r ->
23      parseByte ==> \g ->
24        parseByte ==> \b ->
25          identity (r, g, b)
26
27  parseTimes :: Int -> Parse a -> Parse [a]
28  parseTimes 0 _ = identity []
29  parseTimes n p = p ==> \x -> (x :) <$> parseTimes (n - 1) p

```

上述代码中仅需要注意的是 `parseTimes`，它调用其它解析器来解析一个给定的时间，再构建一个结果的列表。

灰度转化

现在我们有彩色图像的解析，我们需要将彩色数据转换成单色的。一个中间步骤就是转换数据为灰度。将 RGB 图像转换成灰度图像有一个广泛运用的公式，即基于每个颜色通道的感知亮度。

```

1  luminance :: (Pixel, Pixel, Pixel) -> Pixel
2  luminance (r, g, b) = round (r' * 0.30 + g' * 0.59 + b' * 0.11)
3  where
4    r' = fromIntegral r
5    g' = fromIntegral g
6    b' = fromIntegral b

```

Haskell 的数组是 `Functor` typeclass 的成员，所以我们可以使用 `fmap` 来讲整个图像，或是单个扫描行，从彩色转为灰度。


```

1 type Greymap = Array (Int, Int) Pixel
2
3 pixmapToGreymap :: Pixmap -> Greymap
4 pixmapToGreymap = fmap luminance

```

`pixmapToGreymap` 函数仅用于解释。因为我们将仅检查一个条形码图像的某几行，没有必要做额外的工作来转换我们永远用不上的序列。

灰度至二进制以及类型安全

下一个子问题是将灰度图像转换为一个二值图像，即每个像素要么是开要么是闭。

在一个图像处理应用中，需要处理大量的数字，因此很容易为了几个不同的目的重用相同的数字类型。例如使用 `Pixel` 类型来表述 on/off 状态，使用转换将数字一来代表“on”数字零“off”。

然而像这样为了多个目的而重用类型很快会带来潜在的混乱。为了判断一个“Pixel”是否为一个数值或是一个 on/off 值，我们不再可以简单的通过观察类型签名而知。在某些上下文中，我们可以很容易使用包含“错误类型的数字”的值，并且编译器不会对其捕获，因为类型是正确的。

我们可以尝试引入类型别称来解决这个问题。同样的声明 `Pixel` 作为 `Word8` 的同义词，再使用 `Bit` 类型作为 `Pixel` 的同义词。虽然这可以提高可读性，但是类型同义词仍然无法令编译器为我们做为何有效的检查。

编译器会将 `Pixel` 与 `Bit` 视为同样的类型，所以它不能捕获，例如在期望 `Bit` 值为零或一这样的函数中使用 `Pixel` 或是 253 值，这样的异常。

如果我们自定义单色类型，编译器则会阻止我们不小心混合了其它的类型。

```

1 data Bit = Zero | One deriving (Eq, Show)
2
3 threshold :: (Ix k, Integral a) => Double -> Array k a -> Array k Bit
4 threshold n a = binary <$> a
5   where
6     binary i
7       | i < pivot = Zero
8       | otherwise = One
9     pivot = round $ least + (greatest - least) * n
10    least = fromIntegral $ choose (<) a
11    greatest = fromIntegral $ choose (>) a
12    choose f = foldA1 $ \x y -> if f x y then x else y

```

我们的 `threshold` 函数计算了作为输入的数组的最大最小值。它接受这些值和一个介于 0 到 1 之间的阈值，并计算一个“枢轴”值。接着数组中的每个值，如果其值小于枢轴值，返回 `Zero`，否则 `One`。

我们对图像做了什么

略

寻找匹配的位数

我们的第一个问题就是找到给定位置可能被编码的数值。然后是做几个简化的假设：首先处理的是单行，其次确切知道条形码左边缘在一行中的起始位置。

运行长度编码

我们该如何解决条形码粗细的问题呢？答案是运行图像数据的长度编码。

```
1 type Run = Int
2
3 type RunLength a = [(Run, a)]
4
5 runLength :: (Eq a) => [a] -> RunLength a
6 runLength = map rle . group
7   where
8     rle xs = (length xs, head xs)
```

`group` 函数接受一个一系列 id 元素的列表，将它们分组为子列表。

```
1 ghci> group [1,1,2,3,3,3,3]
2 [[1,1],[2],[3,3,3,3]]
```

`runLength` 函数将每组表示为其长度与第一个元素的二元组。

```
1 ghci> let bits = [0,0,1,1,0,0,1,1,0,0,0,0,0,0,1,1,1,1,0,0,0,0]
2 ghci> runLength bits
3 [(2,0),(2,1),(2,0),(2,1),(6,0),(4,1),(4,0)]
```

由于编码的数据都是一和零，我们可以将编码值扔了而不会失去任何有用的信息，即每次运行仅保留长度。

```
1 runLengths :: Eq a => [a] -> [Run]
2 runLengths = map fst . runLength
```

```
1 ghci> runLengths bits
2 [2,2,2,2,6,4,4]
```

上面的位模式不是随机的；它们是左外守卫也是我们捕获图像中每一行的第一个编码数字。如果丢弃守卫条形码，那么留下的长度为 `[2,6,4,4]`。

缩放运行长度并找到近似匹配

一个可行的方法是缩放运行长度，使它们的和为一。我们将使用 `Ratio Int` 类型而不是通常的 `Double` 来管理这些缩放值，因为 `Ratio` 在 `ghci` 中有更可读的打印。这使调试和开发的交互变得更简单。

```

1  type Score = Ratio Int
2
3  scaleToOne :: [Run] -> [Score]
4  scaleToOne xs = map divide xs
5      where
6          divide d = fromIntegral d / divisor
7          divisor = fromIntegral (sum xs)
8
9  -- A more compact alternative that "knows" we're using Ratio Int:
10 -- scaleToOne xs = map (% sum xs) xs
11
12 type ScoreTable = [[Score]]
13
14 -- "SRL" means "scaled run length"
15 asSRL :: [String] -> ScoreTable
16 asSRL = map (scaleToOne . runLengths)
17
18 leftOddSRL = asSRL leftOddList
19
20 leftEvenSRL = asSRL leftEvenList
21
22 rightSRL = asSRL rightList
23
24 paritySRL = asSRL parityList

```

这里使用 `Score` 类型同义词可以让大多数代码不需要考虑其本身的类型是什么。

我们可以使用 `scaleToOne` 来对正在搜寻的一系列数位进行缩放。我们现在纠正了距离而导致的条形码宽度变化，因为缩放的运行长度编码表中的条目与图像中提取的运行长度序列之间应该有相等接近的匹配。

下一个问题是如何将直觉的“相当近”转为可测量的“足够近”。给定的两个缩放运行长度序列，我们可以计算它们之间的一个近似“距离”。

```

1  distance :: [Score] -> [Score] -> Score
2  distance a b = sum . map abs $ zipWith (-) a b

```

完全匹配的距离会得到零，越弱的匹配则有越大的距离。

```

1  ghci> let group = scaleToOne [2,6,4,4]
2  ghci> distance group (head leftEvenSRL)
3  13 % 28
4  ghci> distance group (head leftOddSRL)
5  17 % 28

```

给定一个缩放后的运行长度表，选择表中最优的几个匹配用作输入序列。

```
1 type Digit = Word8
2
3 bestScores :: ScoreTable -> [Run] -> [(Score, Digit)]
4 bestScores srl ps = take 3 . sort $ scores
5 where
6   scores = zip [distance d (scaleToOne ps) | d <- srl] digits
7   digits = [0 .. 9]
```

列表表达式

之前的例子中提到了一个新的概念，即列表表达式，根据一个或多个列表来创建一个列表。

```
1 ghci> [(a,b) | a <- [1,2], b <- "abc"]
2 [(1,'a'),(1,'b'),(1,'c'),(2,'a'),(2,'b'),(2,'c')]
```

在竖线左侧的表达式会在右侧生成器表达式的每个组合后计算。生成器表达式通过 `a <-` 以及右侧列表中的一个元素进行变量的绑定。上述例子中，生成器的组合子按照深度的顺序进行计算：首先是第一个列表的第一个元素，再计算第二个列表中的每个元素，以此类推。

另外关于生成器，我们还可以在列表表达式的右侧指定守护。一个守护就是一个 `Bool` 表达式，如果结果为 `False`，那么元素被跳过。

```
1 ghci> [(a,b) | a <- [1..6], b <- [5..7], even (a + b ^ 2)]
2 [(1,5),(1,7),(2,6),(3,5),(3,7),(4,6),(5,5),(5,7),(6,6)]
```

我们还可以用 `let` 表达式来绑定本地变量。

```
1 ghci> let vowel = ('elem' "aeiou")
2 ghci> [x | a <- "etaoin", b <- "shrdlu", let x = [a,b], all vowel x]
3 ["eu","au","ou","iu"]
```

如果在一个生成器表达式中模式匹配失败了，不会有异常出现。相反，列表的元素被跳过。

```
1 ghci> [a | (3,a) <- [(1, 'y'),(3, 'e'),(5, 'p')]]
2 "e"
```

列表表达式非常的强大已经简洁。然而它们不利于阅读。

记住匹配的奇偶

对于左侧组的每个匹配，我们需要记住其是否在偶数表格还是奇数表格。

```
1 data Parity a = Even a | Odd a | None a deriving (Show)
2
3 fromParity :: Parity a -> a
4 fromParity (Even a) = a
5 fromParity (Odd a) = a
6 fromParity (None a) = a
```

```

7
8 parityMap :: (a -> b) -> Parity a -> Parity b
9 parityMap f (Even a) = Even (f a)
10 parityMap f (Odd a) = Odd (f a)
11 parityMap f (None a) = None (f a)
12
13 instance Functor Parity where
14     fmap = parityMap

```

我们将一个编码好的值包装进 `parity` 内，并为其实现 `Functor` 实例，这样就可以轻松的操控 `parity` 内部的值了。

我们希望根据 `parity` 所包含的值进行排序。`Data.Function` 模块为此还提供了一个方便的组合子，`on`。

```

1 import Data.Function (on)
2
3 -- on :: (a -> a -> b) -> (c -> a) -> c -> c -> b
4 -- on f g x y = g x `f` g y
5
6 compareWithoutParity :: Parity (Score, Digit) -> Parity (Score, Digit) -> Ordering
7 compareWithoutParity = compare `on` fromParity

```

注：文件头引入了 `Data.Function` 的 `on`，另外显式标注了 `compareWithoutParity` 的函数签名。

为了避免理解不清楚，尝试将 `on` 想象为一个带有两个参数的函数，`f` 与 `g`，该函数返回另一个带两个参数的函数，`x` 与 `y`。然后将 `g` 分别应用至 `x` 与 `y`，再将 `f` 应用至上述两个结果。

将匹配包装进一个 `parity` 值很直接。

```

1 bestLeft :: [Run] -> [Parity (Score, Digit)]
2 bestLeft ps =
3     sortBy compareWithoutParity $
4         map Odd (bestScores leftOddSRL ps) ++ map Even (bestScores leftEvenSRL ps)
5
6 bestRight :: [Run] -> [Parity (Score, Digit)]
7 bestRight = map None . bestScores rightSRL

```

有了奇偶表格的最佳左侧匹配后，可以仅基于每个匹配的质量对它们进行排序

另一种惰性，键盘上的

`Parity` 类型可以使用 Haskell 的 record 语法来避免编写 `fromParity` 函数。换言之，可以这么写：

```

1 data AltParity a
2     = AltEven {fromAltParity :: a}
3     | AltOdd {fromAltParity :: a}

```

```

4 | AltNone {fromAltParity :: a}
5 deriving (Show)

```

那么为什么不这么做呢？答案有点令人羞愧，这与 **ghci** 中的交互式调试有关。当我们告诉 GHC 自动派生类型的 **Show** 实例时，它会根据是否使用记录语法声明该类型而生成不同的代码。

```

1 ghci> show $ Even 1
2 "Even 1"
3 ghci> show $ AltEven 1
4 "AltEven {fromAltParity = 1}"
5 ghci> length . show $ Even 1
6 6
7 ghci> length . show $ AltEven 1
8 27

```

Show 实例对于使用 record 语义而言显然更加冗长。

列表切分

处理列表的通常做法就是对它们进行切分。例如，条形码中的每个位数是由四个位数所编码的。我们可以将平整的列表表示为四个元素所构成的列表的列表。

```

1 chunkWith :: ([a] -> ([a], [a])) -> [a] -> [[a]]
2 chunkWith _ [] = []
3 chunkWith f xs = let (h, t) = f xs in h : chunkWith f t
4
5 chunksOf :: Int -> [a] -> [[a]]
6 chunksOf n = chunkWith $ splitAt n

```

生成候选位数列表

有了这些帮助函数后，编写一个用于生成每个数位组的候选匹配列表函数就很简单了。首先需要提前检查匹配偶数是否合理。所有的运行都需要以黑（**Zero**）条码开始，并包含足够的条码。

```

1 candidateDigits :: RunLength Bit -> [[Parity Digit]]
2 candidateDigits ((_, One) : _) = []
3 candidateDigits rle | length rle < 59 = []

```

对于 **bestLeft** 或 **bestRight** 返回的空列表，并不能被匹配。否则返回一个由校验编码的候选数字列表组成的列表。外部列表由十二个元素构成，条形码的每个数位有一个元素。每个子列表中的数字按匹配质量排序。

```

1 candidateDigits rle
2 | any null match = []
3 | otherwise = map (map (fmap snd)) match

```

```

4   where
5       match = map bestLeft left ++ map bestRight right
6       left  = chunksOf 4 . take 24 . drop 3 $ runLengths
7       right = chunksOf 4 . take 24 . drop 32 $ runLengths
8       runLengths = map fst rle

```

测试：

```

1  ghci> let input = zip (runLengths $ encodeEAN13 "978013211467") (cycle [Zero, One])
2  ghci> :t input
3  input :: [(Run, Bit)]
4  ghci> take 7 input
5  [(1,Zero),(1,One),(1,Zero),(1,One),(3,Zero),(1,One),(2,Zero)]
6  ghci> mapM_ print $ candidateDigits input
7  [Odd 7,Even 1,Even 2,Odd 3,Even 4,Odd 8]
8  [Even 8,Odd 0,Odd 1,Odd 2,Even 6,Even 7]
9  [Even 0,Even 1,Odd 2,Odd 4,Odd 6,Even 9]
10 [Odd 1,Odd 0,Even 1,Odd 2,Even 2,Even 4]
11 [Even 3,Odd 4,Odd 5,Even 7,Even 0,Odd 1]
12 [Odd 1,Even 1,Odd 2,Even 2,Odd 0,Even 0]
13 [None 1,None 0,None 2]
14 [None 1,None 0,None 2]
15 [None 4,None 2,None 5]
16 [None 6,None 8,None 2]
17 [None 7,None 3,None 8]
18 [None 7,None 3,None 8]

```

注：原文缺失：`let input = ...`。

没有数组或哈希表的世界

在命令式语言中，数组就跟 Haskell 的列表和元组那样，作为核心。在命令式语言中数组通常是可变的。

而 Haskell 中的数组是不可变的。这就意味着“修改”数组中的一个元素，整个数组会被拷贝，同时将那个元素设为新值。很明显，这样性能很差。

可变数组是由另一个常用的命令式数据结构所构建的，即哈希表。经典的实现中，数组像是在“旋转”表，每个元素包含了一个列表。为哈希表添加一个元素时，将哈希元素来找到列表的偏移量，并在该处修改列表并添加元素。

如果数组不是可变的，更新一个哈希表则需要创建一个新的。拷贝数组，将一个新的列表放在偏移处。我们无需拷贝其它偏移处的列表，但是我们已经拷贝了整个哈希表导致了更差的性能。

不可变数组从我们的工具箱中消除了两个规范的命令式数据结构。相较于很多其它语言，在纯 Haskell 中数组显得没有这么有用。不过很多数组代码只会在构建时更新，然后以只读的方式使用它。

树作为解决方案

尽管情况看上去好像并没有那么不好。数组和哈希表通常作为由键作为索引的集合，而在 Haskell 中我们则使用树。

在 Haskell 实现一个简单的树类型非常的容易。除此之外，更有用的树类型也非常容易实现。自平衡结构，如红黑树，其平衡算法是出奇的难以被正确的使用。

Haskell 的代数数据类型组合，模式匹配，以及守护，将最复杂的平衡操作简化成了几行代码。

对于函数式程序员而言树的魅力则是其廉价的修改。我们不会打破不变的规则：树是不变的。当需要修改一个树时，则创建一个新树，这一过程可以将新旧版本的树中绝大多数的结构进行共享。例如一个树包含了 10,000 个节点，那么在添加或删除一个元素时，可以预期新旧版本共享了 9,985 个元素。换言之，每次更新需要改变元素的数量是取决于树的高度，或者树大小的对数。

Haskell 标准库提供了两种由平衡树所实现的集合类型：`Data.Map` 用于键值对，`Data.Set` 用于值集合。

Tip

关于性能

相较于哈希表，一个实现良好的存函数树数据结构将具有竞争力。因此不应该假设代码在处理树时会损失性能。

maps 的简介

`Data.Map` 模块提供了一个参数化类型 `Map k a`。尽管其内部是一个大小平衡的二元树，其实现并不对我们可见。

`Map` 的键是严格的 `strict`，但是值确实非严格的 `non-strict`。换言之，`map` 的结构总是维持更新状态，但是值只会在使用时才会被计算。

记住这点很重要，因为 `Map` 在于值的惰性是导致空间泄漏的常见来源。

由于 `Data.Map` 模块包含了一系列与 `Prelude` 重名的名称，通常需要通过 `qualified` 的方式进行引入。

类型约束

`Map` 类型再起键类型上并没有任何显式的约束，但是模块中大多数有用的函数都需要键是 `Ord` 的实例。这是值得注意的，因为这是 Haskell 代码中常见的设计模式：类型约束只会在真实需要时才进行声明。

`Map` 类型和模块中的任何函数都不会约束用作值的类型。

部分应用程序的笨拙

由于某些原因，`Data.Map` 中的函数类型签名对部分应用 `partial application` 而言并不友好。`map` 的参数总是最后出现，毕竟先出现对于部分应用会更为友好。因此，使用部分应用的映射函数的代码几乎总是包含适配器函数来调整参数顺序。

从 API 开始

`Data.Map` 模块有一个巨大的“表层面”：即导出了大量的函数。

使用 `empty` 可以创建一个空的 `map`。使用 `singleton` 可以创建只包含一对键值的 `map`。

```
1 ghci> import qualified Data.Map as M
2 ghci> M.empty
3 fromList []
4 ghci> M.singleton "foo" True
5 fromList [("foo",True)]
```

由于实现是抽象的，我们无法对 `Map` 值进行模式匹配。相反，它提供了一系列的查询函数，其中有两个是被广泛使用的。`lookup` 函数的函数签名会稍微比较严格，不过无需担心，我们会在第十四章中进行详细的说明。

```
1 ghci> :t M.lookup
2 M.lookup :: Ord k => k -> M.Map k a -> Maybe a
```

通常而言，类型签名 `m` 会是 `Maybe`。换言之，如果 `map` 包含给定键的值，那么 `lookup` 则会返回被 `Just` 所包含的值；否则返回 `Nothing`。

```
1 ghci> let m = M.singleton "foo" 1 :: M.Map String Int
2 ghci> case M.lookup "bar" m of {Just v -> "yay"; Nothing -> "boo"}
3 "boo"
```

`findWithDefault` 函数接受一个值，在键不存在时作为返回。

Warning

小心偏函数！

存在一个 `(!)` 操作符，它执行查找并返回与键相关的未修饰值（即不包装在 `Maybe` 或其它）。不幸的是，它并不是一个完整的函数：它在键不存在时调用 `error`。

要添加一对键值至 `map`，最有用的函数就是 `insert` 与 `insertWith` 了。前者插入一个值，且覆盖可能存在的值。注：原文中的 `insertWith` 已废弃，现直接使用 `insertWith` 即可。

```
1 ghci> :t M.insert
2 M.insert :: Ord k => k -> a -> M.Map k a -> M.Map k a
```

```

3 ghci> M.insert "quux" 10 m
4 fromList [("foo",1),("quux",10)]
5 ghci> M.insert "foo" 9999 m
6 fromList [("foo",9999)]

```

insertWith 函数接受一个合并函数 *combining function* 作为其参数。如果没有匹配的键，则直接插入该值；否则，合并函数被调用用于新旧值，并将函数返回插入 map 中。

```

1 ghci> :t M.insertWith
2 M.insertWith
3   :: Ord k => (a -> a -> a) -> k -> a -> M.Map k a -> M.Map k a
4 ghci> M.insertWith (+) "zippity" 10 m
5 fromList [("foo",1),("zippity",10)]
6 ghci> M.insertWith (+) "foo" 9999 m
7 fromList [("foo",10000)]

```

delete 函数删除 map 中给定的键值。如果键不存在，那么 map 不变。

```

1 ghci> :t M.delete
2 M.delete :: Ord k => k -> M.Map k a -> M.Map k a
3 ghci> M.delete "foo" m
4 fromList []

```

最后还有若干高效的函数类似集合类型的操作作用于 map。例如下面的 **union**，该函数是“偏左值的”：如果两个 map 包含同样的键，那么返回的结果中则是包含左 map 的值。

```

1 ghci> m `M.union` M.singleton "quux" 1
2 fromList [("foo",1),("quux",1)]
3 ghci> m `M.union` M.singleton "foo" 0
4 fromList [("foo",1)]

```

以上我们仅覆盖大约百分之十的 **Data.Map** 的 API。我们将会在下一章进一步讲解 map 的更多细节。

将位数变为答案

现在还需要解决另有一个问题。有了很多条形码最后 12 位数字的候选码。此外还需要使用前六位数的奇偶来计算第一位数，最后需要确保答案符合校验码。

这看起来很有挑战性！我们仍有大量的未确定数据；那么我们该怎么做呢？通过 **ghci** 可以知道有多少组合需要测试。

```

1 ghci> product . map length . candidateDigits $ input
2 34012224

```

注： **input** 在上文有交代过。

这个想法到此为止。同样，我们将先关注子问题，再去考虑后续。

并行的方式解决校验位数

校验条形码的数组可以假设十分之一的数。对于一个给定的奇偶数，什么样的输入序列会导致该位数被计算？

```
1 type Map a = M.Map Digit [a]
```

该 map 的键是一个校验数，值则是该校验书所计算的序列值。另外两个基于该 map 定义的 map：

```
1 type DigitMap = Map Digit
2
3 type ParityMap = Map (Parity Digit)
```

我们通常将它们称为“解映射”，因为它们展示了“解出”每个校验数的数字序列。

```
1 updateMap ::
2   Parity Digit -> -- new digit
3   Digit -> -- existing key
4   [Parity Digit] -> -- existing digit sequence
5   ParityMap -> -- map to update
6   ParityMap
7   updateMap digit key seq = insertMap key (fromParity digit) (digit : seq)
8
9   insertMap :: Digit -> Digit -> [a] -> Map a -> Map a
10  insertMap key digit val m = val `seq` M.insert key' val m
11  where
12    key' = (key + digit) `mod` 10
```

通过从 map 中得到的现有校验数，求解该校验数的序列和一个新的输入数字，该函数用指向新校验数的新序列更新 map。

这可能看起来难以理解，不过使用一个例子就很清楚了。假设需要找的校验数是 4，其序列为 [1,3]，同时需要添加至该 map 的数是 8。那么 4 与 8 之和，取模 10，即为 2，也是需要插入至 map 的键。那么新校验数 2 所携带的序列就是 [8,1,3]，即所需插入至 map 的值。

对于一个序列中的每个位数，将使用位数与旧的解映射来生成新的解映射。

```
1 useDigit :: ParityMap -> ParityMap -> Parity Digit -> ParityMap
2 useDigit old new digit = new `M.union` M.foldrWithKey (updateMap digit) M.empty old
```

注：原文中的 `M.foldWithKey` 已废弃，先用 `M.foldrWithKey` 替代。

通过一些例子来测试一下：

```
1 ghci> let single n = M.singleton n [Even n] :: ParityMap
2 ghci> useDigit (single 1) M.empty (Even 1)
3 fromList [(2,[Even 1,Even 1])]
4 ghci> useDigit (single 1) (single 2) (Even 2)
5 fromList [(2,[Even 2]),(3,[Even 2,Even 1])]
```

我们提供给 `useDigits` 新的解映射为空，通过 `useDigits` fold 输入序列来进行求解。

```
1 incorporateDigits :: ParityMap -> [Parity Digit] -> ParityMap
2 incorporateDigits old = foldl' (useDigit old) M.empty
```

从旧的解映射生成一个新的：

```
1 ghci> incorporateDigits (M.singleton 0 []) [Even 1, Even 5]
2 fromList [(1,[Even 1]),(5,[Even 5])]
```

最后是构建一个完整的解映射。从空 map 开始，fold 每个位数的位置。在每个位置，创建一个新的 map。该 map 在下一轮的 fold 中又变成了旧的 map。

```
1 finalDigits :: [[Parity Digit]] -> ParityMap
2 finalDigits = foldl' incorporateDigits (M.singleton 0 []) . mapEveryOther (map $ fmap (* 3))
```

那么调用 `finalDigits` 所需的列表有多长呢？我们暂且不知道序列的第一个位数，显然还不能提供它。同时我们并不想在校验位数是包含猜测。因此该列表必须有十一位元素那么长。

一旦从 `finalDigits` 返回答案，解映射又是不完整的，因为我们仍未知第一位数是什么。

通过第一位数完成求解 map

我们仍未讨论如何从左侧组的奇偶性中提取第一个位数。实际上只需要复用之前写过的代码即可。

```
1 firstDigit :: [Parity a] -> Digit
2 firstDigit =
3     snd
4     . head
5     . bestScores paritySRL
6     . runLengths
7     . map parityBit
8     . take 6
9     where
10        parityBit (Even _) = Zero
11        parityBit (Odd _) = One
```

奇偶解映射中的每个元素现在包含了一个倒序的数值列表以及奇偶性数据。下一个任务就是创建一个完整的解映射，通过计算每个序列的第一个位数，再使用它创建最后一个解映射。

```
1 addFirstDigit :: ParityMap -> DigitMap
2 addFirstDigit = M.foldrWithKey updateFirst M.empty
3
4 updateFirst :: Digit -> [Parity Digit] -> DigitMap -> DigitMap
5 updateFirst key seq = insertMap key digit $ digit : renormalize qes
6     where
7         renormalize = mapEveryOther (`div` 3) . map fromParity
8         digit = firstDigit qes
9         qes = reverse seq
```

注：与前文相同，这里使用的是 `M.foldrWithKey`。

通过这样的方式，去除了 `Parity` 类型，同时恢复了之前乘上的 `3`。最后一步就是完成校验码的计算。

```
1 buildMap :: [[Parity Digit]] -> DigitMap
2 buildMap =
3   M.mapKeys (10 -)
4     . addFirstDigit
5     . finalDigits
```

寻找正确的序列

现在有了所有可能的校验和以及对应序列的 `map`。剩下的就是猜测校验位，看看是否拥有与之对应的解映射。

```
1 import Data.Maybe (mapMaybe)
2
3 ...
4
5 solve :: [[Parity Digit]] -> [[Digit]]
6 solve [] = []
7 solve xs = mapMaybe (addCheckDigit m) checkDigits
8   where
9     checkDigits = map fromParity $ last xs
10    m = buildMap $ init xs
11    addCheckDigit m k = (++) [k] <$> M.lookup k m
```

注：原文的 `catMaybes $ map (addCheckDigit m) checkDigits` 被 `mapMaybe` 简化。
测试：

```
1 ghci> listToMaybe . solve . candidateDigits $ input
2 Just [7,2,0,0,1,3,1,1,1,4,6,7,3]
```

处理行数据

反复的提到从图像中取出一行：

```
1 withRow :: Int -> Pixmap -> (RunLength Bit -> a) -> a
2 withRow n greymap f = f . runLength . elems $ posterized
3   where
4     posterized = threshold 0.4 . fmap luminance . row n $ greymap
```

`withRow` 函数接受一行，转换成黑白图像，接着调用另一个函数在 `run length` 编码的行数据上。通过调用 `row`，来获取行数据：

```
1 row :: (Ix a, Ix b) => b -> Array (a, b) c -> Array a c
2 row j a = ixmap (1, u) project a
```

```

3   where
4     project i = (i, j)
5     ((l, _), (u, _)) = bounds a

```

该函数需要详细讲解一下。`fmap` 函数用于转换数组中的值，而 `ixmap` 则用于转换数组中的索引。这是一个极为强大的函数，可以让我们随心所欲的堆数组进行“切片”。

`ixmap` 的第一个参数是新数组的边界。该边界可以是与源数组不同维度的。`row` 函数则是从一个二维数组中提取一个意为数组。

第二个参数则是投影函数，它从新数组中获取一个索引，并返回源数组的索引。该投影索引处的值将成为新数组中位于原始索引处的值。例如，传入 `2` 至投影函数，其返回 `(2,2)`，在新数组的索引 `2` 处的元素将被源数组的 `(2,2)` 处的元素所替代。

合并所有

除非是在条形码序列一开始就调用 `candidateDigits` 函数，否则返回一个空结果。我们可以轻松的扫描整个行，直到得到一个匹配的结果。

```

1   import Data.List (find)
2
3   ...
4
5   findMatch :: [(Run, Bit)] -> Maybe [[Digit]]
6   findMatch =
7     find (not . null)
8       . map (solve . candidateDigits)
9       . tails

```

注：原文的 `listToMaybe . filter` 可以被 `find` 简化。

这里利用了惰性求值。对 `tails` 调用 `map` 只会在其结果为非空列表时才会被计算。

接下来从图像中选择一行，并尝试从中找到条形码。

```

1   findEAN13 :: Pixmap -> Maybe [Digit]
2   findEAN13 pixmap = withRow center pixmap $ fmap head . findMatch
3   where
4     (_, (maxX, _)) = bounds pixmap
5     center = (maxX + 1) `div` 2

```

最后是一个简单的封装，用于从命令行中获取 `netpbm` 图像文件，并打印条形码

```

1   main :: IO ()
2   main = do
3     args <- getArgs
4     forM_ args $ \arg -> do
5       e <- parse parseRawPPM <$> L.readFile arg
6       case e of
7         Left err -> print $ "error: " ++ err
8         Right pixmap -> print $ findEAN13 pixmap

```

注意，本章我们已经定义了超过三十个函数，而只有 `main` 是唯一需要 `IO` 的。

一些开发风格的评论

略

13 Data structures

关联列表

通常我们需要处理由键索引的无序数据。例如 Unix 管理员会有一系列的数值 UIDs 以及其关联的用户名。该列表的价值在于能够查找给定 UID 的用户名，而不是按照数据的顺序。换言之，UID 是数据库的键。

在 Haskell 中，有若干方法可以处理类似这样数据的结构。两个最常见的就是关联列表以及由 `Data.Map` 模块提供的 `Map` 类型。关联列表很方便，因为它们很简单。它们就是 Haskell 的列表，因此所有熟悉的列表函数同样可用于关联列表。然而对于较大的数据集，`Map` 则比关联列表具有更大的性能优势。

关联列表就是一个包含了（键，值）元组的普通列表。对于 UID 映射至用户名的类型就是 `[(Integer, String)]`。

Haskell 拥有一个称为 `Data.List.lookup` 的内建函数用于关联列表的查询。其类型为 `Eq a => a -> [(a, b)] -> Maybe b`。

```
1 ghci> let al = [(1, "one"), (2, "two"), (3, "three"), (4, "four")]
2 ghci> lookup 1 al
3 Just "one"
4 ghci> lookup 5 al
5 Nothing
```

`lookup` 函数实际很简单，我们可以自己编写一个：

```
1 myLookup :: (Eq a) => a -> [(a, b)] -> Maybe b
2 myLookup _ [] = Nothing
3 myLookup key ((thisKey, thisVal) : rest) =
4   if key == thisKey
5     then Just thisVal
6     else myLookup key rest
```

让我们来看一个更复杂的关联列表。在 Unix/Linux 机器中存在一个名为 `/etc/passwd` 的文件用于存储用户名，UIDs，home 路径，已经其它数据。我们将编写一个程序用于解析该文件，创建一个关联列表，令用户通过给定的 UID 来查找用户名。

```
1 import Control.Monad (when)
2 import Data.List
3 import System.Environment (getArgs)
4 import System.Exit
5 import System.IO
6
7 main = do
8   -- Load the command-line arguments
9   args <- getArgs
10
```



```

11  -- If we don't have the right amount of args, give an error and abort
12  when (length args /= 2) $ do
13      putStrLn "Syntax: passwd-al filename uid"
14      exitFailure
15
16  -- Read the file lazily
17  content <- readFile $ head args
18
19  -- Compute the username in pure code
20  let username = findByUID content $ read $ args !! 1
21
22  -- Display the result
23  case username of
24      Just x -> putStrLn x
25      Nothing -> putStrLn "Could not find that UID"
26
27      putStrLn "whatever"
28
29  -- Given the entire input and a UID, see if we can find a username.
30  findByUID :: String -> Integer -> Maybe String
31  findByUID content uid =
32      let al = map parseLine . filter ((' #' /=) . head) . lines $ content
33      in lookup uid al
34
35  -- Convert a colon-separated line into fields
36  parseLine :: String -> (Integer, String)
37  parseLine input =
38      let fields = split ':' input
39      in (read (fields !! 2), head fields)
40
41  -- Takes a delimiter and a list. Break up the list based on the delimiter.
42  split :: (Eq a) => a -> [a] -> [[a]]
43  -- If the input is empty, the result is a list of empty lists.
44  split _ [] = [[]]
45  split delim str =
46      -- Find the part of the list before delim and put it in "before".
47      -- The rest of the list, including the leading delim, goes in "remainder".
48      let (before, remainder) = span (/= delim) str
49      in before : case remainder of
50          [] -> []
51          -- If there is more data to process,
52          -- call split recursively to process it
53          x ->
54              split delim $ tail x

```

注：原文的 `findByUID` 函数中的 `let al = map parseLine . lines $ content` 需要修改成 `let al = map parseLine . filter (('##' /=) . head) . lines $ content` 才能对 `/etc/passwd` 文件生效。否则会抛出异常 `Prelude.!!: index too large`。

测试：

```
1 % runhaskell passwd-al.hs /etc/passwd 0
2 root
3 % runhaskell passwd-al.hs /etc/passwd 3
4 Could not find that UID
```

映射

`Data.Map` 模块提供了一个与关联列表行为相似的 `Map` 类型，不过有更优异的性能。

映射提供了其它语言中与哈希表一样功能。其内部的实现为一个平衡二叉树。相较于哈希表，在不可变数据上的表现而言，映射更加的高效。这也是纯函数式编程如何深刻影响我们编写代码的最明显的例子：我们选择的数据结构和算法可以清晰的表达并且高效的执行，但是我们对特定任务的选择通常与命令式语言中的对应选项不同。

`Data.Map` 中有些函数与 Prelude 同名，因此需要 `import qualified Data.Map as Map` 并使用 `Map.name` 来引用模块中对应的名称。

```
1 import Data.Map qualified as Map
2
3 -- Functions to generate a Map that represents an association list as a map
4
5 al = [(1, "one"), (2, "two"), (3, "three"), (4, "four")]
6
7 {-
8 Create a map representation of 'al' by converting the association list using Map.fromList
9 -}
10 mapFromAL = Map.fromList al
11
12 {-
13 Create a map representation of 'al' by doing a fold
14 -}
15 mapFold = foldl (\map (k, v) -> Map.insert k v map) Map.empty al
16
17 {-
18 Manually create a map with the elements of 'al' in it
19 -}
20 mapManual =
21   Map.insert 2 "two"
22     . Map.insert 4 "four"
23     . Map.insert 1 "one"
24     . Map.insert 3 "three"
25   $ Map.empty
```

类似 `Map.insert` 的函数通常在 Haskell 中这么工作：它们返回输入数据的拷贝，并应用所需的修改。这对于映射而言很轻松。这意味着可以使用 `foldl` 来构建一个如 `mapFold` 案

例中那样的映射。或者是调用 `Map.insert` 串联在一起如 `mapManual` 案例中那样。使用 `ghci` 来查看是否如同预期：

```
1 ghci> :l buildmap.hs
2 [1 of 2] Compiling Main             ( buildmap.hs, interpreted )
3 Ok, one module loaded.
4 ghci> mapFromAL
5 fromList [(1,"one"),(2,"two"),(3,"three"),(4,"four")]
6 ghci> mapFold
7 fromList [(1,"one"),(2,"two"),(3,"three"),(4,"four")]
8 ghci> mapManual
9 fromList [(1,"one"),(2,"two"),(3,"three"),(4,"four")]
```

注意 `mapManual` 的输出与之前使用列表来构建映射的输出是不一样的。映射并不确保原始的顺序。

函数也是数据

Haskell 一部分的能力就是创建于操作函数非常的容易。下面是一个将函数存储为 `record` 字段的例子：

```
1 data CustomColor = CustomColor {red :: Int, green :: Int, blue :: Int}
2   deriving (Eq, Show, Read)
3
4 data FuncRec = FuncRec {name :: String, colorCalc :: Int -> (CustomColor, Int)}
5
6 plus5func color x = (color, x + 5)
7
8 purple = CustomColor 255 0 255
9
10 plus5 = FuncRec {name = "plus5", colorCalc = plus5func purple}
11
12 always0 = FuncRec {name = "always0", colorCalc = const (purple, 0)}
```

注意 `colorCalc` 的类型：它是一个函数，接受一个 `Int` 并返回 `(CustomColor, Int)` 的元组。我们创建了两个 `FuncRec` 的 records: `plus5` 与 `always0`。注意两者的 `colorCalc` 总是返回紫色。`FuncRec` 其本身并没有字段用于存储颜色，而是以某种方式使值成为了函数本身。这就是闭包 *closure*。

```
1 ghci> :l funcsecs.hs
2 [1 of 2] Compiling Main             ( funcsecs.hs, interpreted )
3 Ok, one module loaded.
4 ghci> :t plus5
5 plus5 :: FuncRec
6 ghci> name plus5
7 "plus5"
8 ghci> :t colorCalc plus5
9 colorCalc plus5 :: Int -> (CustomColor, Int)
```

```

10 ghci> (colorCalc plus5) 7
11 (CustomColor {red = 255, green = 0, blue = 255},12)
12 ghci> :t colorCalc always0
13 colorCalc always0 :: Int -> (CustomColor, Int)
14 ghci> (colorCalc always0) 7
15 (CustomColor {red = 255, green = 0, blue = 255},0)

```

更高级的方式，例如将数据用于若干地方，使用类型构造函数则会大有帮助：

```

1 data FuncRec = FuncRec
2   { name :: String,
3     calc :: Int -> Int,
4     namedCalc :: Int -> (String, Int)
5   }
6
7 mkFuncRec :: String -> (Int -> Int) -> FuncRec
8 mkFuncRec name calcfunc =
9   FuncRec
10   { name = name,
11     calc = calcfunc,
12     namedCalc = \x -> (name, calcfunc x)
13   }
14
15 plus5 = mkFuncRec "plus5" (+ 5)
16
17 always0 = mkFuncRec "always0" (const 0)

```

这里有一个名为 `mkFuncRec` 的函数，其接受一个 `String` 以及另一个函数作为参数，返回一个新的 `FuncRec` record。注意 `mkFuncRec` 的两个参数在若干地方都被用到了。

```

1 ghci> :l funcsecs2.hs
2 [1 of 2] Compiling Main             ( funcsecs2.hs, interpreted )
3 Ok, one module loaded.
4 ghci> :t plus5
5 plus5 :: FuncRec
6 ghci> name plus5
7 "plus5"
8 ghci> (calc plus5) 5
9 10
10 ghci> (namedCalc plus5) 5
11 ("plus5",10)
12 ghci> let plus5a = plus5 {name = "PLUS5A"}
13 ghci> name plus5a
14 "PLUS5A"
15 ghci> (namedCalc plus5a) 5
16 ("plus5",10)

```

注意 `plus5a` 的创建。我们修改了 `name` 字段而不是 `namedCalc` 字段。这是为什么 `name` 拥有一个新的名称，而 `namedCalc` 仍然返回的是传入 `mkFuncRec` 的名称；它不会改变，除非我们显式的修改它。

拓展案例: /etc/passwd

以下案例从 `/etc/passwd` 文件中解析并存储数据。

```

1  import Control.Monad (when)
2  import Data.List
3  import Data.Map qualified as Map
4  import System.Environment (getArgs)
5  import System.Exit
6  import System.IO
7  import Text.Printf (printf)
8
9  {-
10 The primary piece of data this program will store.
11 It represents the fields in a POSIX /etc/passwd file.
12 -}
13 data PasswdEntry = PasswdEntry
14   { userName :: String,
15     password :: String,
16     uid :: Integer,
17     gid :: Integer,
18     gecos :: String,
19     homeDir :: String,
20     shell :: String
21   }
22   deriving (Eq, Ord)
23
24 {-
25 Define how we get data to a 'PasswdEntry'.
26 -}
27 instance Show PasswdEntry where
28   show pe =
29     printf
30       "%s:%s:%d:%d:%s:%s:%s"
31       (userName pe)
32       (password pe)
33       (uid pe)
34       (gid pe)
35       (gecos pe)
36       (homeDir pe)
37       (shell pe)
38
39 {-
40 Converting data back out of a 'PasswdEntry'.
41 -}
42 instance Read PasswdEntry where
43   readsPrec _ value =
44     case split ':' value of
45       [f1, f2, f3, f4, f5, f6, f7] ->

```

```

46     -- Generate a 'PasswdEntry' the shorthand way:
47     -- using the positional fields. We use 'read' to
48     -- convert the numeric fields to Integers.
49     [(PasswdEntry f1 f2 (read f3) (read f4) f5 f6 f7, [])]
50     x -> error $ "Invalid number of fields in input: " ++ show x
51 where
52     -- Takes a delimiter and a list.
53     -- Break up the list based on the delimiter.
54     split :: (Eq a) => a -> [a] -> [[a]]
55     -- If the input is empty, the result is a list of empty lists.
56     split _ [] = []
57     split delim str =
58         -- Find the part of the list before delim and put it in "before".
59         -- The rest of the list, including the leading delim,
60         -- goes in "remainder".
61         let (before, remainder) = span (/= delim) str
62         in before : case remainder of
63             [] -> []
64             x ->
65                 -- If there is more data to process,
66                 -- call split recursively to process it
67                 split delim (tail x)
68
69 -- Convenience aliases; we'll have two maps: one from UID to entries
70 -- and the other from username to entries
71 type UIDMap = Map.Map Integer PasswdEntry
72
73 type UserMap = Map.Map String PasswdEntry
74
75 {-
76 Converts input data to maps. Returns UID and User maps.
77 -}
78 inputToMaps :: String -> (UIDMap, UserMap)
79 inputToMaps inp = (uidmap, usermap)
80 where
81     -- fromList converts a [(key, value)] list into a Map
82     uidmap = Map.fromList . map (\pe -> (uid pe, pe)) $ entries
83     usermap = Map.fromList . map (\pe -> (userName pe, pe)) $ entries
84     -- Convert the input String to [PasswdEntry]
85     entries = map read $ lines inp
86
87 mainMenu maps@(uidmap, usermap) = do
88     putStr optionText
89     hFlush stdout
90     sel <- getLine
91     -- See what they want to do. For every option except 4,
92     -- return them to the main menu afterwards by calling
93     -- mainMenu recursively
94     case sel of

```

```

95     "1" -> lookupUserName >> mainMenu maps
96     "2" -> lookupUID >> mainMenu maps
97     "3" -> displayFile >> mainMenu maps
98     "4" -> return ()
99     _ -> putStrLn "Invalid selection" >> mainMenu maps
100 where
101     lookupUserName = do
102         putStrLn "Username: "
103         username <- getLine
104         case Map.lookup username usermap of
105             Nothing -> putStrLn "Not found."
106             Just x -> print x
107     lookupUID = do
108         putStrLn "UID: "
109         uidstring <- getLine
110         case Map.lookup (read uidstring) uidmap of
111             Nothing -> putStrLn "Not found."
112             Just x -> print x
113     displayFile =
114         putStr . unlines . map (show . snd) . Map.toList $ uidmap
115     optionText =
116         "\npasswdmap options:\n\
117         \\n\
118         \1   Look up a user name\n\
119         \2   Look up a UID\n\
120         \3   Display entire file\n\
121         \4   Quit\n\n\
122         \Your selection: "
123
124 main = do
125     -- Load the command-line arguments
126     args <- getArgs
127
128     -- If we don't have the right number of args,
129     -- give an error and abort
130
131     when (length args /= 1) $ do
132         putStrLn "Syntax: passwdmap filename"
133         exitFailure
134
135     -- Read the file lazily
136     content <- readFile $ head args
137     let maps = inputToMaps content
138     mainMenu maps

```

该案例维护了两个 maps: username 与 `PasswdEntry` , 以及 UID 与 `PasswdEntry` 。可以将其视为数据库的两个不同的数据索引, 用于加速搜索不同的字段。

拓展案列：数值类型

略

第一步

如果希望为加号操作符实现一些自定义行为，则必须定义一个 `newtype`，并使其成为 `Num` 的一个实例。该类型需要以符号方式存储表达式。为此我们需要存储操作本身，其左侧以及右侧。其中，左右侧也可以是表达式。

```

1  -- The "operators" that we're going to support
2  data Op = Plus | Minus | Mul | Div | Pow
3          deriving (Eq, Show)
4
5  {- The core symbolic manipulation type -}
6  data SymbolicManip a
7      = Number a -- Simple number, such as 5
8      | Arith Op (SymbolicManip a) (SymbolicManip a)
9      deriving (Eq, Show)
10
11 {-
12   SymbolicManip will be an instance of Num.
13   Define how the Num operations are handled over a SymbolicManip.
14   This will implement things like (+) for SymbolicManip.
15 -}
16 instance (Num a) => Num (SymbolicManip a) where
17     a + b = Arith Plus a b
18     a - b = Arith Minus a b
19     a * b = Arith Mul a b
20     negate = Arith Mul (Number (-1))
21     abs a = error "abs is unimplemented"
22     signum _ = error "signum is unimplemented"
23     fromInteger i = Number (fromInteger i)

```

这里定义了名为 `Op` 的类型。该类型代表了某些的操作符。接着是 `SymbolicManip a` 的定义，由于有 `Num a` 约束，任何 `Num` 都可用作于 `a`。一个完整类型会像 `SymbolicManip Int` 这样。

一个 `SymbolicManip` 类型可以是一个简单的数字，或者是某些计算操作。`Arith` 构造函数的类型是递归的，这在 Haskell 中很合理。`Arith` 用一个 `Op` 和另外两个 `SymbolicManip` 创建了一个新的 `SymbolicManip`。

```

1  ghci> :l numsimple.hs
2  [1 of 2] Compiling Main                ( numsimple.hs, interpreted )
3  Ok, one module loaded.
4  ghci> Number 5
5  Number 5
6  ghci> :t Number 5

```



```

7  Number 5 :: Num a => SymbolicManip a
8  ghci> :t Number (5::Int)
9  Number (5::Int) :: SymbolicManip Int
10 ghci> Number 5 * Number 10
11 Arith Mul (Number 5) (Number 10)
12 ghci> (5 * 10)::SymbolicManip Int
13 Arith Mul (Number 5) (Number 10)
14 ghci> (5 * 10 + 2)::SymbolicManip Int
15 Arith Plus (Arith Mul (Number 5) (Number 10)) (Number 2)

```

可以看到非常基础的表达式是如何工作的。注意 Haskell 是如何“转换” `5 * 10 + 2` 成一个 `SymbolicManip` 的，以及其中的处理顺序。这并非一个真实的转换；`SymbolicManip` 如今是一个 first-class 数值。整数数值的字面量会被视为包裹在 `fromInteger` 内，因此 `5` 等同于 `SymbolicManip Int` 等同于 `Int`。

至此我们的任务就简单了：拓展 `SymbolicManip` 类型来表达所有需求的操作，为其实现其他数值 typeclasses 的实例，并实现自身的 `Show`。

完整代码

以下是完整的 `num.hs` 代码：

```

1  import Data.List
2
3  -- Symbolic/units manipulation
4  data Op
5      = Plus
6      | Minus
7      | Mul
8      | Div
9      | Pow
10     deriving (Eq, Show)
11
12 {-
13     The core symbolic manipulation type.
14     It can be a simple number, a symbol, a binary arithmetic operation (such as +),
15     or a unary arithmetic operation (such as cos)
16
17     Notice the types of BinaryArith and UnaryArith: it's a recursive type.
18     So, we could represent a (+) over two SymbolicManips.
19 -}
20 data SymbolicManips a
21     = Number a -- Simple number, such as 5
22     | Symbol String -- A symbol, such as x
23     | BinaryArith Op (SymbolicManips a) (SymbolicManips a)
24     | UnaryArith String (SymbolicManips a)
25     deriving (Eq)

```

上述代码定义了之前使用过的 `Op`，以及 `SymbolicManips`，与之前类似。这个版本支持了单元算法符（即仅接受一个参数）例如 `abs` 以及 `cos`。接下来定义 `Num`。

```

1 {-
2   SymbolicManips will be an instance of Num.
3   Define how the Num operations are handled over a SymbolicManips.
4   This will implement things like (+) for SymbolicManips.
5 -}
6 instance (Num a) => Num (SymbolicManips a) where
7   a + b = BinaryArith Plus a b
8   a - b = BinaryArith Minus a b
9   a * b = BinaryArith Mul a b
10  negate = BinaryArith Mul (Number (-1))
11  abs = UnaryArith "abs"
12  signum _ = error "signum is unimplemented"
13  fromInteger = Number . fromInteger

```

以上代码简单直接。注意之前的版本并不支持 `abs`，不过现在有了 `UnaryArith` 构造函数就能支持了。接下来是更多的实例：

```

1 {-
2   Make SymbolicManips an instance of Fractional
3 -}
4 instance (Fractional a) => Fractional (SymbolicManips a) where
5   a / b = BinaryArith Div a b
6   recip = BinaryArith Div $ Number 1
7   fromRational = Number . fromRational
8
9 {-
10  Make SymbolicManips an instance of Floating
11 -}
12 instance (Floating a) => Floating (SymbolicManips a) where
13   pi = Symbol "pi"
14   exp = UnaryArith "exp"
15   log = UnaryArith "log"
16   sqrt = UnaryArith "sqrt"
17   a ** b = BinaryArith Pow a b
18   sin = UnaryArith "sin"
19   cos = UnaryArith "cos"
20   tan = UnaryArith "tan"
21   asin = UnaryArith "asin"
22   acos = UnaryArith "acos"
23   atan = UnaryArith "atan"
24   sinh = UnaryArith "sinh"
25   cosh = UnaryArith "cosh"
26   tanh = UnaryArith "tanh"
27   asinh = UnaryArith "asinh"
28   acosh = UnaryArith "acosh"
29   atanh = UnaryArith "atanh"

```

以上代码实现的是 `Fractional` 与 `Floating` 。接下来是将表达式以字符串形式展示的实现：

```

1 {-
2   Show a SymbolicManips as a String, using conventional algebraic notation
3 -}
4 prettyShow :: (Show a, Num a) => SymbolicManips a -> String
5 -- Show a number or symbol as a bare number or serial
6 prettyShow (Number x) = show x
7 prettyShow (Symbol x) = x
8 prettyShow (BinaryArith op a b) =
9     let pa = simpleParen a
10        pb = simpleParen b
11        pop = op2str op
12    in pa ++ pop ++ pb
13 prettyShow (UnaryArith opstr a) =
14     opstr ++ "(" ++ prettyShow a ++ ")"
15
16 op2str :: Op -> String
17 op2str Plus = "+"
18 op2str Minus = "-"
19 op2str Mul = "*"
20 op2str Div = "/"
21 op2str Pow = "**"
22
23 {-
24   Add parenthesis where needed. This function is fairly conservative and will
25   add parenthesis when not needed in some cases.
26
27   Haskell will have already figured out precedence for us while building up
28   the SymbolicManips.
29 -}
30 simpleParen :: (Show a, Num a) => SymbolicManips a -> String
31 simpleParen (Number x) = prettyShow (Number x)
32 simpleParen (Symbol x) = prettyShow (Symbol x)
33 simpleParen x@(BinaryArith _ _ _) = "(" ++ prettyShow x ++ ")"
34 simpleParen x@(UnaryArith _ _) = prettyShow x
35
36 {-
37   Showing a SymbolicManips calls the prettyShow function on it
38 -}
39 instance (Show a, Num a) => Show (SymbolicManips a) where
40     show = prettyShow

```

首先是定义 `prettyShow` 函数，它用于渲染表达式。接下来实现将一个表达式转换为 RPN 格式字符串的算法：

```

1 {-
2   Show a SymbolicManip using RPN. HP calculator users may find this familiar.
3 -}

```

```

4  rnpShow :: (Show a, Num a) => SymbolicManips a -> String
5  rnpShow i =
6      let toList (Number x) = [show x]
7          toList (Symbol x) = [x]
8          toList (BinaryArith op a b) = toList a ++ toList b ++ [op2str op]
9          toList (UnaryArith op a) = toList a ++ [op]
10         join :: [a] -> [[a]] -> [a]
11         join = intercalate
12     in join " " (toList i)

```

接下来实现一个函数来对表达式进行一些基本的简化：

```

1  {-
2      Perform some basic algebraic simplifications on a SymbolicManip
3  -}
4  simplify :: (Num a, Eq a) => SymbolicManips a -> SymbolicManips a
5  simplify (BinaryArith op ia ib) =
6      let sa = simplify ia
7          sb = simplify ib
8      in case (op, sa, sb) of
9          (Mul, Number 1, b) -> b
10         (Mul, a, Number 1) -> a
11         (Mul, Number 0, _) -> Number 0
12         (Mul, _, Number 0) -> Number 0
13         (Div, a, Number 1) -> a
14         (Plus, a, Number 0) -> a
15         (Plus, Number 0, b) -> b
16         (Minus, a, Number 0) -> a
17         _ -> BinaryArith op sa sb
18  simplify (UnaryArith op a) = UnaryArith op (simplify a)
19  simplify x = x

```

现在需要为已构建好的库来增加单位测量。这将允许我们表达例如“5 米”这样的数量。首先是定义一个类型：

```

1  {-
2      New data type: Units.
3      A Units type contains a number and a SymbolicManip, which represents the units of measure.
4      A simple label would be something like (Symbol "m")
5  -}
6  data Units a = Units a (SymbolicManips a) deriving (Eq)

```

一个 `Units` 包含一个数值与一个标签。标签本身是一个 `SymbolicManip`。接下来就是 `Units` 的 `Num` 实例了：

```

1  {-
2      Implement Units for Num.
3      We don't know how to convert between arbitrary units, so we generate an error if we try to
4      add
5      numbers with different units. For multiplication, generate the appropriate new units.
6  -}

```

```

5  -}
6  instance (Num a, Eq a) => Num (Units a) where
7    (Units xa ua) + (Units xb ub)
8      | ua == ub = Units (xa + xb) ua
9      | otherwise = error "Mis-matched units in add or subtract"
10   (Units xa ua) - (Units xb ub) = Units xa ua + Units (xb * (-1)) ub
11   (Units xa ua) * (Units xb ub) = Units (xa * xb) (ua * ub)
12   negate (Units xa ua) = Units (negate xa) ua
13   abs (Units xa ua) = Units (abs xa) ua
14   signum (Units xa _) = Units (signum xa) (Number 1)
15   fromInteger i = Units (fromInteger i) (Number 1)

```

这里就能明白为什么使用 `SymbolicManip` 而不是 `String` 来存储测量单位了。例如在乘法计算出现时，测量的单位也同样发生了变化。例如 5 米乘以 2 米，那么得到的是 10 平方米。我们强制加法时的单位匹配，并由加法来实现减法。下面是更多的 `typeclass`：

```

1  {-
2    Make Units an instance of Fractional
3  -}
4  instance (Fractional a, Eq a) => Fractional (Units a) where
5    (Units xa ua) / (Units xb ub) = Units (xa / xb) (ua / ub)
6    recip a = 1 / a
7    fromRational r = Units (fromRational r) (Number 1)
8
9  {-
10   Floating implementation for Units.
11   Use some intelligence for angle calculations: support deg and rad
12 -}
13 instance (Floating a, Eq a) => Floating (Units a) where
14   pi = Units pi $ Number 1
15   exp = error "exp not yet implemented in Units"
16   log = error "log not yet implemented in Units"
17   sqrt (Units xa ua) = Units (sqrt xa) (sqrt ua)
18   (Units xa ua) ** (Units xb ub)
19     | ub == Number 1 = Units (xa ** xb) (ua ** Number xb)
20     | otherwise = error "units for RHS of ** not supported"
21   logBase _ = error "logBase not yet implemented in Units"
22   sin (Units xa ua)
23     | ua == Symbol "rad" = Units (sin xa) (Number 1)
24     | ua == Symbol "deg" = Units (sin (deg2rad xa)) (Number 1)
25     | otherwise = error "Units for sin must be empty"
26   cos (Units xa ua)
27     | ua == Symbol "rad" = Units (cos xa) (Number 1)
28     | ua == Symbol "deg" = Units (cos (deg2rad xa)) (Number 1)
29     | otherwise = error "Units for cos must be empty"
30   tan (Units xa ua)
31     | ua == Symbol "rad" = Units (tan xa) (Number 1)
32     | ua == Symbol "deg" = Units (tan (deg2rad xa)) (Number 1)
33     | otherwise = error "Units for tan must be empty"

```

```

34  asin (Units xa ua)
35      | ua == Number 1 = Units (rad2deg $ asin xa) (Symbol "deg")
36      | otherwise = error "Units for asin must be empty"
37  acos (Units xa ua)
38      | ua == Number 1 = Units (rad2deg $ acos xa) (Symbol "deg")
39      | otherwise = error "Units for acos must be empty"
40  atan (Units xa ua)
41      | ua == Number 1 = Units (rad2deg $ atan xa) (Symbol "deg")
42      | otherwise = error "Units for atan must be empty"
43  sinh = error "sinh not yet implemented in Units"
44  cosh = error "cosh not yet implemented in Units"
45  tanh = error "tanh not yet implemented in Units"
46  asinh = error "asinh not yet implemented in Units"
47  acosh = error "acosh not yet implemented in Units"
48  atanh = error "atanh not yet implemented in Units"

```

接下来是一些处理单位的公用函数：

```

1  {-
2    A simple function that takes a number and a String and returns an appropriate Units type to
3    represent the number and its unit of measure
4  -}
5  units :: (Num z) => z -> String -> Units z
6  units a b = Units a $ Symbol b
7
8  {-
9    Extract the number only out of a Units type
10 -}
11 dropUnits :: (Num z) => Units z -> z
12 dropUnits (Units x _) = x
13
14 {-
15   Utilities for the Unit implementation
16 -}
17 deg2rad x = 2 * pi * x / 360
18
19 rad2deg x = 360 * x / (2 * pi)

```

这里首先是 `units` 用于简化了表达式的构建。接着是 `dropUnits` 用于去掉单位返回裸值 `Num`。最后是定义了之前所需的角度与弧度之间的转换函数。接下来是 `Units` 的 `Show` 实例：

```

1  {-
2    Showing units: we show the numeric component, an underscore, then the prettyShow version of
3    the simplified units
4  -}
5  instance (Show a, Num a, Eq a) => Show (Units a) where
6      show (Units xa ua) = show xa ++ "_" ++ prettyShow (simplify ua)

```

最后是测试：

```

1 test :: (Num a) => a
2 test = 2 * 5 + 3

```

函数作为数据的优点

在命令式语言中，两个列表的追加廉价且容易。这里是一个简单的 C 结构，其中维护了两个列表的头尾指针：

```

1 struct list {
2     struct node *head, *tail;
3 };

```

当我们想将一个列表附加到另一个列表时，需要修改前一个列表的最后一个节点并指向后一个列表的 `head` 节点，接着更新尾指针指向尾结点。

显然在 Haskell 中，如果想要保持纯粹性，这种方法是不行的。因为数据是不可变的，所以不能随意修改列表。Haskell 用来附加列表的 `(++)` 操作符：

```

1 (++) :: [a] -> [a] -> [a]
2 (x : xs) ++ ys = x : xs ++ ys
3 _ ++ ys = ys

```

从代码中可知，创建一个新的列表取决于初始列表的长度。

如果是不断的附加，那么这个代价是巨大的。`(++)` 操作符是右关联的：

```

1 ghci> :info (++)
2 (++) :: [a] -> [a] -> [a]    -- Defined in GHC.Base
3 infixr 5 ++

```

这就意味着 Haskell 将表达式 `"a" ++ "b" ++ "c"` 转换为 `"a" ++ ("b" ++ "c")`。这样可以使性能达到最优，因为它令左侧的操作尽可能的简短。

当我们重复的对一个列表进行附加时，我们便破坏了这种结合性。

与此同时，命令式程序员却在呵呵笑，因为他们重复追加的成本只取决于他们执行追加的次数。它们的性能是线性的，而我们则是平方函数。

像这样重复对列表进行附加的常见操作会造成性能损失时，就需要从另一个角度来看待这个问题了。

表达式 `"a"++` 是一个偏应用函数。它的类型是什么？

```

1 ghci> :type ("a" ++)
2 ("a" ++) :: [Char] -> [Char]

```

由于这是一个函数，我们可以使用 `(.)` 操作符与其它操作进行组合，例如 `("b"++)`。

```

1 ghci> :type ("a" ++) . ("b" ++)
2 ("a" ++) . ("b" ++) :: [Char] -> [Char]

```

新的函数拥有同样的类型。那么当停止组合函数时，并将一个 `String` 提供给函数会得到什么呢？

```
1 ghci> let f = ("a" ++) . ("b" ++)
2 ghci> f []
3 "ab"
```

我们附加了字符串！我们用这些偏应用函数来存储数据，通过一个空列表来获取它。每个偏应用的 `(++)` 与 `(.)` 代表着一个附加，但是又不实际执行附加。

关于这个方法有两点非常有趣的地方。首先偏应用的成本是恒定的，因此很多偏应用的成本是线性的。其次当我们最终提供 `[]` 值来获取偏应用中的最终列表时，应用的顺序是从右至左的。这使得 `(++)` 的左侧操作最小，从而让附加操作的整体成本是线性而不是二次方程的。

将差异列表变为一个合理的库

第一步是使用 `newtype` 声明对用户隐藏基础类型。我们将创建一个名为 `DList` 的新类型。

```
1 newtype DList a = DL
2 { unDL :: [a] -> [a]
3 }
```

`unDL` 函数即解构函数，即移除 `DL` 构造函数。

```
1 append :: DList a -> DList a -> DList a
2 append xs ys = DL (unDL xs . unDL ys)
```

`append` 函数看起来有点复杂，但它实际上只是之前展示的 `(.)` 操作符的相同用法。对函数进行组合，首先需要从 `DL` 构造函数中解包，因此使用了 `unDL`。接着通过 `DL` 构造函数重新将结果包装起来。

下面是另一种写法，这里使用了模式匹配来解包 `xs` 与 `ys`：

```
1 append' :: DList a -> DList a -> DList a
2 append' (DL xs) (DL ys) = DL (xs . ys)
```

`DList` 类型如果不能与正常列表互相转换的话，那么用处也不会很大：

```
1 fromList :: [a] -> DList a
2 fromList xs = DL (xs ++)
3
4 toList :: DList a -> [a]
5 toList (DL xs) = xs []
```

如果我们还想要 `DList` 变得跟普通列表一样好用，那么还需要提供一些列表的常规操作。

```
1 empty :: DList a
2 empty = DL id
3
```



```

4  -- equivalent of the list type's (:) operator
5  cons :: a -> DList a -> DList a
6  cons x (DL xs) = DL $ (x :) . xs
7
8  dfoldr :: (a -> b -> b) -> b -> DList a -> b
9  dfoldr f z xs = foldr f z $ toList xs

```

尽管 `DList` 令列表附加的成本下降，但并不是所有的类列表操作都是可用的。`head` 函数的成本对于普通列表是常数，而 `DList` 需要将整个 `DList` 转换为正常列表才能进行，因此这使成本变得额外高昂。

```

1  safeHead :: DList a -> Maybe a
2  safeHead xs = case toList xs of
3    (y : _) -> Just y
4    _ -> Nothing

```

对于 `map` 而言，我们需要将 `DList` 类型成为一个函子。

```

1  dmap :: (a -> b) -> DList a -> DList b
2  dmap f = dfoldr go empty
3  where
4    go x = cons (f x)
5
6  instance Functor DList where
7    fmap = dmap

```

最终可在源文件头部添加模块头进行导出：

```

1  module DList
2  ( DList,
3    fromList,
4    toList,
5    empty,
6    append,
7    cons,
8    dfoldr,
9  )
10 where

```

列表，差异列表，以及幺半群

在抽象代数中，存在一个名为幺半群 *monoid* 的简单抽象结构。很多数学对象就是幺半群，它们必须有两个属性：

- 一个二维操作符。名为 `(*)`：表达式 `a * (b * c)` 的结果必须等同于 `(a * b) * c` 的结果。
- 一个 id 值。如果是 `e`，它必须遵循两个规则：`a * e == a` 以及 `e * a == a`。

具体实现:

```

1 instance Semigroup (DList a) where
2   (<>) = append
3
4 instance Monoid (DList a) where
5   mempty = empty

```

注: 与原文不同, `Monoid` 现在需要 `Semigroup` 作为约束。

通用序列

Haskell 内建的列表类型和我们所构建的 `DList` 类型在某些特定场景下的性能非常的差。而 `Data.Sequence` 模块中定义的 `Seq` 容器类型则是在大量的操作下具有优异的性能。

```

1 import Data.Sequence qualified as Seq
2
3 import Data.Sequence ((<|), (>|), (|>))
4
5 import Data.Foldable qualified as Foldable

```

示例:

```

1 ghci> :l DataSequence.hs
2 [1 of 2] Compiling Main             ( DataSequence.hs, interpreted )
3 Ok, one module loaded.
4 ghci> Seq.empty
5 fromList []
6 ghci> Seq.singleton 1
7 fromList [1]
8 ghci> let a = Seq.fromList [1,2,3]
9 ghci> Seq.singleton 1 |> 2
10 fromList [1,2]
11 ghci> let left = Seq.fromList [1,3,3]
12 ghci> let right = Seq.fromList [7,1]
13 ghci> left >< right
14 fromList [1,3,3,7,1]
15 ghci> Foldable.toList (Seq.fromList [1,2,3])
16 [1,2,3]
17 ghci> Foldable.foldl' (+) 0 (Seq.fromList [1,2,3])
18 6

```

14 Monads

简介

在第七章 I/O 中，我们讨论了 `IO` 单子，然而当时只局限于如何它与外界进行交互，并没有讨论什么是单子。

...

重构早期代码

Maybe 链

回忆一下第十章的 `PNM.hs` 代码中的 `parseP5` 函数：

```

1  -- header
2  matchHeader :: L.ByteString -> L.ByteString -> Maybe L.ByteString
3  matchHeader prefix str
4  | prefix `L8.isPrefixOf` str = Just (L8.dropWhile isSpace (L.drop (L.length prefix) str))
5  | otherwise = Nothing
6
7  -- nat = natural number
8  getNat :: L.ByteString -> Maybe (Int, L.ByteString)
9  getNat s = case L8.readInt s of
10     Nothing -> Nothing
11     Just (num, rest)
12       | num <= 0 -> Nothing
13       | otherwise -> Just (fromIntegral num, rest)
14
15  getBytes :: Int -> L.ByteString -> Maybe (L.ByteString, L.ByteString)
16  getBytes n str =
17     let count = fromIntegral n
18         both@(prefix, _) = L.splitAt count str
19     in if L.length prefix < count
20        then Nothing
21        else Just both
22
23  -- parse function
24  parseP5 :: L.ByteString -> Maybe (GreyMap, L.ByteString)
25  parseP5 s =
26     case matchHeader (L8.pack "PS") s of
27       Nothing -> Nothing
28       Just s1 ->
29         case getNat s1 of
30           Nothing -> Nothing
31           Just (width, s2) ->
32             case getNat (L8.dropWhile isSpace s2) of
33               Nothing -> Nothing

```

```

34         Just (height, s3) ->
35         case getNat (L8.dropWhile isSpace s3) of
36             Nothing -> Nothing
37             Just (maxGrey, s4)
38                 | maxGrey > 255 -> Nothing
39                 | otherwise ->
40                 case getBytes 1 s4 of
41                     Nothing -> Nothing
42                     Just (_, s5) ->
43                     case getBytes (width * height) s5 of
44                         Nothing -> Nothing
45                         Just (bitmap, s6) ->
46                         Just (GreyMap width height maxGrey bitmap, s6)

```

接着又使用了 `(>>?)` 函数来减少模式匹配:

```

1  (>>?) :: Maybe a -> (a -> Maybe b) -> Maybe b
2  Nothing >>? _ = Nothing
3  Just v >>? f = f v

```

...

隐式状态

简化后的代码如下:

```

1  parseP5_take2 :: L.ByteString -> Maybe (GreyMap, L.ByteString)
2  parseP5_take2 s =
3      matchHeader (L8.pack "P5") s
4      >>? \s ->
5          skipSpace ((), s)
6          >>? (getNat . snd)
7          >>? skipSpace
8          >>? \(width, s) ->
9              getNat s
10             >>? skipSpace
11             >>? \(height, s) ->
12                 getNat s
13                 >>? \(maxGrey, s) ->
14                     getBytes 1 s
15                     >>? (getBytes (width * height) . snd)
16                     >>? \(bitmap, s) -> Just (GreyMap width height maxGrey bitmap, s)

```

我们仍然遇到了重复的行为模式: 消费一些字符串, 返回一个结果, 接着将剩下的字符串传入下一个函数进行消费。然而这套模式有隐藏的风险: 如果希望将其余部分的信息在链中传递, 那么就需要将链中的每个元素都进行修改, 将每个二元元组改为三元元组!

我们解决这个问题的方法是将管理当前字符串的责任从链中的单个函数移动到将函数进行串联的函数中:

```

1  (==>) :: Parse a -> (a -> Parse b) -> Parse b
2  firstParser ==> secondParser = Parse chainedParser
3  where
4      chainedParser initState =
5          case runParse firstParser initState of
6              Left errorMessage ->
7                  Left errorMessage
8              Right (firstResult, newState) ->
9                  runParse (secondParser firstResult) newState

```

我们同样将解析状态的细节隐藏进 `ParseState` 类型中。即使 `getState` 与 `putState` 函数都不会检查解析状态，因此对 `ParseState` 的任何修改都不会影响现有的代码。

寻找共有模式

...

单子 typeclass

我们可以在 Haskell 的 typeclass 中捕获链接和注入的概念，同时又希望它们具有类型。标准的 Prelude 已经定义了这样的一个 typeclass，即 `Monad`。

```

1  class Monad m where
2      -- chain
3      (>>=) :: m a -> (a -> m b) -> m b
4      -- inject
5      return :: a -> m a

```

注：新的 Haskell 版本中 `Monad` 还需要先实现 `Applicative` 实例，而 `Applicative` 又需要先实现 `Functor` 实例。

这里的 `>>=` 是链接函数，而 `return` 则是注入函数。

...

行话时间

- “Monadic” 即为“与单子有关”。一个 monadic 类型就是 `Monad` typeclass 的一个实例；一个 monadic 值拥有一个 monadic 类型。
- 当我们说一个类型“是一个单子”，这实际上在简述它是 `Monad` typeclass 的一个实例。作为 `Monad` 的一个实例则拥有必要的 monadic 的类型构造函数，注入函数，以及链接函数。
- 同样的，引用“`Foo` 单子”意味着正在谈论名为 `Foo` 的类型，且它是 `Monad` 的一个实例。

- “动作 action”则是一个 monadic 值的别称。这个词的这种用法可能起源于用于 I/O 单子的引入，其中像 `print "foo"` 这样的单子值可能具有可观察到的副作用。具有 monadic 返回类型的函数也可以称为操作，尽管这种情况不太常见。

使用一个新的单子：展示你的成果！

在介绍单子时，我们展示了一些已经具备 monadic 形态的预定义代码。现在让我们定义其接口。

纯 Haskell 代码编写很简单，但是它不能执行 I/O。有时我们希望在记录一些决策时，不必将 log 信息写入文件中。现在来开发一个这样的库。

回忆在之前将 glob 模式转为正则表达式的小节中所开发的 `globToRegex` 函数。我们将修改它，使其保存它所翻译的每个特殊模式序列的记录。

作为开始，首先将返回类型通过 `Logger` 类型构造函数进行包装。

```
1 globToRegex :: String -> Logger String
```

隐藏信息

```
1 module Logger
2   ( Logger,
3     Log,
4     runLogger,
5     record,
6   )
7 where
```

像这样隐藏细节有两个好处：它为我们实现单子提供了相当大的灵活性，更重要的是，它为用户提供了一个简单的界面。

我们的 `Logger` 类型存粹是一个类型构造函数。没有导出用户创建这种类型的值所需的值构造函数，用户仅仅可以使用 `Logger` 来编写类型签名。

`Log` 类型即字符串列表的同义词，将若干签名变得更有可读性。使用列表字符串是为了实现起来更加的方便。

```
1 type Log = [String]
```

相较于提供用户一个值构造函数，更好的是提供一个 `runLogger` 函数，用于记录操作。它将同时返回操作的结果，以及计算时的日志。

```
1 runLogger :: Logger a -> (a, Log)
```

控制退出

`Monad` typeclass 不提供任何方法让值脱离它们的 monadic 束缚。可以使用 `return` 将值注入进一个单子中，可以使用 `(>>=)` 将值从单子中提取出来，而位于其右侧的函数将会得到解包的值，该函数再将其结果重新进行包装。

多数单子拥有一个或多个类似 `runLogger` 的函数。最典型的的就是 `IO`，通常只会在退出程序时才退出。

一个单子执行函数会将代码运行在单子内部并展开其结果。这样的函数通常是提供给值从其一元包装器中转义的唯一方法。因此单子的作者可以完全控制单子内发生的任何事物。

留痕

当在一个 `Logger` 中操作时，用户代码调用 `record` 来记录事物。

```
1 record :: String -> Logger ()
```

由于记录发生在单子的管道中，所以操作的结果不提供任何信息。

通常来说，一个单子会提供一个或多个像 `record` 一样的帮助函数。它们是访问该单子的特殊行为的方法。

模块中也为 `Logger` 类型提供了 `Monad` 实例。它们是客户端模块为了能使用这个单子所需的全部定义。

使用 Logger 单子

下面是将 `glob-to-regexp` 在 `Logger` 单子中转换的方法：

```
1 globToRegex :: String -> Logger String
2 globToRegex cs =
3   globToRegex' cs >>= \ds -> return $ '^' : ds
```

注意 `(>>=)` 的类型：它从左侧的 `Logger` 包裹中中提取值，然后将解包的值传递给右侧的函数。在其右侧的函数则需要使用 `Logger` 包装其结果。这也正是 `return` 所做的：获取一个纯值，包裹该值进入一个单子类型的构造函数中。

```
1 ghci> :type (>>=)
2 (>>=) :: (Monad m) => m a -> (a -> m b) -> m b
3 ghci> :type (globToRegex "" >>=)
4 (globToRegex "" >>=) :: (String -> Logger b) -> Logger b
```

即使编写一个不做任何事情的函数也要调用 `return` 将结果包装成正确的类型。

```
1 globToRegex' :: String -> Logger String
2 globToRegex' "" = return "$"
```

当调用 `record` 来存储一个日志，需要用 `(>>)` 而不是 `(>>=)` 来进行串联。

```

1 globToRegex' ('?' : cs) =
2   record "any"
3   >> globToRegex' cs
4   >>= \ds -> return ('.' : ds)

```

记住它是 `(>>=)` 的一种变体，它会忽略左侧的结果。我们知道 `record` 的结果总是 `()`，因此无需捕获该结果。

我们可以使用 `do` 符号来整洁一下代码：

```

1 globToRegex' ('*' : cs) = do
2   record "kleene star"
3   ds <- globToRegex' cs
4   return $ "." ++ ds

```

解析字符类主要遵循以上这种模式：

```

1 globToRegex' ('[' : '!' : c : cs) =
2   record "character class, negative"
3   >> charClass cs
4   >>= \ds -> return $ "[" ++ c : ds
5 globToRegex' ('[' : c : cs) =
6   record "character class"
7   >> charClass cs
8   >>= \ds -> return $ "[" ++ c : ds
9 globToRegex' ('[' : _) =
10  fail "unterminated character class"

```

混合纯代码与单子代码

根据目前所见到的代码，单子看起来有一个很大的缺点：使用普通的纯函数处理被包装后的值时，会变得很棘手。以下是这个浅显问题的一个简单说明。假设有一个简单的代码运行在 `Logger` 单子中并返回一个字符串。

```

1 ghci> let m = return "foo" :: Logger String

```

如果希望得知该字符串的长度，我们无法直接调用 `length`：因为字符串被封装了，类型并不匹配。

迄今为止我们所能做的就是类似这样：

```

1 ghci> :type m >>= \s -> return (length s)
2 m >>= \s -> return (length s) :: Logger Int

```

通过 `(>>=)` 解包该字符串，接着用一个小的匿名函数来调用 `length` 并通过 `return` 重新进行包装。

由于 `Monad` `typeclass` 已经提供了 `(>>=)` 与 `return` 函数用于解包与打包一个值，`liftM` 函数则不需要知道任何单子实现的细节。


```

1 liftM :: (Monad m) => (a -> b) -> m a -> m b
2 liftM f m = m >>= \i -> return $ f i

```

当我们为一个类型声明 `Functor` `typeclass` 实例时，我们需要编写自己的 `fmap`。相反的，`liftM` 不需要了解单子的内部，因为它们抽象出来了 `(>>=)` 与 `return`。我们只需要通过适当的类型约束编写它一次。

`liftM` 函数预定义在 `Control.Monad` 模块中。

为了检测 `liftM` 是如何提高可读性的，我们来比较连个相同作用的代码。首先是不使用 `liftM` 的情况：

```

1 charClass_wordy (']' : cs) =
2   globToRegex' cs >>= \ds -> return $ ']' : ds
3 charClass_wordy (c : cs) =
4   charClass_wordy cs >>= \ds -> return $ c : ds

```

然后是使用 `liftM` 来去除 `(>>=)` 以及匿名函数：

```

1 charClass (']' : cs) = (']' :) `liftM` globToRegex' cs
2 charClass (c : cs) = (c :) `liftM` globToRegex' cs

```

与 `fmap` 一样，我们经常以中缀形式使用 `liftM`。阅读这种表达式的一种简单方法是“将左侧的纯函数应用于右侧一元操作的结果”。

`liftM` 太有用了以至于 `Control.Monad` 定义了若干变体，用于结合更长的操作链。来看一下 `globToRegex'` 函数的最后一个分句：

```

1 globToRegex' (c : cs) = liftM2 (++) (escape c) (globToRegex' cs)
2
3 escape :: Char -> Logger String
4 escape c
5   | c `elem` regexChars = record "escape" >> return ['\\', c]
6   | otherwise = return [c]
7 where
8   regexChars = "\\+()~$.{}|!"

```

而上述的 `liftM2` 函数定义如下：

```

1 liftM2 :: (Monad m) => (a -> b -> c) -> m a -> m b -> m c
2 liftM2 f m1 m2 =
3   m1 >>= \a ->
4     m2 >>= \b ->
5       return (f a b)

```

它首先执行第一个操作，接着第二个，再使用纯函数 `f` 组合它们的结果，并封装该结果。除了 `liftM2`，`Control.Monad` 中的变体达到了 `liftM5`。

消除一些误解

略。

构建一个 Logger 单子

`Logger` 类型的定义非常简单：

```
1 newtype Logger a = Logger {execLogger :: (a, Log)}
```

它是一个二元元组，第一个元素是具体操作的结果，而第二个元素则是该操作中产生的所有日志的列表。

我们将该元组包装进了一个 `newtype` 中使其变为独立类型。`runLogger` 函数则是从包装中提取出该元组。而暴露出来执行一个日志行为的函数，`runLogger`，则是 `execLogger` 的一个别名。

```
1 runLogger = execLogger
```

`record` 帮助函数则是通过传入的信息，创建了一个单例列表：

```
1 record s = Logger ((), [s])
```

该操作的结果为 `()`，这也是为什么在结果位置放置它的缘故。

现在开始 `Monad` 实例的 `return`，它很简单：不记录日志，并将其入参存储值结果位置。

```
1 instance Functor Logger where
2   fmap f (Logger (a, l)) = Logger (f a, l)
3
4 instance Applicative Logger where
5   pure a = Logger (a, [])
6   (<*>) (Logger (h, w)) (Logger (a, x)) = Logger (h a, w ++ x)
7
8 instance Monad Logger where
9   m >>= k =
10     let (a, w) = execLogger m
11         n = k a
12         (b, x) = execLogger n
13     in Logger (b, w ++ x)
```

注：现代版本 Haskell 在实现 `Monad` 实例之前还需要先实现 `Functor` 与 `Applicative` 实例。

序列化日志，而不是序列化计算

我们定义的 `(>>=)` 确保了左侧的日志信息将会附加到右侧的新的日志信息中。然而，它并没有涉及 `a` 与 `b` 的计算：`(>>=)` 是惰性的。

正如其它大多数的单子行为一样，严格程度是由单子的实现者所控制的。它不是所有单子都共享的常量。实际上一些单子由多种风格，每一种都会有不同的严格程度。

Writer 单子

我们的 `Logger` 单子实际上是标准 `Writer` 单子的特殊版本，可以在 `mtl` 库的 `Control.Monad.Writer` 模块中找到。关于 `Writer` 的例子将会在下一章中讲到。

Maybe 单子

`Maybe` 类型是最简单的 `Monad` 实例。它代表着计算可能会返回一个结果。

当我们将一系列返回 `Maybe` 的计算通过 `(>>=)` 或 `>>` 串联起时，它们任意一个返回 `Nothing` 时，后续的计算都不再进行。

注意，尽管链条并不是完全短路的。链路中的每个 `(>>=)` 或 `(>>)` 仍然会匹配左侧的 `Nothing`，并生产一个 `Nothing` 在右侧，直至最后。容易遗忘的点：当链路中的一个计算失败时，剩下的链路以及 `Nothing` 值的消费在运行时是廉价的，但并不是全免。

执行 Maybe 单子

执行 `Maybe` 单子的函数名为 `maybe`。（记住“执行”一个单子包括对单子求值并返回一个去掉单子类型包装的结果。）

```
1 maybe :: b -> (a -> b) -> Maybe a -> b
2 maybe n _ Nothing = n
3 maybe _ f (Just x) = f x
```

它第一个参数是当结果为 `Nothing` 时，给定的返回；第二个参数则是当结果为 `Just` 时，应用到解包值上的函数。

Maybe 作为优秀的 API 设计

下面是一个 `Maybe` 作为单子使用的例子。给定一个用户名，希望通过移动电话运营商找到账单地址。

```
1 import Data.Map qualified as M
2
3 type PersonName = String
4
5 type PhoneNumber = String
6
7 type BillingAddress = String
8
9 data MobileCarrier
10   = Honest_Bobs_Phone_Network
11   | Morrisas_Marvelous_Mobiles
12   | Petes_Plutocratic_Phones
13   deriving (Eq, Ord)
14
15 findCarrierBillingAddress ::
16   PersonName ->
17   M.Map PersonName PhoneNumber ->
18   M.Map PhoneNumber MobileCarrier ->
19   M.Map MobileCarrier BillingAddress ->
20   Maybe BillingAddress
```

我们的第一个版本是可怕的梯形代码，带着一堆 `case` 表达式：

```
1 variation1 :: (Ord k1, Ord k2, Ord k3) => k1 -> M.Map k1 k2 -> M.Map k2 k3 -> M.Map k3 a ->
  Maybe a
2 variation1 person phoneMap carrierMap addressMap =
3   case M.lookup person phoneMap of
4     Nothing -> Nothing
5     Just number ->
6       case M.lookup number carrierMap of
7         Nothing -> Nothing
8         Just carrier -> M.lookup carrier addressMap
```

`Data.Map` 模块的 `lookup` 函数拥有一个 monadic 返回类型

```
1 ghci> :module +Data.Map
2 ghci> :type Data.Map.lookup
3 Data.Map.lookup :: (Ord k, Monad m) => k -> Map k a -> m a
```

换言之，如果给定的键在 `map` 中，`lookup` 通过 `return` 将值注入至单子；反之，调用 `fail`。这是很有意思的 API 设计，尽管我们会认为这是一个糟糕的选择。

- 正面来看，成功与失败的行为是根据调用 `lookup` 所产生的单子自动定值的。更妙的是，`lookup` 本身并不需要知道或关系这些行为。
- 反面来看，问题在于错误的单子中使用 `file` 会抛出令人厌恶的异常。

使用 `do` 简化：

```
1 variation2 :: (Ord k1, Ord k2, Ord k3) => k1 -> M.Map k1 k2 -> M.Map k2 k3 -> M.Map k3 b ->
  Maybe b
2 variation2 person phoneMap carrierMap addressMap = do
3   number <- M.lookup person phoneMap
4   carrier <- M.lookup number carrierMap
5   M.lookup carrier addressMap
```

如果这些 `lookup` 中任意一个失败了，那么 `(>=)` 以及 `(>>)` 的定义意味着函数的整个结果将会变为 `Nothing`，正如第一次尝试中显式的使用 `case` 那样。

通过 `flip` 还可以将函数体变为一行：

```
1 variation3 :: Ord a => a -> M.Map a a -> M.Map a a -> M.Map a b -> Maybe b
2 variation3 person phoneMap carrierMap addressMap =
3   lookup phoneMap person >= lookup carrierMap >= lookup addressMap
4   where
5     lookup = flip M.lookup
```

列单子

虽然 `Maybe` 类型即可以表示没有值，也可以表示一个值，但在很多情况下，我们希望返回一些事先不知道的结果。显然列表非常适合这个目的。列表的类型表明我们可以将它用作单子，因为他的类型构造函数有一个自由变量。当然，我们可以使用列表作为单子。

略。

如果将 `(>>=)` 应用至列表，我们会发现其类型为 `[a] -> (a -> [b]) -> [b]`。这看起来就跟 `map` 类似：

```
1 ghci> :type (>>=)
2 (>>=) :: (Monad m) => m a -> (a -> m b) -> m b
3 ghci> :type map
4 map :: (a -> b) -> [a] -> [b]
```

看上去并不匹配，不过可以使用 `flip`：

```
1 ghci> :type (>>=)
2 (>>=) :: (Monad m) => m a -> (a -> m b) -> m b
3 ghci> :type flip map
4 flip map :: [a] -> (a -> b) -> [b]
```

仍有一个问题：`flip map` 的第二个参数的类型是 `a -> b`，而 `(>>=)` 对于列表的第二个参数是 `a -> [a]`。我们该怎么做呢？

函数 `flip map` 可以返回任何 `b` 类型作为其结果。如果在签名中将 `b` 替换为 `[b]`，那么类型签名就变为 `a -> (a -> [a]) -> [[b]]`。换言之，如果我们 `map` 一个函数返回列表的列表，我们则会得到列表的列表作为返回。

```
1 ghci> flip map [1,2,3] (\a -> [a,a+100])
2 [[1,101],[2,102],[3,103]]
```

有趣的是，我们并没有真正改变类型签名。`(>>=)` 的类型是 `[a] -> (a -> [b]) -> [b]`，而当映射函数返回列表时，`flip map` 的类型为 `[a] -> (a -> [b]) -> [[b]]`。这里仍有一个类型不匹配；我们仅仅将一项从类型签名的中间移到了末尾。不过这并没有白费功夫：现在需要一个接受 `[[b]]` 并返回 `[b]` 的函数，这里推荐 `concat`：

```
1 ghci> :type concat
2 concat :: [[a]] -> [a]
```

该类型建议我们翻转参数给到 `map`，接着 `concat` 结果得到一个非嵌套的列表。

```
1 ghci> :type \xs f -> concat (map f xs)
2 \xs f -> concat (map f xs) :: [a] -> (a -> [a1]) -> [a1]
```

这正是列表的 `(>>=)` 的定义：

```
1 instance Monad [] where
2   return x = [x]
3   xs >>= f = concat (map f xs)
```

注：以上是旧版的定义。略。

理解列表单子

列表单子类似于我们熟悉的 Haskell 工具：列表表达式。我们可以通过两个列表的笛卡尔积来说明它们的相似性。首先是写一个列表表达式：

```
1 comprehensive xs ys = [(x, y) | x <- xs, y <- ys]
```

这一次，我们将对一元代码使用花括号表示法。这将突出一元代码在结构上与列表表达式的相似之处：

```
1 monadic xs ys = do x <- xs; y <- ys; return (x, y)
```

注：原本的 `monadic xs ys = do { x <- xs; y <- ys; return (x, y) }` 会被 HLS 简化成上述代码。

这里的唯一真正的区别是，我们构造的值出现在一系列表达式的末尾，而不是像列表表达式那样出现在开头。两者的结果是完全一致的：

```
1 ghci> comprehensive [1,2] "bar"
2 [(1,'b'),(1,'a'),(1,'r'),(2,'b'),(2,'a'),(2,'r')]
3 ghci> comprehensive [1,2] "bar" == monadic [1,2] "bar"
4 True
```

刚开始的时候很容易对列表单子感到困惑，让我们再次看一下单子笛卡尔乘积代码。这次将重新排列函数：

```
1 blockyDo :: Monad m => m a -> m b -> m (a, b)
2 blockyDo xs ys = do
3   x <- xs
4   y <- ys
5   return (x, y)
```

对于列表 `xs` 中的每个元素，函数的其余部分求值一次，每次都将 `x` 绑定到列表中的不同的值上。然后对于列表 `ys` 中的每个元素，函数的剩余部分求值一次，每次都将 `y` 绑定到列表中的不同值。

这里我们真正拥有的是一个双重嵌套循环！这突出了一个关于单子的重要事实：除非知道一个单子代码块将在那个单子中执行，否则无法预测它的行为。

现在让嵌套的循环更加明显一些：

```
1 blockyPlain :: (Monad m) => m a -> m b -> m (a, b)
2 blockyPlain xs ys =
3   xs
4   >>= \x ->
5     ys
6     >>= \y -> return (x, y)
7
```

```

8  blockyPlain_reloaded :: [a] -> [b] -> [(a, b)]
9  blockyPlain_reloaded xs ys =
10  concatMap (\x -> concatMap (\y -> return (x, y)) ys) xs

```

注：这里与文中不同之处在于使用了 `concatMap` 而不是 `concat (map (...)`。

令列表单子生效

下面是一个简单的暴力约束求解。给定一个整数，它会找到所有正整数对，这些正整数相乘后会得到该值（这就是要解决的约束）。

```

1  guarded :: Bool -> [a] -> [a]
2  guarded True xs = xs
3  guarded False _ = []
4
5  multiplyTo :: Int -> [(Int, Int)]
6  multiplyTo n = do
7    x <- [1 .. n]
8    y <- [x .. n]
9    guarded (x * y == n) $ return (x, y)

```

测试：

```

1  ghci> multiplyTo 8
2  [(1,8),(2,4)]
3  ghci> multiplyTo 100
4  [(1,100),(2,50),(4,25),(5,20),(10,10)]
5  ghci> multiplyTo 891
6  [(1,891),(3,297),(9,99),(11,81),(27,33)]

```

do 代码块的脱糖

略。

单子作为可编程的分号

略。

为何选择无糖？

略。

State 单子

我们早在第十章节中就发现了 `Parse` 是一个单子。它有着两个独立的角度：一个是当解析失败时，提供带有细节的错误信息（通过 `Either` 类型达成）；另一个就是它携带着隐式状态，该例子中为 `ByteString`。

这种读写状态的需求在 Haskell 程序中很常见，因此标准库提供了一个名为 `State` 的单子，其位于 `Control.Monad.State` 模块中。

当我们的 `Parse` 类型携带一个 `ByteString` 作为它的状态片段时，`state` 单子可以携带任何类型的状态。我们将状态的未知类型称为 `s`。

给定一个状态值，我们检查它，然后产生一个结果和一个新的状态值。假设结果可以是任何类型 `a`。捕获这个想法的类型签名是 `s -> (a, s)`；获取一个状态 `s`，对其做某些操作，然后返回结果 `a`，以及一个新的状态 `s`。

状态单子

我们先来开发一个 `State` 单子

```
1 type SimpleState s a = s -> (a, s)
```

我们的单子是一个函数，用于转换一个状态，并返回一个值。正因如此，状态单子有时候会被称为状态转换单子。

这是一个类型别名，而不是一个新类型，所以这里有点作弊，不过这可以简化很多之后的描述。

早在本章开头，我们提到了一个单子有一个单类型入参的类型构造函数，而这里则是带两个参数的类型。这里的关键是要理解我们可以部分应用类型，正如部分应用普通函数那样。接下来是一个简单的示例：

```
1 type StringState a = SimpleState String a
```

这里将变量 `s` 绑定至 `String`。`StringState` 类型仍有一个类型参数 `a`。现在有了一个适合单子的类型构造函数，换言之，单子的类型构造函数是 `SimpleState s`，而不仅仅只是 `SimpleState`。

接下来是单子所需的 `return` 函数：

```
1 returnSt :: a -> SimpleState s a
2 returnSt a = \s -> (a, s)
```

它所做的就是获取结果和当前状态，并将它们“元组化”。到目前为止，我们可能已经习惯了这样的想法，即带有多个参数的 Haskell 函数知识一个单参数的函数链，不过以防万一，这里有一种更熟悉的编写 `returnSt` 的方法，它使这个函数更明显且简单：

```
1 returnAlt :: a -> SimpleState s a
2 returnAlt a s = (a, s)
```


单子最后一块定义就是 `(>>=)`，这里使用了标准库中 `State` 定义中的实际变量名：

```
1 bindSt :: SimpleState s a -> (a -> SimpleState s b) -> SimpleState s b
2 bindSt m k = \s -> let (a, s') = m s in k a s'
```

这些单字母变量名对可读性并没有什么好处，所以让我们看看是否可以替换一些更有意义的名称。

```
1 -- m == step
2 -- k == makeStep
3 -- s == oldState
4 bindAlt step makeStep oldState =
5   let (result, newState) = step oldState
6   in makeStep result newState
```

为了理解这个定义，记住 `step` 是一个类型为 `s -> (a, s)` 的函数。当需要计算它时，将会得到一个元组，我们需要使用它返回一个新的 `s -> (a, s)` 类型的函数。这恐怕比从 `bindAlt` 类型中去掉 `SimpleState` 类型别名，以及测试类型的入参与返回值，要来的更简单。

```
1 bindAlt :: (s -> (a, s)) -> (a -> s -> (b, s)) -> (s -> (b, s))
```

读取与修改状态

状态单子的 `(>>=)` 与 `return` 定义很简单：移动状态，但并不触碰该状态。

```
1 getSt :: SimpleState s s
2 getSt = \s -> (s, s)
3
4 putSt :: s -> SimpleState s ()
5 putSt s = const ((), s)
```

注：原文的 `putSt s = _ -> ((), s)` 现改为 `putSt s = const ((), s)`。

真正的状态单子可用吗？

我们在前一节中使用的唯一简化技巧是为 `SimpleState` 使用类型别名而不是类型定义。如果引入了一个线类型的包装器，那么额外的包装和解包将使代码变得更难理解。

为了定义一个 `Monad` 实例，我们需要提供合适的类型构造函数，以及 `(>>=)` 与 `return` 的定义。这就是真实的 `State` 定义。

```
1 newtype State s a = State
2   { runState :: s -> (a, s)
3   }
```

这里所做的是包装 `s -> (a, s)` 类型进一个 `State` 构造函数中。使用 Haskell 的 record 语义来进行类型定义，自动获得一个 `runState` 函数用来从其构造函数中解包一个 `State` 值。`runState` 的类型是 `State s a -> s -> (a, s)`。

`return` 的定义与 `SimpleState` 几乎相同，除了需要通过 `State` 构造函数来进行包装。

```
1 returnState :: a -> State s a
2 returnState a = State (a,)
```

注：原文为 `returnState a = State $ s-> (a, s)`，这里进行了简化。

`(>>=)` 的定义有点复杂，因为它需要使用 `runState` 来移除 `State` 包装器。

```
1 bindState :: State s a -> (a -> State s b) -> State s b
2 bindState m k = State $ \s ->
3   let (a, s') = runState m s
4   in runState (k a) s'
```

该函数与早前定义的 `bindSt` 仅在多了包装和解包动作上有所不同。

读与修改状态也是同理，添加包装：

```
1 get :: State s s
2 get = State $ \s -> (s, s)
3
4 put :: s -> State s ()
5 put s = State $ const (((), s))
```

注：原文为 `put s = State $ _ -> (((), s))`。

使用状态单子：生成随机值

我们已经使用过状态单子的前身 `Parse` 用于解析二进制数据。这个例子中，是将要操作的状态类型直接连接到 `Parse` 类型中。

`State` 单子，相反，接受任意状态类型作为参数，`State ByteString`。

如果拥有命令式语言的背景，那么状态单子可能会比其它单子更易接受。因为命令式语言都是关于携带一些隐式状态，读取一部分，并通过赋值修改其它部分，而这正是状态单子的作用。

Haskell 生成随机值的标准库名为 `System.Random`。它允许任何类型的随机值生成，不仅仅是数值。该模块包含了若干方便的函数于 `IO` 单子中。例如，等同于 C 的 `rand` 函数：

```
1 import System.Random
2
3 rand :: IO Int
4 rand = getStdRandom $ randomR (0, maxBound)
```

(`randomR` 函数接受一个闭合的区间作为随机值的范围)

`System.Random` 模块提供一个 typeclass, `RandomGen`, 其供我们定义随机的 `Int` 源。`StdGen` 类型则是标准 `RandomGen` 的实例。它生成伪随机值。如果我们拥有真正随机数据的外部来源, 便可以将其作为 `RandomGen` 的实例, 并获得真正随机的值, 而不仅仅是伪随机值。

另一个 typeclass, `Random`, 则指示如何生成特定类型的随机值。模块定义了所有常用类型的 `Random` 实例。

顺带一提, 上面 `rand` 的定义读取并修改了一个内置的全局随机生成器, 该生成器位于 `IO` 单子中。

尝试纯粹性

到目前为止, 我们一直强调尽可能避免使用 `IO` 单子, 如果只是为了生成一些随机值而使用它, 那就很可惜。实际上 `System.Random` 包含了纯随机数生成函数。

纯函数的缺点是, 我们必须获得或创建一个随机生成器, 在调用它时, 它返回一个新的随机数生成器: 记住, 在纯函数中我们不能修改现有生成器的状态。

如果我们忘记不可变性, 并且在函数中重用相同的生成器, 那么则会得到完全相同的“随机”数:

```
1 twoBadRandoms :: (RandomGen g) => g -> (Int, Int)
2 twoBadRandoms gen = (fst $ random gen, fst $ random gen)
```

`random` 函数使用了一个隐式范围而不是 `randomR` 里用户提供的范围。`getStdGen` 函数从 `IO` 单子中获取当前的全局标准数值生成器。

不幸的是, 正确的传递和使用生成器的连续版本并不会产生令人满意的数。

```
1 twoGoodRandoms :: (RandomGen g) => g -> ((Int, Int), g)
2 twoGoodRandoms gen =
3   let (a, gen') = random gen
4       (b, gen'') = random gen'
5   in ((a, b), gen'')
```

现在我们了解了状态单子, 但是它看起来是隐藏生成器的一个很好的候选。状态单子允许我们规整的管理可变状态, 同时保证我们的代码不会有其它意想不到的副作用, 比如修改文件或建立网络连接。这使得我们更容易推断代码的行为。

状态单子中的随机值

下面是一个状态单子, 携带一个 `StdGen` 作为它的状态:

```
1 type RandomState a = State StdGen a
```

生成一个随机值现在只需要获取当前的生成器, 使用它, 然后修改状态, 用新的生成器替换它。

```

1 getRandom :: (Random a) => RandomState a
2 getRandom =
3   get >>= \gen ->
4     let (val, gen') = random gen
5     in put gen' >> return val

```

我们现在可以使用之前看到的一些一元机制来编写一个更简洁的函数，用于给出一对随机数：

```

1 getTwoRandoms :: (Random a) => RandomState (a, a)
2 getTwoRandoms = liftM2 (,) getRandom getRandom

```

运行状态单子

正如之前提到的那样，每个单子都有其特化的计算函数。本例的状态单子，有若干种选择：

- `runState` 同时返回结果和最终状态；
- `evalState` 仅返回结果，丢弃最终状态；
- `execState` 丢弃结果，仅返回最终状态。

`evalState` 与 `execState` 函数仅仅是 `runState` 以及 `fst` 与 `snd` 的组合。因此最需要记住的是 `runState` 函数。

以下是如何实现 `getTwoRandoms` 函数的完整用例：

```

1 runTwoRandoms :: IO (Int, Int)
2 runTwoRandoms = do
3   oldState <- getStdGen
4   let (result, newState) = runState getTwoRandoms oldState
5   setStdGen newState
6   return result

```

`runState` 的调用遵循一个标准模式：传递它至一个包含在状态单子中的函数，以及一个初始状态；它返回函数的结果以及最终状态。

围绕 `runState` 调用的代码仅仅获取当前全局 `StdGen` 值，然后替换它，以便后续对 `runTwoRandoms` 或其他随机生成函数的调用可以获得更新的状态。

更多特性

很难想象编写只有一个状态值要传递的代码。当我们想要一次跟踪多个状态片段时，通常的技巧是在一个数据类型中维护它们。这里有一个例子：跟踪我们分发的随机数的数量：

```

1 data CountedRandom = CountedRandom
2   { crGen :: StdGen,

```

```

3     crCount :: Int
4   }
5
6   type CRState = State CountedRandom
7
8   getCountedRandom :: (Random a) => CRState a
9   getCountedRandom = do
10     st <- get
11     let (val, gen') = random (crGen st)
12     put CountedRandom {crGen = gen', crCount = crCount st + 1}
13     return val

```

该例子碰巧消耗了状态的两个元素，并构造了一个全新的状态。更普遍的场景是，可能只读取或修改状态的一部分。该函数获取到目前为止生成的随机值数量。

```

1   getCount :: CRState Int
2   getCount = gets crCount

```

注：原文为 `getCount = crCount `liftM` get`，这里用 `gets crCount` 进行简化。
如果我们希望更新部分状态，那么代码可能就没那么显而易见了。

```

1   putCount :: Int -> CRState ()
2   putCount a = do
3     st <- get
4     put st {crCount = a}

```

还存在一个名为 `modify` 的函数，其组合了 `get` 与 `put` 步骤。它将状态转换函数作为参数，但它并不令人满意：我们仍然无法摆脱笨拙的记录更新语法：

```

1   putCountModify :: Int -> CRState ()
2   putCountModify a = modify $ \st -> st {crCount = a}

```

单子与函子

略（原文 Haskell 版本过旧，已不适合现在的学习）。

15 Programming with monads

练习：关联列表

Web 客户端与服务端之间经常通过简单的键值对列表进行信息传输。

```
1 name=Attila+%42The+Hun%42&occupation=Khan
```

这里的编码名为 `application/x-www-form-urlencoded`，同时非常便于理解。每个键值对都被一个“&”符号分隔。在一个兼职对中，键为“=”符号之前的所有字符，而值为之后的所有字符。

显然可以将一个 `String` 作为键，但是 HTTP 并不清楚该键是否必须跟着一值。可以通过 `Maybe String` 来表示一个模糊的值。如果值为 `Nothing`，即无值可展示。当使用 `Just` 包裹一个值时则以为着有值。使用 `Maybe` 让我们可以区分“无值”与“空值”。

Haskell 程序员使用类型为 `[(a, b)]` 的关联列表，可以视作关联列表中的每个元素都是键与值的关联。

假设我们想用这些列表中的一个来填充一个数据结构。

```
1 data MovieReview = MovieReview
2   { revTitle :: String,
3     revUser  :: String,
4     revReview :: String
5   }
```

从一个朴素的函数开始：

```
1 simpleReview :: [(String, Maybe String)] -> Maybe MovieReview
2 simpleReview alist =
3   case lookup "title" alist of
4     Just (Just title@(_ : _)) ->
5       case lookup "user" alist of
6         Just (Just user@(_ : _)) ->
7           case lookup "review" alist of
8             Just (Just review@(_ : _)) ->
9               Just (MovieReview title user review)
10          _ -> Nothing -- no review
11        _ -> Nothing -- no user
12      _ -> Nothing -- no title
```

当关联列表包含了所有必要值且不为空值时，它将返回一个 `MovieReview`。

我们对 `Maybe` 单子已经很熟悉了，因此可以简化一下上述的阶梯式代码：

```
1 maybeReview :: [(String, Maybe [Char])] -> Maybe MovieReview
2 maybeReview alist = do
3   title <- lookup1 "title" alist
4   user  <- lookup1 "user"  alist
5   review <- lookup1 "review" alist
```

```

6     return $ MovieReview title user review
7
8     lookup1 :: Eq a1 => a1 -> [(a1, Maybe [a2])] -> Maybe [a2]
9     lookup1 key alist =
10         case lookup key alist of
11             Just (Just s@(_ : _)) -> Just s
12             _ -> Nothing

```

尽管这看起来简洁多了，但仍然在重复自身。我们可以利用 `MovieReview` 构造函数作为一个普通的纯函数，通过 *lifting* 它至单子：

```

1     liftedReview :: [(String, Maybe [Char])] -> Maybe MovieReview
2     liftedReview alist =
3         liftM3
4             MovieReview
5             (lookup1 "title" alist)
6             (lookup1 "user" alist)
7             (lookup1 "review" alist)

```

这里仍然有一些重复，不过更难简化。

泛化的 lifting

尽管使用 `liftM3` 得以简化代码，但是我们无法使用 `liftM` 家族的函数来解决更泛化的问题，因为标准库仅定义到了 `liftM5`。虽然可以自定义此类型的变体函数，但是这个数量仍然是一个问题。

假设一个构造函数或者纯函数接收十个参数，并决定坚持使用标准库，这可能就不太合适了。

在 `Control.Monad` 中，有一个名为 `ap` 的函数拥有着有趣的类型签名。

```

1     ghci> :m +Control.Monad
2     ghci> :type ap
3     ap :: (Monad m) => m (a -> b) -> m a -> m b

```

我们可能会疑惑谁会将单参数纯函数放在单子中，且为什么。不过回想一下所有的 Haskell 函数实际上只接受一个参数，这里开始将看到其与 `MovieReview` 构造函数的管理。

```

1     ghci> :type MovieReview
2     MovieReview :: String -> String -> String -> MovieReview

```

我们当然可以简单的将类型写作 `String -> (String -> (String -> MovieReview))`。如果使用旧的 `liftM` 将 `MovieReview` 提升至 `Maybe` 单子，那么我们将会得到类型为 `Maybe (String -> (String -> (String -> MovieReview)))` 的值。现在可以看出来该类型适用于单个参数的 `ap`。我们可以一次将这个传递给 `ap`，并继续链式执行，直到得到该定义。

```

1  apReview :: [(String, Maybe [Char])] -> Maybe MovieReview
2  apReview alist =
3      MovieReview
4      `liftM` lookup1 "title" alist
5      `ap` lookup1 "user" alist
6      `ap` lookup1 "review" alist

```

注：以下为步骤拆解后的类型变化。

```

1  MovieReview :: String -> ( String -> String -> MovieReview )
2  MovieReview `liftM` :: Maybe String -> Maybe ( String -> String -> MovieReview )
3  MovieReview `liftM` lookup1 "title" alist :: Maybe ( String -> String -> MovieReview )
4  MovieReview `liftM` lookup1 "title" alist `ap` :: Maybe String -> Maybe ( String ->
    MovieReview )
5  MovieReview `liftM` lookup1 "title" alist `ap` lookup1 "user" alist :: Maybe ( String ->
    MovieReview )

```

我们可以像这样把 `ap` 的应用链接起来，只要有需要就可以多次链接，从而绕过 `liftM` 系列函数。

看待 `ap` 的另一种有用的方式是，它是我们熟悉的 `($\$$)` 操作符的一元等价物：可以把 `ap` 读作 *apply*。当比较两者的函数签名时可知：

```

1  ghci> :type ( $\$$ )
2  ( $\$$ ) :: (a -> b) -> a -> b
3  ghci> :type ap
4  ap :: (Monad m) => m (a -> b) -> m a -> m b

```

实际上，`ap` 通常被定义为 `liftM2 id` 或是 `liftM2 ($\$$)`。

寻找其它方案

以下是某人的电话号码：

```

1  data Context = Home | Mobile | Business deriving (Eq, Show)
2
3  type Phone = String
4
5  albulena :: [(Context, String)]
6  albulena = [(Home, "+355-652-55512")]
7
8  nils :: [(Context, String)]
9  nils =
10     [ (Mobile, "+47-922-55-512"),
11       (Business, "+47-922-12-121"),
12       (Home, "+47-925-55-121"),
13       (Business, "+47-922-25-551")
14     ]
15
16  twalumba :: [(Context, String)]

```



```
17 twalumba = [(Business, "+260-02-55-5121")]
```

假设我们想要通过打电话来联系某人。我们不希望通过商务号码，更倾向于使用家庭号码（如果存在的话）而不是移动电话。

```
1 onePersonalPhone :: [(Context, Phone)] -> Maybe Phone
2 onePersonalPhone ps =
3   case lookup Home ps of
4     Nothing -> lookup Mobile ps
5     Just n -> Just n
```

当然我们可以使用 `Maybe` 作为返回类型，我们无法考虑到某人可能拥有多个号码的可能性。因此我们需要一个列表：

```
1 allBusinessPhones :: [(Context, Phone)] -> [Phone]
2 allBusinessPhones ps = map snd numbers
3   where
4     numbers =
5       case filter (contextIs Business) ps of
6         [] -> filter (contextIs Mobile) ps
7         ns -> ns
8
9 contextIs :: Eq a => a -> (a, b) -> Bool
10 contextIs a (b, _) = a == b
```

注意这两个函数的 `case` 表达式结构类似：一个替代方法处理第一次查找返回空值，而另一个方法处理非空情况。

```
1 ghci> onePersonalPhone twalumba
2 Nothing
3 ghci> onePersonalPhone albulena
4 Just "+355-652-55512"
5 ghci> allBusinessPhones nils
6 ["+47-922-12-121", "+47-922-25-551"]
```

Haskell 的 `Control.Monad` 模块定义了一个 `typeclass`，`MonadPlus`，它让我们可以出 `case` 表达式中抽象出公共模式。

```
1 ghci> import Control.Monad
2 ghci> :i MonadPlus
3 type MonadPlus :: (* -> *) -> Constraint
4 class (GHC.Base.Alternative m, Monad m) => MonadPlus m where
5   mzero :: m a
6   mplus :: m a -> m a -> m a
7   -- Defined in 'GHC.Base'
8 instance MonadPlus IO -- Defined in 'GHC.Base'
9 instance MonadPlus [] -- Defined in 'GHC.Base'
10 instance MonadPlus Maybe -- Defined in 'GHC.Base'
```

`mzero` 代表一个空值，而 `mplus` 则是将两个结果合并成一个。我们现在可以使用 `mplus` 来完全的移除 `case` 表达式了。

```

1 oneBusinessPhone :: [(Context, Phone)] -> Maybe Phone
2 oneBusinessPhone ps = lookup Business ps `mplus` lookup Mobile ps
3
4 allPersonalPhones :: [(Context, Phone)] -> [Phone]
5 allPersonalPhones ps =
6   map snd $
7     filter (contextIs Home) ps
8     `mplus` filter (contextIs Mobile) ps

```

这些函数中由于我们知道 `lookup` 返回一个类型为 `Maybe` 的值，以及 `filter` 返回一个列表，那么 `mplus` 使用的版本就很明显了。

更有趣的是，我们可以使用 `mzero` 与 `mplus` 来编写任何对 `MonadPlus` 实例都有用的函数。以下是标准查找函数的例子：

```

1 lookup :: (Eq a) => a -> [(a, b)] -> Maybe b
2 lookup _ [] = Nothing
3 lookup k ((x,y):xys) | x == k = Just y
4                       | otherwise = lookup k xys

```

我们可以轻易地泛化返回类型至任意 `MonadPlus` 的实例：

```

1 lookupM :: (MonadPlus m, Eq a) => a -> [(a, b)] -> m b
2 lookupM _ [] = mzero
3 lookupM k ((x, y) : xys)
4   | x == k = return y `mplus` lookupM k xys
5   | otherwise = lookupM k xys

```

如果结果类型是 `Maybe`，即要么没有结果，要么一个结果；如果类型是列表，即所有结果；或者其它更适合于 `MonadPlus` 的奇异实例。

`mplus` 并不是加法

尽管 `mplus` 函数包含了“plus”，但这并不意味着将两个值求和。

根据单子定义 `mplus` 可能会实现类似于加法的操作。例如立标单子中的 `mplus` 是作为 `(++)` 操作符实现的。

```

1 ghci> [1,2,3] `mplus` [4,5,6]
2 [1,2,3,4,5,6]

```

然而切换到其他单子，类似加法的行为并不成立：

```

1 ghci> Just 1 `mplus` Just 2
2 Just 1

```

`MonadPlus` 的规则

`MonadPlus` typeclass 的实例相较于普通的单子规则还需要遵循某些其他简单的规则。

如果 `mzero` 出现在绑定表达式的左侧，则实例必须短路。换言之，表达式 `mzero >= f` 的计算结果必须与 `mzero` 单独计算的结果相同。

```
1 mzero >= f == mzero
```

如果 `mzero` 出现在序列表达式的右侧，那么该实例必须短路。

```
1 v >> mzero == mzero
```

失败安全的 MonadPlus

早在“The Monad typeclass”章节中提到的 `fail` 函数，被告知不要使用它：在很多 `Monad` 中，它被实现为对 `error` 的调用，这会产生令人不快的后果。

`MonadPlus` typeclass 为我们提供了一种更温和的方式来失败计算，而不会出现 `fail` 或 `error`。上面介绍的规则允许我们在任何需要的地方在代码中引入 `mzero`，并在该点上短路。

在 `Control.Monad` 模块中，标准函数 `guard` 将此理念打包成了方便的样式。

```
1 guard      :: (MonadPlus m) => Bool -> m ()
2 guard True  = return ()
3 guard False = mzero
```

下面是个简单的例子，一个函数接受一个值 `x` 并计算它对另一个数字 `n` 取模。如果结果为零返回 `x`，否则返回当前单子的 `mzero`。

```
1 x `zeroMod` n = guard ((x `mod` n) == 0) >> return x
```

隐藏管道的冒险

略。

我们给单子取名为 `Supply`，将执行函数 `runSupply` 提供一个列表；需要确保列表中每一个元素都是唯一的。

```
1 runSupply :: Supply s a -> [s] -> (a, [s])
```

单子并不会关心内部的值：它们有可能是随机数，临时文件的名称，或是 HTTP cookies 的 ID。

单子内部，消费者每次需求一个值，`next` 则会从列表中获取下一个元素并给到消费者。每个值都会被 `Maybe` 构造函数包装，以防列表长度不够。

```
1 next :: Supply s (Maybe s)
```

为了隐藏管道，模块声明时仅导出类型构造函数，执行函数，以及 `next` 操作函数：

```
1 module Supply (Supply, next, runSupply) where
```

管道非常的简单：使用 `newtype` 声明来包装一个 `State` 单子：

```

1 import Control.Monad.State
2
3 newtype Supply s a = S (State [s] a)

```

这里型参 `s` 是我们将要提供的唯一值类型，而 `a` 是为了类型成为单子而必须提供的通常类型参数。

我们对 `Supply` 类型的 `newtype` 的使用和模块头文件联合起来防止用户使用 `State` 单子的 `get` 和 `set` 操作。由于模块没有导出 `S` 的构造函数，所以外部无法通过编程的方式看到包装的 `State` 单子，也无法访问它。

此刻有了类型 `Supply`，我们需要它来创建 `Monad` typeclass 实例。我们可以遵循 `(>>=)` 以及 `return` 的通常模式，但这只是存粹的样板代码。我们所要做的是包装盒解包 `State` 单子的 `(>>=)` 版本，并使用 `S` 值构造函数返回。

```

1 unwrapS :: Supply s a -> State [s] a
2 unwrapS (S s) = s
3
4 instance Functor (Supply s) where
5     fmap f s = S $ fmap f $ unwrapS s
6
7 instance Applicative (Supply s) where
8     pure = S . return
9     f <*> a = S $ unwrapS f <*> unwrapS a
10
11 instance Monad (Supply s) where
12     s >>= m = S $ unwrapS s >>= unwrapS . m

```

注：与原文不同，实现单子实例前还需分别实现函子实例与应用函子实例。

Haskell 程序员不喜欢样板文件，且可以肯定的是，GHC 有一个可爱的语言扩展用于消除这些样板。使用它需要再源文件顶部添加：

```

1 {-# LANGUAGE GeneralisedNewtypeDeriving #-}

```

通常而言，我们只能自动派生一些标准 typeclass 的实例，例如 `Show` 和 `Eq`。顾名思义，`GeneralisedNewtypeDeriving` 扩展了派生 typeclass 实例的能力，且它是特定于 `newtype` 声明的。如果我们包装的类型是任何 typeclass 的实例，扩展可以自动将我们的新类型作为该 typeclass 的实例，如下所示：

```

1 newtype Supply s a = S (State [s] a)
2     deriving (Monad)

```

那么接下来就是 `next` 与 `runSupply` 的定义了：

```

1 runSupply :: Supply s a -> [s] -> (a, [s])
2 runSupply (S m) xs = runState m xs
3
4 next :: Supply s (Maybe s)

```

```

5 next = S $ do
6   st <- get
7   case st of
8     [] -> return Nothing
9     (x : xs) -> do
10      put xs
11      return $ Just x

```

加载模块至 `ghci` 测试一下：

```

1 ghci> :l Supply
2 [1 of 1] Compiling Supply          ( Supply.hs, interpreted )
3 Ok, one module loaded.
4 ghci> runSupply next [1,2,3]
5 (Just 1,[2,3])
6 ghci> import Control.Monad
7 ghci> runSupply (liftM2 (,) next next) [1,2,3]
8 ((Just 1,Just 2),[3])
9 ghci> runSupply (liftM2 (,) next next) [1]
10 ((Just 1,Nothing),[])

```

我们还可以验证 `State` 单子是否泄漏了。

```

1 ghci> :browse Supply
2 type role Supply nominal nominal
3 type Supply :: * -> * -> *
4 newtype Supply s a = S (State [s] a)
5 runSupply :: Supply s a -> [s] -> (a, [s])
6 next :: Supply s (Maybe s)
7 ghci> :info Supply
8 type role Supply nominal nominal
9 type Supply :: * -> * -> *
10 newtype Supply s a = S (State [s] a)
11   -- Defined at Supply.hs:12:1
12 instance Applicative (Supply s) -- Defined at Supply.hs:32:10
13 instance Functor (Supply s) -- Defined at Supply.hs:29:10
14 instance Monad (Supply s) -- Defined at Supply.hs:36:10

```

支持随机数

如果想要使用 `Supply` 单子作为随机数的源，那么我们将会遇到一个小困难。理想情况下，我们希望能够为它提供无限流式的随机数。我们可以在 `IO` 单子中获得一个 `StdGen`，但是当完成时必须“放回”一个不同的 `StdGen`。如果不这么做，那么下一次获取 `StdGen` 的代码将获得相同的状态，即产生一样的随机数，这可是灾难性的事故。

从 `System.Random` 模块中可知，目前的需求很难被调和。我们可以使用 `getStdRandom`，其类型可以确保获得一个 `StdGen` 时，又会放回一个。

```

1 ghci> :type getStdRandom

```

```
2 getStdRandom :: (StdGen -> (a, StdGen)) -> IO a
```

在给到一个随机值后,可以使用 `random` 来获取一个新的 `StdGen` ;可以使用 `randoms` 来获取一个随机数的无限列表。但是我们该怎么得到一个无限的随机数列表和一个新的 `StdGen` 呢?

答案就在 `RandomGen` typeclass 的 `split` 函数内,该函数接受一个随机数生成器,并将其转换为两个生成器。像这样拆分随机生成器是最不寻常的事:它在纯函数设置中显然非常有用,但本质上既不是必须的,也不是由非纯语言提供的。

使用 `split` 函数时,可以使用 `StdGen` 来生成一个无限随机数列表用于 `runSupply` ,另一个则是用于 `IO` 单子:

```
1 import Supply
2 import System.Random hiding (next)
3
4 randomsIO :: (Random a) => IO [a]
5 randomsIO = getStdRandom $ \g ->
6   let (a, b) = split g in (randoms a, b)
```

如果我们正确的编写了这个函数,那么示例应该在每次调用时打印一个不同的随机数:

```
1 ghci> :l RandomSupply.hs
2 [1 of 2] Compiling Supply          ( Supply.hs, interpreted )
3 [2 of 2] Compiling RandomSupply    ( RandomSupply.hs, interpreted )
4 Ok, two modules loaded.
5 ghci> (fst . runSupply next) `fmap` randomsIO
6 Just (-8154423328023582499)
7 ghci> (fst . runSupply next) `fmap` randomsIO
8 Just (-4209314352233312889)
```

回忆一下, `runSupply` 函数即返回执行一元操作的结果,也返回列表中未使用的剩余部分。由于我们向它传递了一个随机数的无限列表,因此使用 `fst` 以确保在 `ghci` 尝试打印结果时不会被随机数淹没。

再一次尝试

将函数应用与一对元组中的一个元素,并在不改变另一个原始元素的情况下构造一个新元组的模式,这在 Haskell 代码中很常见,以至于它已经变成了标准代码。

`Control.Arrow` 模块中有两个函数, `first` 与 `second` ,即实现了该操作。

```
1 ghci> :m +Control.Arrow
2 ghci> first (+3) (1,2)
3 (4,2)
4 ghci> second odd ('a',1)
5 ('a',True)
```

从实现中分离接口

之前的小节中，我们见识到了在使用 `State` 用于维护 `Supply` 的状态时，是如何隐藏实现的。

另一个让代码更模块化的方式则是分离其接口 – 即代码可以做的，与实现 – 即代码如何做的。

`System.Random` 模块中的标准随机数生成器是很低效的。如果使用 `randomsIO` 函数来提供随机数，那么 `next` 操作的性能不会很好。

一个简单高效的处理方式就是为 `Supply` 提供一个更好的随机数据源。现在让我们将这个想法放在一旁，而去考虑另一种方式，一种在很多设置中有用的方式。我们将单子可以执行的操作与它使用的 `typeclass` 的工作方式分离开来。

```
1 class (Monad m) => MonadSupply s m | m -> s where
2   next :: m (Maybe s)
```

该 `typeclass` 定义了任何 `supply` 单子必须实现的接口。他需要仔细检查，因为它使用了几个暂不熟悉的 Haskell 语言扩展。我们将在接下来的章节中逐一介绍。

若干参数的 `typeclasses`

我们该如何阅读代码切片 `MonadSupply s m` 这个 `typeclass` 呢？如果添加圆括号，那么一个相同的表达式就是 `(MonadSupply s) m`，这更清晰一点。换言之，给定某身为 `Monad` 的类型变量 `m`，我们可以使其成为 `MonadSupply s` 的实例。有别于通常的 `typeclass`，它有一个参数。

语言扩展允许一个 `typeclass` 拥有多个参数，其名称为 `MultiParamTypeClasses`。参数 `s` 的作用与同名的 `Supply` 类型参数相同：它表示下一个函数传递的值的类型。

注意，我们不需要在 `MonadSupply s` 的定义中提到 `(>>=)` 或 `return`，因为 `typeclass` 的上下文（superclass）要求 `MonadSupply s` 必须是 `Monad`。

函数式依赖

回忆一下早前忽略的代码片段，`| m -> s` 是一个函数式依赖，通常称为 *fundep*。我们可以将竖线 `|` 读作“such that”，而箭头 `->` 读作“uniquely determines”。函数式依赖建立了关于 `m` 与 `s` 之间的关系。

功能依赖是由 `FunctionalDependencies` 的语言 `pragma` 控制的。

声明关系的目的是帮助类型检查器。回想一下，Haskell 类型检查器本质上是一个定理证明器，且他的操作方式是保守的：它坚持它的证明必须终止。非终止证明会导致编译器放弃或陷入无限循环。

通过函数式依赖，我们告诉类型检查器，每当它看到 `MonadSupply s` 的上下文中使用某些单子 `m` 时，类型 `s` 是唯一可接受的类型。如果我们忽略函数式依赖，类型检查器将简单的放弃并显示一条错误信息。

现在看一下这个 `typeclass` 的实例：

```
1 import qualified Supply as S
2
3 instance MonadSupply s (S.Supply s) where
4     next = S.next
```

这里类型变量 `m` 被类型 `S.Supply s` 取代。由于函数式依赖，类型检查器知道了当它看到 `S.Supply s` 时，该类型可以用作 `typeclass MonadSupply s` 的实例。

为了去掉最后一层抽象，试着考虑一下类型 `S.Supply Int`。如果没有函数式依赖，我们可以将其声明为 `MonadSupply s` 的实例。然而在尝试使用该实例时，编译器将无法确定类型的 `Int` 参数需要与 `typeclass` 的 `s` 参数相同，同时编译器将报告一个错误。

函数式依赖关系可能很难理解，一旦超出了简单使用，它们在实践中将很难使用。幸运的是，函数依赖关系最常见的使用方式就是像上述这样的简单情况，它们造成的麻烦会少很多。

完善我们的模块

当我们将 `typeclass` 与实例保存在一个名为 `SupplyClass.hs` 的源文件中，我们需要添加一个模块头如下：

```
1 {-# LANGUAGE FlexibleInstances #-}
2 {-# LANGUAGE FunctionalDependencies #-}
3 {-# LANGUAGE MultiParamTypeClasses #-}
4
5 module SupplyClass
6     ( MonadSupply (..),
7       S.Supply,
8       S.runSupply,
9     )
10    where
```

为了让编译器接受我们的实例声明，必须使用 `FlexibleInstances` 扩展。这个扩展放宽了某些情况下编写实例的正常规则，在某种程度上仍然让编译器的类型检查器保证它被终止。这里还需要使用 `FlexibleInstances` 是因为使用了函数式依赖的关系，不过这些细节超出了本书的范围。

Tip

如何知道哪个语言扩展是被需要的

如果 GHC 因为需要某些语言扩展被开启才能对某部分代码进行编译，它会直接告诉用户需要那些扩展。例如，如果它认为我们的代码需要灵活实例的支持，它会建议我们使用 `-XFlexibleInstances` 选项来进行编译。一个 `-X` 选项与 `LANGUAGE` 的效果一样：开启特定的扩展。

最后请注意，我们将重新导出这个模块中的 `runSupply` 以及 `Supply`。从一个模块导出名称是完全合法的，即使它是在另一个模块中定义的。这个例子中，意味着用户只需要导入 `SupplyClass` 模块，而不需要导入 `Supply` 模块。这减少了用户需要记住的“移动部件”的数量。

对单子接口进行编程

以下是一个简单的函数用于提取 `Supply` 单子的两个值，并格式化它们成为一个字符串进行返回：

```
1 showTwo :: (Show s) => Supply s String
2 showTwo = do
3   a <- next
4   b <- next
5   return $ show "a: " ++ show a ++ ", b: " ++ show b
```

这段代码通过其结果类型绑定到 `Supply` 单子。通过修改函数类型，我们可以很容易的泛化到任何实现 `MonadSupply` 接口的单子。注意函数体保持不变：

```
1 showTwoClass :: (Show s, Monad m, MonadSupply s m) => m String
2 showTwoClass = do
3   a <- next
4   b <- next
5   return $ show "a: " ++ show a ++ ", b: " ++ show b
```

reader 单子

`State` 单子允许我们在代码中探索可变状态。有时我们希望能够传递一些不可变的状态，例如程序的配置数据。我们可以使用 `State` 单子来实现这个目的，但是我们可能会发现自己意外的修改了应该保持不变的数据。

现在考虑一下具有上述期望的特征的函数应该做什么。它应该接受我们传入的数据某种类型 `e` 的值，并返回另一种类型 `a` 的值作为结果，即 `e -> a`。

将该类型转换为一个方便的 `Monad` 实例仅需 `newtype` 包装：

```
1 newtype Reader e a = R {runReader :: e -> a}
```

使其成为 `Monad` 实例并不需要做太多的工作：

```

1 instance Functor (Reader a) where
2   fmap f m = R $ f . runReader m
3
4 instance Applicative (Reader a) where
5   pure = R . const
6   f <*> m = R $ \r -> runReader f r $ runReader m r
7
8 instance Monad (Reader e) where
9   m >>= k = R $ \r -> runReader (k $ runReader m r) r

```

注：原文代码过旧。

我们可以把类型 `e` 看作是计算某个表达式的环境。无论环境是什么，`return` 操作都应该具有相同效果，即忽略它的环境。

`(>>=)` 的定义稍微复杂一些，仅仅是因为需要将环境 – 这里是变量 `r` – 在当前计算与链接的计算上都适用。

在单子中执行的一段代码如何知道它的环境中有什么？仅需 `ask`：

```

1 ask :: Reader e e
2 ask = R id

```

在给定的操作链中，每次调用 `ask` 都将返回相同的值，因为存储在环境中的值不会改变。在 `ghci` 中进行测试：

```

1 ghci> :l SupplyClass.hs
2 [1 of 2] Compiling Supply          ( Supply.hs, interpreted )
3 [2 of 2] Compiling SupplyClass    ( SupplyClass.hs, interpreted )
4 Ok, two modules loaded.
5 ghci> runReader (ask >>= \x -> return (x * 3)) 2
6 6

```

`Reader` 单子包含在标准的 `mtl` 库中，这个库通常与 `GHC` 捆绑在一起。我们可以在 `Control.Monad.Reader` 模块中找到它。这个单子的动机最初看起来可能有点单薄，但是它在复杂的代码中很有用。我们经常要访问程序内部深处的配置信息；将这些信息作为普通参数传递进来，则需要对代码进行痛苦的重构。将这些信息隐藏在单子中，不关心配置信息的中间函数则不需要看到它。

`Reader` 单子的最明确动机将出现在第 18 章单子转换中的合并几个单子来构建一个新的单子的讨论中。这里我们则是看到了如何更好的控制状态，以便我们的代码可以通过 `State` 单子修改一些值，而其他值则通过 `Reader` 单子保持不变。

返回自动推导

现在我们了解了 `Reader` 单子，让我们用它创建 `MonadSupply` typeclass。为了保持示例的简单性，在这里违背 `MonadSupply` 的精神：下一个动作总是返回相同的值，而不是总返回不同的值。

直接将 `Reader` 类型转为 `MonadSupply` 类的实例是一个坏主意，因为这样任何 `Reader` 都可以充当 `MonadSupply`，这通常没有任何意义。

创建一个基于 `Reader` 的 `newtype`，它隐藏了内部使用 `Reader` 的事实。现在必须使类型成为我们所关心的两个 typeclasses 实例。启用了 `GeneralizedNewtypeDeriving` 扩展后，GHC 将会帮我们做大部分的苦力活。

```
1 {-# LANGUAGE FlexibleInstances #-}
2 {-# LANGUAGE GeneralizedNewtypeDeriving #-}
3 {-# LANGUAGE MultiParamTypeClasses #-}
4
5 import Control.Monad
6 import SupplyClass
7
8 newtype MySupply e a = MySupply {runMySupply :: Reader e a}
9     deriving (Functor, Applicative, Monad)
10
11 instance MonadSupply e (MySupply e) where
12     next = MySupply $ Just `liftM` ask
```

注：`deriving Monad` 之前还要加上 `Functor` 和 `Applicative`。

注意我们必须让类型成为 `MonadSupply e` 的一个实例，而不是 `MonadSupply`。如果省略类型变量，编译器会报错。

测试 `MySupply` 类型：

```
1 xy :: (Num s, MonadSupply s m, MonadFail m) => m s
2 xy = do
3     Just x <- next
4     Just y <- next
5     return $ x * y
```

注：与原文不同，这里 HLS 加上了 `MonadFail m` 的约束。

略（原文此处难以理解）。

隐藏 IO 单子

`IO` 单子的优点和缺点是它非常强大。如果谨慎使用类型可以帮助我们避免编程错误，那么 `IO` 单子应该是一个很大的不安来源。因为 `IO` 单子对我们能做的事情没有任何限制，它使我们容易受到各种事故的影响。

我们怎样才能驯服它呢？假设我们想要保证一段代码可以读写本地文件系统上的文件，但它不能访问网络。我们不能使用普通的 `IO` 单子，因为它不会限制我们。

使用 newtype

现在让我们创建一个模块提供一些用于读写文件的功能。首先是创建一个约束版本的 `IO` 将其用 `newtype` 包裹。

```
1 newtype HandleIO a = HandleIO {runHandleIO :: IO a}
2 deriving (Functor, Applicative, Monad)
```

从模块导出类型构造函数和 `runHandleIO` 执行函数，但不导出数据构造函数。这可以防止在 `HandleIO` 单子内运行的代码获得它所包装的 `IO` 单子。

剩下要做的就是包装希望 `monad` 允许的每个动作。这是一个用 `HandleIO` 数据构造函数包装每个 `IO`。

```
1 import qualified System.IO as IO
2
3 openFile :: FilePath -> IO.IOMode -> HandleIO IO.Handle
4 openFile path mode = HandleIO $ IO.openFile path mode
5
6 hClose :: IO.Handle -> HandleIO ()
7 hClose = HandleIO . IO.hClose
8
9 hPutStrLn :: IO.Handle -> String -> HandleIO ()
10 hPutStrLn h s = HandleIO $ IO.hPutStrLn h s
```

现在我们可以使用被约束的 `HandleIO` 单子了：

```
1 safeHello :: FilePath -> HandleIO ()
2 safeHello path = do
3   h <- openFile path IO.WriteMode
4   hPutStrLn h "hello world"
5   hClose h
```

测试：

```
1 ghci> :l HandleIO.hs
2 [1 of 1] Compiling HandleIO          ( HandleIO.hs, interpreted )
3 Ok, one module loaded.
4 ghci> runHandleIO (safeHello "hello_world_101.txt")
```

为意想不到的用途设计

`HandleIO` 单子有一个小但却重要的问题：它没有考虑到我们可能偶尔需要一个逃生舱口的可能性。如果像这样定义一个单子，可能会偶尔需要执行单子不允许的 I/O 操作。

我们这样定义单子的目的是为了更容易在常见情况下编写可靠的代码，而不是让极端情况变得不可能。

`Control.Monad.Trans` 模块定义了一个“标准逃生舱”，`MonadIO` typeclass。它定义了一个单独的函数，`liftIO`，让我们可以将一个 `IO` 操作镶嵌进另一个单子中。

```

1 ghci> :m +Control.Monad.Trans
2 ghci> :i MonadIO
3 type MonadIO :: (* -> *) -> Constraint
4 class Monad m => MonadIO m where
5     liftIO :: IO a -> m a
6     {-# MINIMAL liftIO #-}
7     -- Defined in 'Control.Monad.IO.Class'
8 instance [safe] MonadIO IO -- Defined in 'Control.Monad.IO.Class'

```

这个 typeclass 的实现很简单：仅用构造函数来包装 `IO`。

```

1 import Control.Monad.Trans (MonadIO (..))
2
3 instance MonadIO HandleIO where
4     liftIO = HandleIO

```

通过明智的使用 `liftIO`，我们可以摆脱束缚并在必要时调用 `IO` 操作。

```

1 import System.Directory (removeFile)
2
3 tidyHello :: FilePath -> HandleIO ()
4 tidyHello path = do
5     safeHello path
6     liftIO $ removeFile path

```

Tip

自动派生与 MonadIO

通过将 typeclass 添加到 `HandleIO` 的派生子句中，可以让编译器自动派生 `MonadIO` 实例。实际上在生产代码中，这是常用的策略。我们在这里避免了这点，仅仅是为了将早期材料的呈现与 `MonadIO` 分开。

使用 typeclasses

将 `IO` 隐藏在另一个单子中的缺点就是：它仍然与具体的实现绑定到了一起。如果想要将 `HandleIO` 替换为其他单子，则必须更改使用 `HandleIO` 的每个操作类型。

作为替换方法，我们可以创建一个 typeclass 指定从操作文件的单子中获得接口。

```

1 {-# LANGUAGE FunctionalDependencies #-}
2 {-# LANGUAGE MultiParamTypeClasses #-}
3
4 module MonadHandle
5     ( MonadHandle (..),
6     )
7 where
8

```

```

9  import qualified System.IO as IO
10
11  class (Monad m) => MonadHandle h m | m -> h where
12      openFile :: FilePath -> IO.IOMode -> m h
13      hPutStr  :: h -> String -> m ()
14      hClose   :: h -> m ()
15      hGetContents :: h -> m String
16
17      hPutStrLn :: h -> String -> m ()
18      hPutStrLn h s = hPutStr h s >> hPutStr h "\n"

```

这里我们选择抽象单子的类型和文件句柄的类型。为了满足类型检查器的要求，我们添加了一个函数依赖：对于 `MonadHandle` 的任何实例，只能使用一种句柄类型。当我们将 `IO` 单子作为该类的一个实例时，我们使用一个普通的 `Handle`。

```

1  {-# LANGUAGE FunctionalDependencies #-}
2  {-# LANGUAGE MultiParamTypeClasses #-}
3
4  import Control.Monad.Trans (MonadIO (..), MonadTrans (..))
5  import MonadHandle
6  import SafeHello
7  import System.Directory (removeFile)
8  import System.IO (IOMode (..))
9  import qualified System.IO as IO
10
11  instance MonadHandle IO.Handle IO where
12      openFile = IO.openFile
13      hPutStr  = IO.hPutStr
14      hClose   = IO.hClose
15      hGetContents = IO.hGetContents
16      hPutStrLn = IO.hPutStrLn

```

由于任何 `MonadHandle` 也必须同样是一个 `Monad`，因此可以使用 `do` 来操作文件，而无需关心它最终会在那个单子中执行。

```

1  safeHello :: (MonadHandle h m) => FilePath -> m ()
2  safeHello path = do
3      h <- openFile path WriteMode
4      hPutStrLn h "hello world"
5      hClose h

```

因为我们让 `IO` 成为这个 typeclass 的一个实例，所以可以从 `ghci` 执行这个动作：

```

1  ghci> :l MonadHandleIO.hs
2  [1 of 2] Compiling MonadHandle      ( MonadHandle.hs, interpreted )
3  [2 of 2] Compiling MonadHandleIO   ( MonadHandleIO.hs, interpreted )
4  Ok, two modules loaded.
5  ghci> safeHello "hello to my fans in domestic surveillance"
6  ghci> removeFile "hello to my fans in domestic surveillance"

```

typeclass 方法的美妙之处在于，我们可以在不涉及太多代码的情况下将一个底层单子交换为另一个单子，因为大多数代码都不知道或不关心实现。例如，可以用某个写入文件时压缩文件的单子来代替 `IO`，

通过 typeclass 定义单子的接口还有一个好处。它允许其他人将我们的实现隐藏在新的类型包装器中，并自动派生他们想要公开的 typeclass 实例。

隔离与测试

实际上，由于 `safeHello` 函数并没有使用 `IO` 类型，我们甚至可以用一个无法执行 I/O 的单子。这允许我们在一个完全纯净、可控的环境中测试通常会产生副作用的代码。

为此我们将创建一个无法执行 I/O 的单子，不过将会记录每个文件相关的事件。

```
1 data Event
2   = Open FilePath IOMode
3   | Put String String
4   | Close String
5   | GetContents String
6   deriving (Show)
```

我们在之前的章节中开发了 `Logger` 类型，不过这里将使用标准且更泛用的 `Writer` 单子。正如其它 `mtl` 单子，由 `Writer` 提供的 API 定义在一个 typeclass 中，即 `MonadWriter`。它最有用的方法是 `tell`，即记录一个值。

```
1 ghci> :m +Control.Monad.Writer
2 ghci> :type tell
3 tell :: (MonadWriter w m) => w -> m ()
```

我们记录的值可以是任何 `Monoid` 类型。由于列表类型是一个 `Monoid`，那么记录到 `Event` 列表。

我们可以让 `Writer [Event]` 成为 `MonadHandle` 的一个实例，但它更便宜，更容易，也更安全。

注：书中遗漏了 `MonadHandle` 的 `WriterIO` 实例：

```
1 {-# LANGUAGE FlexibleInstances #-}
2 {-# LANGUAGE MultiParamTypeClasses #-}
3
4 import MonadHandle (MonadHandle (..))
5 import MonadHandleIO
6
7 instance MonadHandle FilePath WriterIO where
8   openFile path mode = tell [Open path mode] >> return path
9   hPutStr h str = tell [Put h str]
10  hClose h = tell [Close h]
11  hGetContents h = tell [GetContents h] >> return "fake contents"
```

通过 `ghci` 测试：

```

1 ghci> :l WriterIO.hs
2 [1 of 3] Compiling MonadHandle      ( MonadHandle.hs, interpreted )
3 [2 of 3] Compiling MonadHandleIO    ( MonadHandleIO.hs, interpreted )
4 [3 of 3] Compiling WriterIO         ( WriterIO.hs, interpreted )
5 Ok, three modules loaded.
6 ghci> runWriterIO (safeHello "foo")
7 ((),[Open "foo" WriteMode,Put "foo" "hello world",Put "foo" "\n",Close "foo"])

```

writer 单子与列表

每次使用 `tell` 时，writer 单子都会使用单子的 `mappend` 函数，由于列表的 `mappend` 是 `(++)`，因此列表不是与 `Writer` 一起使用的使用选择：重复追加的代价很高。上面的示例使用列表存粹是为了简单。

在生产代码中，如果使用 `Writer` 单子同时需要一个类似列表这样的方式记录日志，那么请使用更高效的追加特征的容器。这里有一种选择就是之前介绍的差异列表（十三章）。我们无需自己造轮子，而是可以在 Haskell 的库中寻找一个合适的替代。另一种方案就是使用由 `Data.Sequence` 模块所提供的 `Seq` 类型（十三章）。

任意 I/O 的重新访问

如果使用 `typeclass` 方式来限制 `IO`，我们可能仍然希望保留执行任意 I/O 操作的能力。我们可以尝试在 `typeclass` 上添加 `MonadIO` 作为约束。

```

1 class (MonadHandle h m, MonadIO m) => MonadHandleIO h m | m -> h
2
3 instance MonadHandleIO Handle IO
4
5 tidierHello :: (MonadHandleIO h m) => FilePath -> m ()
6 tidierHello path = do
7     safeHello path
8     liftIO $ removeFile path

```

不过这种方式存在一个问题：添加的 `MonadIO` 约束使我们失去了在纯环境中测试代码的能力，因为我们不能再判断测试是否可能具有破坏性的副作用。另一种方法是将这个约束从 `typeclass`（它影响所有函数）移到那些真正需要执行 I/O 的函数。

```

1 tidyHello :: (MonadIO m, MonadHandle h m) => FilePath -> m ()
2 tidyHello path = do
3     safeHello path
4     liftIO $ removeFile path

```

我们可以对缺少 `MonadIO` 约束的函数使用纯函数式测试，而对其余的函数使用传统的单元测试。

不幸的是，我们用一个问题替代了另一个问题：我们不能从单独具有 `MonadHandle` 约束的代码中调用具有 `MonadIO` 和 `MonadHandle` 约束的代码。如果我们发现在 `MonadHandle` 代码深处的某个地方，我们确实需要 `MonadIO` 约束，我们必须将它添加到通向这一点的所有代码路径中。

允许任意 I/O 是有风险的，并且会对我们如何开发和测试代码产生深远的影响。当我们必须在宽容和更容易的推理和测试之间做出选择时，我们通常会选择后者。

16 The Parsec parsing library

WIP

17 The foreign function interface

WIP

18 Monad transformers

WIP

19 Error handling

WIP

20 Systems programming

WIP

21 Working with databases

WIP

22 Web client programming

WIP

23 GUI programming

WIP

24 Basic concurrent and parallel programming

WIP

25 Profiling and tuning for performance

WIP

26 Advanced library design: building a Bloom filter

WIP

27 Network programming

WIP

28 Software transactional memory

WIP